

Theorem Proving in Lean

Jeremy Avigad
Leonardo de Moura
Soonho Kong

Version **f931b84**, updated at 2016-12-12 21:06:27 -0500

Contents

Contents	3
1 Introduction	6
1.1 Computers and Theorem Proving	6
1.2 About Lean	7
1.3 About this Book	7
1.4 Acknowledgments	8
2 Dependent Type Theory	9
2.1 Simple Type Theory	9
2.2 Types as Objects	11
2.3 Function Abstraction and Evaluation	13
2.4 Introducing Definitions	17
2.5 Local Definitions	18
2.6 Variables and Sections	19
2.7 Namespaces	21
2.8 Dependent Types	23
2.9 Implicit Arguments	26
3 Propositions and Proofs	30
3.1 Propositions as Types	30
3.2 Working with Propositions as Types	33
3.3 Propositional Logic	36
3.4 Introducing Auxiliary Subgoals	41
3.5 Classical Logic	42
3.6 Examples of Propositional Validities	43
4 Quantifiers and Equality	46
4.1 The Universal Quantifier	46
4.2 Equality	50

4.3	Calculational Proofs	52
4.4	The Existential Quantifier	55
4.5	More on the Proof Language	59
5	Tactics	62
5.1	Entering Tactic Mode	62
5.2	Basic Tactics	65
5.3	More tactics	70
5.4	Structuring Tactic Proofs	72
5.5	Rewriting and the Simplifier	77
6	Interacting with Lean	82
6.1	Displaying Information	82
6.2	Setting Options	84
6.3	Using the Library	85
6.4	Using Lean with Emacs	87
6.5	Imports, Object Files, and Projects	89
6.6	Notation and Abbreviations	91
7	Inductive Types	93
7.1	Enumerated Types	94
7.2	Constructors with Arguments	97
7.3	Inductively Defined Propositions	101
7.4	Defining the Natural Numbers	102
7.5	Tactics	105
7.6	Other Inductive Types	108
8	Induction and Recursion	111
8.1	Pattern Matching	111
8.2	Structural Recursion and Induction	113
8.3	Dependent Pattern-Matching	114
8.4	Variations on Pattern Matching	116
8.5	Inaccessible Terms	117
8.6	Match Expressions	118
8.7	Well-Founded Recursion	119
9	Structures and Records	120
9.1	Declaring Structures	120
9.2	Objects	122
9.3	Inheritance	124
10	Type Classes	125

10.1 Type Classes and Instances	125
10.2 Chaining Instances	129
10.3 Decidable Propositions	129
10.4 Overloading with Type Classes	131
10.5 Managing Type Class Inference	132
Bibliography	134

Introduction

1.1 Computers and Theorem Proving

Formal verification involves the use of logical and computational methods to establish claims that are expressed in precise mathematical terms. These can include ordinary mathematical theorems, as well as claims that pieces of hardware or software, network protocols, and mechanical and hybrid systems meet their specifications. In practice, there is not a sharp distinction between verifying a piece of mathematics and verifying the correctness of a system: formal verification requires describing hardware and software systems in mathematical terms, at which point establishing claims as to their correctness becomes a form of theorem proving. Conversely, the proof of a mathematical theorem may require a lengthy computation, in which case verifying the truth of the theorem requires verifying that the computation does what it is supposed to do.

The gold standard for supporting a mathematical claim is to provide a proof, and twentieth-century developments in logic show most if not all conventional proof methods can be reduced to a small set of axioms and rules in any of a number of foundational systems. With this reduction, there are two ways that a computer can help establish a claim: it can help find a proof in the first place, and it can help verify that a purported proof is correct.

Automated theorem proving focuses on the “finding” aspect. Resolution theorem provers, tableau theorem provers, fast satisfiability solvers, and so on provide means of establishing the validity of formulas in propositional and first-order logic. Other systems provide search procedures and decision procedures for specific languages and domains, such as linear or nonlinear expressions over the integers or the real numbers. Architectures like SMT (“satisfiability modulo theories”) combine domain-general search methods with domain-

specific procedures. Computer algebra systems and specialized mathematical software packages provide means of carrying out mathematical computations, establishing mathematical bounds, or finding mathematical objects. A calculation can be viewed as a proof as well, and these systems, too, help establish mathematical claims.

Automated reasoning systems strive for power and efficiency, often at the expense of guaranteed soundness. Such systems can have bugs, and it can be difficult to ensure that the results they deliver are correct. In contrast, *interactive theorem proving* focuses on the “verification” aspect of theorem proving, requiring that every claim is supported by a proof in a suitable axiomatic foundation. This sets a very high standard: every rule of inference and every step of a calculation has to be justified by appealing to prior definitions and theorems, all the way down to basic axioms and rules. In fact, most such systems provide fully elaborated “proof objects” that can be communicated to other systems and checked independently. Constructing such proofs typically requires much more input and interaction from users, but it allows us to obtain deeper and more complex proofs.

The *Lean Theorem Prover* aims to bridge the gap between interactive and automated theorem proving, by situating automated tools and methods in a framework that supports user interaction and the construction of fully specified axiomatic proofs. The goal is to support both mathematical reasoning and reasoning about complex systems, and to verify claims in both domains.

1.2 About Lean

The *Lean* project was launched by Leonardo de Moura at Microsoft Research Redmond in 2012. It is an ongoing, long-term effort, and much of the potential for automation will be realized only gradually over time. Lean is released under the Apache 2.0 license, a permissive open source license that permits others to use and extend the code and mathematical libraries freely.

There are currently two ways to use Lean. The first is to run it from the web: a Javascript version of Lean, a standard library of definitions and theorems, and an editor are actually downloaded to your browser and run there. This provides a quick and convenient way to begin experimenting with the system.

The second way to use Lean is to install and run it natively on your computer. The native version is much faster than the web version, and is more flexible in other ways, too. It comes with an Emacs mode that offers powerful support for writing and debugging proofs, and is much better suited for serious use.

1.3 About this Book

This book is designed to teach you to develop and verify proofs in Lean. Much of the background information you will need in order to do this is not specific to Lean at all. To

start with, we will explain the logical system that Lean is based on, a version of *dependent type theory* that is powerful enough to prove almost any conventional mathematical theorem, and expressive enough to do it in a natural way. More specifically, Lean is based on a variant of a system known as the *Calculus of Inductive Constructions*[1, 3], or *CIC*. We will explain not only how to define mathematical objects and express mathematical assertions in dependent type theory, but also how to use it as a language for writing proofs.

Because fully detailed axiomatic proofs are so complicated, the challenge of theorem proving is to have the computer fill in as many of the details as possible. We will describe various methods to support this in dependent type theory. For example, we will discuss term rewriting, and Lean’s automated methods for simplifying terms and expressions automatically. Similarly, we will discuss methods of *elaboration* and *type inference*, which can be used to support flexible forms of algebraic reasoning.

Finally, of course, we will discuss features that are specific to Lean, including the language with which you can communicate with the system, and the mechanisms Lean offers for managing complex theories and data.

If you are reading this book within Lean’s online tutorial system, you will see a copy of the Lean editor at right, with an output buffer beneath it. At any point, you can type things into the editor, press the “play” button, and see Lean’s response. Notice that you can resize the various windows if you would like.

Throughout the text you will find examples of Lean code like the one below:

```
theorem and_commutative (p q : Prop) : p ∧ q → q ∧ p :=
  assume hpq : p ∧ q,
  have hp : p, from and.left hpq,
  have hq : q, from and.right hpq,
  show q ∧ p, from and.intro hq hp
```

Once again, if you are reading the book online, you will see a button that reads “try it yourself.” Pressing the button copies the example into the Lean editor with enough surrounding context to make the example compile correctly, and then runs Lean. We recommend running the examples and experimenting with the code on your own as you work through the chapters that follow.

1.4 Acknowledgments

This tutorial is an open access project maintained on Github. Many people have contributed to the effort, providing corrections, suggestions, examples, and text. We are grateful to Ulrik Buchholz, Nathan Carter, Amine Chaieb, Floris van Doorn, Anthony Hart, Sean Leather, Christopher John Mazey, Assia Mahboubi, Sebastian Ullrich, Daniel Velleman, and Théo Zimmerman for their contributions, and we apologize to those whose names we have inadvertently omitted.

Dependent Type Theory

Dependent type theory is a powerful and expressive language, allowing us to express complex mathematical assertions, write complex hardware and software specifications, and reason about both of these in a natural and uniform way. Lean is based on a version of dependent type theory known as the *Calculus of Inductive Constructions*, with a countable hierarchy of non-cumulative universes and inductive types. By the end of this chapter, you will understand much of what this means.

2.1 Simple Type Theory

As a foundation for mathematics, set theory has a simple ontology that is rather appealing. Everything is a set, including numbers, functions, triangles, stochastic processes, and Riemannian manifolds. It is a remarkable fact that one can construct a rich mathematical universe from a small number of axioms that describe a few basic set-theoretic constructions.

But for many purposes, including formal theorem proving, it is better to have an infrastructure that helps us manage and keep track of the various kinds of mathematical objects we are working with. “Type theory” gets its name from the fact that every expression has an associated *type*. For example, in a given context, $x + 0$ may denote a natural number and f may denote a function on the natural numbers.

Here are some examples of how we can declare objects in Lean and check their types.

```
/- declare some constants -/

constant m : nat      -- m is a natural number
constant n : nat
```

```

constants b1 b2 : bool -- declare two constants at once

/- check their types -/

check m          -- output: nat
check n
check n + 0       -- nat
check m * (n + 0) -- nat
check b1          -- bool
check b1 && b2     -- "∧" is boolean and
check b1 || b2    -- boolean or
check tt         -- boolean "true"

-- Try some examples of your own.

```

The `/-` and `-/` annotations indicate that the next line is a comment block that is ignored by Lean. Similarly, two dashes indicate that the rest of the line contains a comment that is also ignored. Comment blocks can be nested, making it possible to “comment out” chunks of code, just as in many programming languages.

The `constant` and `constants` commands introduce new constant symbols into the working environment, and the `check` command asks Lean to report their types. You should test this, and try typing some examples of your own. Declaring new objects in this way is a good way to experiment with the system, but it is ultimately undesirable: Lean is a foundational system, which is to say, it provides us with powerful mechanisms to *define* all the mathematical objects we need, rather than simply postulating them to the system. We will explore these mechanisms in the chapters to come.

What makes simple type theory powerful is that one can build new types out of others. For example, if α and β are types, $\alpha \rightarrow \beta$ denotes the type of functions from α to β , and $\alpha \times \beta$ denotes the cartesian product, that is, the type of ordered pairs consisting of an element of α paired with an element of β .

```

constants m n : nat

constant f : nat → nat          -- type the arrow as "\to" or "\r"
constant f' : nat -> nat        -- alternative ASCII notation
constant f'' : ℕ → ℕ            -- \nat is alternative notation for nat
constant p : nat × nat          -- type the product as "\times"
constant q : prod nat nat       -- alternative notation
constant g : nat → nat → nat
constant g' : nat → (nat → nat) -- has the same type as g!
constant h : nat × nat → nat

constant F : (nat → nat) → nat -- a "functional"

check f          -- ℕ → ℕ
check f n         -- ℕ
check g m n       -- ℕ
check g m         -- ℕ → ℕ
check (m, n)      -- ℕ × ℕ
check p.1         -- ℕ

```

```

check p.2          -- N
check (m, n).1      -- N
check (p.1, n)      -- N × N
check F f           -- N

-- Try it on your own: write down some types, declare some constants,
-- and check some expressions.

```

Let us dispense with some basic syntax. You can enter the unicode arrow \rightarrow by typing `\to` or `\r`. You can also use the ASCII alternative `->`, so that the expression `nat -> nat` and `nat → nat` mean the same thing. Both expressions denote the type of functions that take a natural number as input and return a natural number as output. The symbol \mathbb{N} is alternative unicode notation for `nat`; you can enter it by typing `\nat`. The unicode symbol \times for the cartesian product is entered `\prod`. We will generally use lower-case greek letters like α , β , and γ to range over types. You can enter these particular ones with `\a`, `\b`, and `\g`.

There are a few more things to notice here. First, the application of a function `f` to a value `x` is denoted `f x`. Second, when writing type expressions, arrows associate to the *right*; for example, the type of `g` is `nat → (nat → nat)`. Thus we can view `g` as a function that takes natural numbers and returns another function that takes a natural number and returns a natural number. In type theory, this is generally more convenient than writing `g` as a function that takes a pair of natural numbers as input, and returns a natural number as output. For example, it allows us to “partially apply” the function `g`. The example above shows that `g m` has type `nat → nat`, that is, the function that “waits” for a second argument, `n`, and then returns `g m n`. Taking a function `h` of type `nat × nat → nat` and “redefining” it to look like `g` is a process known as *currying*, something we will come back to below.

By now you may also have guessed that, in Lean, `(m, n)` denotes the ordered pair of `m` and `n`, and if `p` is a pair, `fst p` and `snd p` denote the two projections.

2.2 Types as Objects

One way in which Lean’s dependent type theory extends simple type theory is that types themselves – entities like `nat` and `bool` – are first-class citizens, which is to say that they themselves are objects of study. For that to be the case, each of them also has to have a type.

```

check nat          -- Type
check bool         -- Type
check nat → bool   -- Type
check nat × bool   -- Type
check nat → nat     -- ...
check nat × nat → nat
check nat → nat → nat

```

```

check nat → (nat → nat)
check nat → nat → bool
check (nat → nat) → nat

```

We see that each one of the expressions above is an object of type `Type`. We can also declare new constants and constructors for types:

```

constants α β : Type
constant F : Type → Type
constant G : Type → Type → Type

check α          -- Type
check F α        -- Type
check F nat      -- Type
check G α        -- Type → Type
check G α β      -- Type
check G α nat    -- Type

```

Indeed, we have already seen an example of a function of type `Type → Type → Type`, namely, the Cartesian product.

```

constants α β : Type

check prod α β    -- Type
check prod nat nat -- Type

```

Here is another example: given any type α , the type `list α` denotes the type of lists of elements of type α .

```

constant α : Type

check list α    -- Type
check list nat  -- Type

```

For those more comfortable with set-theoretic foundations, it may be helpful to think of a type as nothing more than a set, in which case, the elements of the type are just the elements of the set. Given that every expression in Lean has a type, it is natural to ask: what type does `Type` itself have?

```

check Type    -- Type2

```

We have actually come up against one of the most subtle aspects of Lean's typing system. Lean's underlying foundation has an infinite hierarchy of types:

```

check Type 1 -- Type2
check Type 2 -- Type3
check Type 3 -- Type4
check Type 4 -- Type5

```

Think of `Type 1` as a universe of “small” or “ordinary” types. `Type 2` is then a larger universe of types, which contains `Type 1` as an element, and `Type 3` is an even larger universe of types, which contains `Type 2` as an element. The list is indefinite, so that there is a `Type n` for every natural number `n`. Lean introduces abbreviations for the first three levels:

```
check Type    -- same as Type 1
check Type2  -- same as Type 2
check Type3  -- same as Type 3
```

It is rare to have to use more than those. There is also a `Type 0`, which is also denoted `Prop`. This type has special properties, and will be discussed in the next chapter.

We want some operations, however, to be *polymorphic* over type universes. For example, `list α` should make sense for any type `α`, no matter which type universe `α` lives in. This explains the type annotation of the function `list`:

```
check list    -- Type u_1 → Type (max 1 u_1)
```

Here `u_1` is a variable ranging over type levels. The output of the `check` command means that whenever `α` has type `Type n`, `list α` also has type `Type n` if `n` is at least 1, and has `Type 1` if `α` has type 0. The function `prod` is similarly polymorphic:

```
check prod    -- Type u_1 → Type u_2 → Type (max 1 u_1 u_2)
```

To define polymorphic constants and variables, Lean allows us to declare universe variables explicitly:

```
universe variable u
constant α : Type u
check α
```

Throughout this book, you will see us do this in examples when we want type constructions to have as much generality as possible. We will see that the ability to treat type constructors as instances of ordinary mathematical functions is a powerful feature of dependent type theory.

2.3 Function Abstraction and Evaluation

We have seen that if we have `m n : nat`, then we have `(m, n) : nat × nat`. This gives us a way of creating pairs of natural numbers. Conversely, if we have `p : nat × nat`, then we have `fst p : nat` and `snd p : nat`. This gives us a way of “using” a pair, by extracting its two components.

We already know how to “use” a function $f : \alpha \rightarrow \beta$, namely, we can apply it to an element $a : \alpha$ to obtain $f a : \beta$. But how do we create a function from another expression?

The companion to application is a process known as “abstraction,” or “lambda abstraction.” Suppose that by temporarily postulating a variable $x : \alpha$ we can construct an expression $t : \beta$. Then the expression `fun x : α , t`, or, equivalently, $\lambda x : \alpha, t$, is an object of type $\alpha \rightarrow \beta$. Think of this as the function from α to β which maps any value x to the value t , which depends on x . For example, in mathematics it is common to say “let f be the function which maps any natural number x to $x + 5$.” The expression $\lambda x : \text{nat}, x + 5$ is just a symbolic representation of the right-hand side of this assignment.

```
check fun x : nat, x + 5
check λ x : nat, x + 5
```

Here are some more abstract examples:

```
constants α β : Type
constants a1 a2 : α
constants b1 b2 : β

constant f : α → α
constant g : α → β
constant h : α → β → α
constant p : α → α → bool

check fun x : α, f x           -- α → α
check λ x : α, f x             -- α → α
check λ x : α, f (f x)         -- α → α
check λ x : α, h x b1          -- α → α
check λ y : β, h a1 y          -- β → α
check λ x : α, p (f (f x)) (h (f a1) b2) -- α → bool
check λ x : α, λ y : β, h (f x) y -- α → β → α
check λ (x : α) (y : β), h (f x) y -- α → β → α
check λ x y, h (f x) y         -- α → β → α
```

Lean interprets the final three examples as the same expression; in the last expression, Lean infers the type of x and y from the types of f and h .

Be sure to try writing some expressions of your own. Some mathematically common examples of operations of functions can be described in terms of lambda abstraction:

```
constants α β γ : Type
constant f : α → β
constant g : β → γ
constant b : β

check λ x : α, x           -- α → α
check λ x : α, b           -- α → β
check λ x : α, g (f x)     -- α → γ
check λ x, g (f x)
```

```

-- we can abstract any of the constants in the previous definitions

check λ b : β, λ x : α, x      -- β → α → α
check λ (b : β) (x : α), x    -- equivalent to the previous line
check λ (g : β → γ) (f : α → β) (x : α), g (f x)
                                -- (β → γ) → (α → β) → α → γ
-- we can even abstract over the type

check λ (α β : Type) (b : β) (x : α), x
check λ (α β γ : Type) (g : β → γ) (f : α → β) (x : α), g (f x)

```

Think about what these expressions mean. The expression $\lambda x : \alpha, x$ denotes the identity function on α , the expression $\lambda x : \alpha, b$ denotes the constant function that always returns b , and $\lambda x : \alpha, g (f x)$, denotes the composition of f and g . We can, in general, leave off the type annotations on the variable and let Lean infer it for us. So, for example, we can write $\lambda x, g (f x)$ instead of $\lambda x : \alpha, g (f x)$.

We can abstract over any of the constants in the previous definitions:

```

check λ b : β, λ x : α, x      -- β → α → α
check λ (b : β) (x : α), x    -- β → α → α
check λ (g : β → γ) (f : α → β) (x : α), g (f x)
                                -- (β → γ) → (α → β) → α → γ

```

Lean lets us combine lambdas, so the second example is equivalent to the first. We can even abstract over the type:

```

check λ (α β : Type) (b : β) (x : α), x
check λ (α β γ : Type) (g : β → γ) (f : α → β) (x : α), g (f x)

```

The last expression, for example, denotes the function that takes three types, α , β , and γ , and two functions, $g : \beta \rightarrow \gamma$ and $f : \alpha \rightarrow \beta$, and returns the composition of g and f . (Making sense of the type of this function requires an understanding of dependent products, which we will explain below.) Within a lambda expression $\lambda x : \alpha, t$, the variable x is a “bound variable”: it is really a placeholder, whose “scope” does not extend beyond t . For example, the variable b in the expression $\lambda (b : \beta) (x : \alpha), x$ has nothing to do with the constant b declared earlier. In fact, the expression denotes the same function as $\lambda (u : \beta) (z : \alpha), z$. Formally, the expressions that are the same up to a renaming of bound variables are called *alpha equivalent*, and are considered “the same.” Lean recognizes this equivalence.

Notice that applying a term $t : \alpha \rightarrow \beta$ to a term $s : \alpha$ yields an expression $t s : \beta$. Returning to the previous example and renaming bound variables for clarity, notice the types of the following expressions:

```

constants  $\alpha$   $\beta$   $\gamma$  : Type
constant f :  $\alpha \rightarrow \beta$ 
constant g :  $\beta \rightarrow \gamma$ 
constant h :  $\alpha \rightarrow \alpha$ 
constants (a :  $\alpha$ ) (b :  $\beta$ )

check (λ x :  $\alpha$ , x) a           --  $\alpha$ 
check (λ x :  $\alpha$ , b) a           --  $\beta$ 
check (λ x :  $\alpha$ , b) (h a)       --  $\beta$ 
check (λ x :  $\alpha$ , g (f x)) (h (h a)) --  $\gamma$ 

check (λ (v :  $\beta \rightarrow \gamma$ ) (u :  $\alpha \rightarrow \beta$ ) x, v (u x)) g f a --  $\gamma$ 

check (λ (Q R S : Type) (v :  $R \rightarrow S$ ) (u :  $Q \rightarrow R$ ) (x : Q),
      v (u x))  $\alpha$   $\beta$   $\gamma$  g f a --  $\gamma$ 

```

As expected, the expression $(\lambda x : \alpha, x) a$ has type α . In fact, more should be true: applying the expression $(\lambda x : \alpha, x)$ to a should “return” the value a . And, indeed, it does:

```

constants  $\alpha$   $\beta$   $\gamma$  : Type
constant f :  $\alpha \rightarrow \beta$ 
constant g :  $\beta \rightarrow \gamma$ 
constant h :  $\alpha \rightarrow \alpha$ 
constants (a :  $\alpha$ ) (b :  $\beta$ )

eval (λ x :  $\alpha$ , x) a           -- a
eval (λ x :  $\alpha$ , b) a           -- b
eval (λ x :  $\alpha$ , b) (h a)       -- b
eval (λ x :  $\alpha$ , g (f x)) a     -- g (f a)

eval (λ (v :  $\beta \rightarrow \gamma$ ) (u :  $\alpha \rightarrow \beta$ ) x, v (u x)) g f a -- g (f a)

eval (λ (Q R S : Type) (v :  $R \rightarrow S$ ) (u :  $Q \rightarrow R$ ) (x : Q),
      v (u x))  $\alpha$   $\beta$   $\gamma$  g f a -- g (f a)

```

The command `eval` tells Lean to *evaluate* an expression. The process of simplifying an expression $(\lambda x, t)s$ to $t[s/x]$ – that is, t with s substituted for the variable x – is known as *beta reduction*, and two terms that beta reduce to a common term are called *beta equivalent*. But the `eval` command carries out other forms of reduction as well:

```

constants m n : nat
constant b : bool

print "reducing pairs"
eval (m, n).1 -- m
eval (m, n).2 -- n

print "reducing boolean expressions"
eval tt && ff -- ff
eval b && ff  -- ff

```

```
print "reducing arithmetic expressions"
eval n + 0      -- n
eval n + 2      -- succ (succ n)
eval 2 + 3      -- 5
```

In a later chapter, we will explain how these terms are evaluated. For now, we only wish to emphasize that this is an important feature of dependent type theory: every term has a computational behavior, and supports a notion of reduction, or *normalization*. In principle, two terms that reduce to the same value are called *definitionally equal*. They are considered “the same” by the underlying logical framework, and Lean does its best to recognize and support these identifications.

2.4 Introducing Definitions

As we have noted above, declaring constants in the Lean environment is a good way to postulate new objects to experiment with, but most of the time what we really want to do is *define* objects in Lean and prove things about them. The `definition` command provides one important way of defining new objects.

```
definition foo : (ℕ → ℕ) → ℕ := λ f, f 0

check foo      -- ℕ
print foo      -- λ (f : ℕ → ℕ), f 0
```

We can omit the type when Lean has enough information to infer it:

```
definition foo' := λ f : ℕ → ℕ, f 0
```

The general form of a definition is `definition foo : T := bar`. Lean can usually infer the type `T`, but it is often a good idea to write it explicitly. This clarifies your intention, and Lean will flag an error if the right-hand side of the definition does not have the right type.

Because function definitions are so common, Lean provides the shorthand `def` for `definition`, and an alternative notation, which puts the abstracted variables before the colon and omits the lambda:

```
def double (x : ℕ) : ℕ := x + x
print double
check double 3
eval double 3      -- 6

def square (x : ℕ) := x * x
print square
check square 3
eval square 3      -- 9
```

```
def do_twice (f :  $\mathbb{N} \rightarrow \mathbb{N}$ ) (x :  $\mathbb{N}$ ) :  $\mathbb{N}$  := f (f x)

eval do_twice double 2    -- 8
```

These definitions are equivalent to the following:

```
def double :  $\mathbb{N} \rightarrow \mathbb{N}$  :=  $\lambda x, x + x$ 
def square :  $\mathbb{N} \rightarrow \mathbb{N}$  :=  $\lambda x, x * x$ 
def do_twice : ( $\mathbb{N} \rightarrow \mathbb{N}$ )  $\rightarrow \mathbb{N} \rightarrow \mathbb{N}$  :=  $\lambda f x, f (f x)$ 
```

We can even use this approach to specify arguments that are types:

```
def compose ( $\alpha \beta \gamma$  : Type) (g :  $\beta \rightarrow \gamma$ ) (f :  $\alpha \rightarrow \beta$ ) (x :  $\alpha$ ) :  $\gamma$  :=
g (f x)
```

As an exercise, we encourage you to use `do_twice` and `double` to define functions that quadruple their input, and multiply the input by 8. As a further exercise, we encourage you to try defining a function `Do_Twice` : $((\mathbb{N} \rightarrow \mathbb{N}) \rightarrow (\mathbb{N} \rightarrow \mathbb{N})) \rightarrow (\mathbb{N} \rightarrow \mathbb{N}) \rightarrow (\mathbb{N} \rightarrow \mathbb{N})$ which iterates *its* argument twice, so that `Do_Twice do_twice` a function which iterates *its* input four times, and evaluate `Do_Twice do_twice double 2`.

Above, we discussed the process of “currying” a function, that is, taking a function `f` (`a`, `b`) that takes an ordered pair as an argument, and recasting it as a function `f' a b` that takes two arguments successively. As another exercise, we encourage you to complete the following definitions, which “curry” and “uncurry” a function.

```
def curry ( $\alpha \beta \gamma$  : Type) (f :  $\alpha \times \beta \rightarrow \gamma$ ) :  $\alpha \rightarrow \beta \rightarrow \gamma$  := sorry

def uncurry ( $\alpha \beta \gamma$  : Type) (f :  $\alpha \rightarrow \beta \rightarrow \gamma$ ) :  $\alpha \times \beta \rightarrow \gamma$  := sorry
```

2.5 Local Definitions

Lean also allows you to introduce “local” definitions using the `let` construct. The expression `let a := t1 in t2` is definitionally equal to the result of replacing every occurrence of `a` in `t2` by `t1`.

```
check let y := 2 + 2 in y * y    --  $\mathbb{N}$ 
eval  let y := 2 + 2 in y * y    -- 16

def t (x :  $\mathbb{N}$ ) :  $\mathbb{N}$  :=
let y := x + x in y * y

eval t 2    -- 16
```

Here, `t` is definitionally equal to the term `(x + x) * (x + x)`. You can combine multiple assignments in a single `let` statement:

```
check let y := 2 + 2, z := y + y in z * z -- 16
eval  let y := 2 + 2, z := y + y in z * z -- 64
```

Notice that the meaning of the expression `let a := t1 in t2` is very similar to the meaning of `(λ a, t2) t1`, but the two are not the same. In the first expression, you should think of every instance of `a` in `t2` as a syntactic abbreviation for `t1`. In the second expression, `a` is a variable, and the expression `λ a, t2` has to make sense independently of the value of `a`. The `let` construct is a stronger means of abbreviation, and there are expressions of the form `let a := t1 in t2` that cannot be expressed as `(λ a, t2) t1`. As an exercise, try to understand why the definition of `foo` below type checks, but the definition of `bar` does not.

```
def foo := let a := nat in λ x : a, x + 2

/-
def bar := (λ a, λ x : a, x + 2) nat
-/
```

2.6 Variables and Sections

This is a good place to introduce some organizational features of Lean that are not a part of the axiomatic framework *per se*, but make it possible to work in the framework more efficiently.

We have seen that the `constant` command allows us to declare new objects, which then become part of the global context. Declaring new objects in this way is somewhat crass. Lean enables us to *define* all of the mathematical objects we need, and *declaring* new objects willy-nilly is therefore somewhat lazy. In the words of Bertrand Russell, it has all the advantages of theft over honest toil. We will see in the next chapter that it is also somewhat dangerous: declaring a new constant is tantamount to declaring an axiomatic extension of our foundational system, and may result in inconsistency.

So far, in this tutorial, we have used the `constant` command to create “arbitrary” objects to work with in our examples. For example, we have declared types α , β , and γ to populate our context. This can be avoided, using implicit or explicit lambda abstraction in our definitions to declare such objects “locally”:

```
def compose (α β γ : Type) (g : β → γ) (f : α → β) (x : α) :
  γ := g (f x)

def do_twice (α : Type) (h : α → α) (x : α) : α := h (h x)
```

```
def do_thrice (α : Type) (h : α → α) (x : α) : α := h (h (h x))
```

Repeating declarations in this way can be tedious, however. Lean provides us with the `variable` and `variables` commands to make such declarations look global:

```
variables (α β γ : Type)

def compose (g : β → γ) (f : α → β) (x : α) : γ := g (f x)
def do_twice (h : α → α) (x : α) : α := h (h x)
def do_thrice (h : α → α) (x : α) : α := h (h (h x))
```

We can declare variables of any type, not just `Type` itself:

```
variables (α β γ : Type)
variables (g : β → γ) (f : α → β) (h : α → α)
variable x : α

def compose := g (f x)
def do_twice := h (h x)
def do_thrice := h (h (h x))

print compose
print do_twice
print do_thrice
```

Printing them out shows that all three groups of definitions have exactly the same effect.

The `variable` and `variables` commands look like the `constant` and `constants` commands we have used above, but there is an important difference: rather than creating permanent entities, the declarations simply tell Lean to insert the variables as bound variables in definitions that refer to them. Lean is smart enough to figure out which variables are used explicitly or implicitly in a definition. We can therefore proceed as though α , β , γ , g , f , h , and x are fixed objects when we write our definitions, and let Lean abstract the definitions for us automatically.

When declared in this way, a variable stays in scope until the end of the file we are working on, and we cannot declare another variable with the same name. Sometimes, however, it is useful to limit the scope of a variable. For that purpose, Lean provides the notion of a `section`:

```
section useful
  variables (α β γ : Type)
  variables (g : β → γ) (f : α → β) (h : α → α)
  variable x : α

  def compose := g (f x)
  def do_twice := h (h x)
  def do_thrice := h (h (h x))
end useful
```

When the section is closed, the variables go out of scope, and become nothing more than a distant memory.

You do not have to indent the lines within a section, since Lean treats any blocks of returns, spaces, and tabs equivalently as whitespace. Nor do you have to name a section, which is to say, you can use an anonymous `section / end` pair. If you do name a section, however, you have to close it using the same name. Sections can also be nested, which allows you to declare new variables incrementally.

2.7 Namespaces

Lean provides us with the ability to group definitions, notation, and other information into nested, hierarchical *namespaces*:

```
namespace foo
  def a : ℕ := 5
  def f (x : ℕ) : ℕ := x + 7

  def fa : ℕ := f a
  def ffa : ℕ := f (f a)

  print "inside foo"

  check a
  check f
  check fa
  check ffa
  check foo.fa
end foo

print "outside the namespace"

-- check a -- error
-- check f -- error
check foo.a
check foo.f
check foo.fa
check foo.ffa

open foo

print "opened foo"

check a
check f
check fa
check foo.fa
```

When we declare that we are working in the namespace `foo`, every identifier we declare has a full name with prefix “`foo.`” Within the namespace, we can refer to identifiers by their shorter names, but once we end the namespace, we have to use the longer names.

The `open` command brings the shorter names into the current context. Often, when we import a theory file, we will want to open one or more of the namespaces it contains, to have access to the short identifiers, notations, and so on. But sometimes we will want to leave this information hidden, for example, when they conflict with identifiers and notations in another namespace we want to use. Thus namespaces give us a way to manage our working environment.

For example, Lean groups definitions and theorems involving lists into a namespace `list`.

```
check list.nil
check list.cons
check list.append
```

We will discuss their types, below. The command `open list` allows us to use the shorter names:

```
open list

check nil
check cons
check append
```

Like sections, namespaces can be nested:

```
namespace foo
  def a : ℕ := 5
  def f (x : ℕ) : ℕ := x + 7

  def fa : ℕ := f a

  namespace bar
    def ffa : ℕ := f (f a)

    check fa
    check ffa
  end bar

  check fa
  check bar.ffa
end foo

check foo.fa
check foo.bar.ffa

open foo

check fa
check bar.ffa
```

Namespaces that have been closed can later be reopened, even in another file:

```

namespace foo
  def a : ℕ := 5
  def f (x : ℕ) : ℕ := x + 7

  def fa : ℕ := f a
end foo

check foo.a
check foo.f

namespace foo
  def ffa : ℕ := f (f a)
end foo

```

Like sections, nested namespaces have to be closed in the order they are opened. Also, a namespace cannot be opened within a section; namespaces have to live on the outer levels.

Namespaces and sections serve different purposes: namespaces organize data and sections declare variables for insertion in theorems. A namespace can be viewed as a special kind of section, however. In particular, if you use the **variable** command within a namespace, its scope is limited to the namespace. Similarly, if you use an **open** command within a namespace, its effects disappear when the namespace is closed.

As scoping mechanisms, namespaces and sections govern more than just variables and identifier names. We will later see that notations defined in a namespace are operant only when the namespace is open, and notation defined in a section has scope limited to the section. Similarly, if we use the **open** command inside a section or namespace, it only remains in effect until that section or namespace is closed. As a result, namespaces and sections provide useful ways of managing the background context while we work with Lean.

2.8 Dependent Types

You now have rudimentary ways of defining functions and objects in Lean, and we will gradually introduce you to many more. Our ultimate goal in Lean is to *prove* things about the objects we define, and the next chapter will introduce you to Lean’s mechanisms for stating theorems and constructing proofs. Meanwhile, let us remain on the topic of defining objects in dependent type theory for just a moment longer, in order to explain what makes dependent type theory *dependent*, and why that is useful.

The short explanation is that what makes dependent type theory dependent is that types can depend on parameters. You have already seen a nice example of this: the type `list α` depends on the argument α , and this dependence is what distinguishes `list ℕ` and `list bool`. For another example, consider the type `vec α n`, the type of vectors of elements of α of length n . This type depends on *two* parameters: the type $\alpha : \mathbf{Type}$ of the elements in the vector and the length $n : \mathbb{N}$.

Suppose we wish to write a function `cons` which inserts a new element at the head of a list. What type should `cons` have? Such a function is *polymorphic*: we expect the `cons`

function for \mathbb{N} , `bool`, or an arbitrary type α to behave the same way. So it makes sense to take the type to be the first argument to `cons`, so that for any type, α , `cons α` is the insertion function for lists of type α . In other words, for every α , `cons α` is the function that takes an element $a : \alpha$ and a list $l : \text{list } \alpha$, and returns a new list, so we have `cons α a l : list α` .

It is clear that `cons α` should have type $\alpha \rightarrow \text{list } \alpha \rightarrow \text{list } \alpha$. But what type should `cons` have? A first guess might be `Type $\rightarrow \alpha \rightarrow \text{list } \alpha \rightarrow \text{list } \alpha$` , but, on reflection, this does not make sense: the α in this expression does not refer to anything, whereas it should refer to the argument of type `Type`. In other words, *assuming* $\alpha : \text{Type}$ is the first argument to the function, the type of the next two elements are α and `list α` . These types vary depending on the first argument, α .

This is an instance of a *Pi type* in dependent type theory. Given $\alpha : \text{Type}$ and $\beta : \alpha \rightarrow \text{Type}$, think of β as a family of types over α , that is, a type $\beta \ a$ for each $a : \alpha$. In that case, the type $\Pi x : \alpha, \beta \ x$ denotes the type of functions f with the property that, for each $a : \alpha$, $f \ a$ is an element of $\beta \ a$. In other words, the type of the value returned by f depends on its input.

Notice that $\Pi x : \alpha, \beta$ makes sense for any expression $\beta : \text{Type}$. When the value of β depends on x (as does, for example, the expression $\beta \ x$ in the previous paragraph), $\Pi x : \alpha, \beta$ denotes a dependent function type. When β doesn't depend on x , $\Pi x : \alpha, \beta$ is no different from the type $\alpha \rightarrow \beta$. Indeed, in dependent type theory (and in Lean), the Pi construction is fundamental, and $\alpha \rightarrow \beta$ is nothing more than notation for $\Pi x : \alpha, \beta$ when β does not depend on α .

Returning to the example of lists, we can model some basic list operations as follows. We use `namespace hide` to avoid a naming conflict with the `list` type defined in the standard library.

```
namespace hide

universe variable u

constant list : Type u  $\rightarrow$  Type u

constant cons :  $\Pi \alpha : \text{Type } u, \alpha \rightarrow \text{list } \alpha \rightarrow \text{list } \alpha$ 
constant nil :  $\Pi \alpha : \text{Type } u, \text{list } \alpha$ 
constant head :  $\Pi \alpha : \text{Type } u, \text{list } \alpha \rightarrow \alpha$ 
constant tail :  $\Pi \alpha : \text{Type } u, \text{list } \alpha \rightarrow \text{list } \alpha$ 
constant append :  $\Pi \alpha : \text{Type } u, \text{list } \alpha \rightarrow \text{list } \alpha \rightarrow \text{list } \alpha$ 

end hide
```

You can enter the symbol Π by typing `\Pi`. Here, `nil` is intended to denote the empty list, `head` and `tail` return the first element of a list and the remainder, respectively. The constant `append` is intended to denote the function that concatenates two lists.

We emphasize that these constant declarations are only for the purposes of illustration. The `list` type and all these operations are, in fact, *defined* in Lean’s standard library, and are proved to have the expected properties. In fact, as the next example shows, the types indicated above are essentially the types of the objects that are defined in the library. (We will explain the `@` symbol and the difference between the round and curly brackets momentarily.)

```
open list

check list      -- Type u_1 → Type u_1

check @cons     -- Π {α : Type u_1}, α → list α → list α
check @nil      -- Π {α : Type u_1}, list α
check @head     -- Π {α : Type u_1} [_inst_1 : inhabited α], list α → α
check @tail     -- Π {α : Type u_1}, list α → list α
check @append   -- Π {α : Type u_1}, list α → list α → list α
```

There is a subtlety in the definition of `head`: the type α is required to have at least one element, and when passed the empty list, the function must determine a default element of the relevant type. We will explain how this is done in a later chapter.

Vector operations are handled similarly:

```
universe variable u
constant vec : Type u → ℕ → Type u

namespace vec
  constant empty : Π α : Type u, vec α 0
  constant cons :
    Π (α : Type u) (n : ℕ), α → vec α n → vec α (n + 1)
  constant append :
    Π (α : Type u) (n m : ℕ), vec α m → vec α n → vec α (n + m)
end vec
```

In the coming chapters, you will come across many instances of dependent types. Here we will mention just one more important and illustrative example, the *Sigma types*, $\Sigma x : \alpha, \beta x$, sometimes also known as *dependent pairs*. These are, in a sense, companions to the Pi types. The type $\Sigma x : \alpha, \beta x$ denotes the type of pairs `sigma.mk a b` where $a : \alpha$ and $b : \beta a$.

Just as Pi types $\Pi x : \alpha, \beta x$ generalize the notion of a function type $\alpha \rightarrow \beta$ by allowing β to depend on α , Sigma types $\Sigma x : \alpha, \beta x$ generalize the cartesian product $\alpha \times \beta$ in the same way: in the expression `sigma.mk a b`, the type of the second element of the pair, $b : \beta a$, depends on the first element of the pair, $a : \alpha$.

```
variable α : Type
variable β : α → Type
variable a : α
variable b : β a
```

```

check sigma.mk a b --  $\Sigma (a : \alpha), \beta a$ 
check (sigma.mk a b).1 --  $\alpha$ 
check (sigma.mk a b).2 --  $\beta (\text{sigma.fst } (\text{sigma.mk } a \ b))$ 

eval (sigma.mk a b).1 --  $a$ 
eval (sigma.mk a b).2 --  $b$ 

```

Notice that when p is a dependent pair the expressions $(\text{sigma.mk } a \ b).1$ and $(\text{sigma.mk } a \ b).2$ are short for $\text{sigma.fst } (\text{sigma.mk } a \ b)$ and $\text{sigma.snd } (\text{sigma.mk } a \ b)$, respectively, and that these reduce to a and b , respectively.

2.9 Implicit Arguments

Suppose we have an implementation of lists as described above.

```

namespace hide
universe variable u
constant list : Type u → Type u

namespace list
  constant cons :  $\Pi \alpha : \text{Type } u, \alpha \rightarrow \text{list } \alpha \rightarrow \text{list } \alpha$ 
  constant nil :  $\Pi \alpha : \text{Type } u, \text{list } \alpha$ 
  constant append :  $\Pi \alpha : \text{Type } u, \text{list } \alpha \rightarrow \text{list } \alpha \rightarrow \text{list } \alpha$ 
end list
end hide

```

Then, given a type α , some elements of α , and some lists of elements of α , we can construct new lists using the constructors.

```

open hide.list

variable  $\alpha : \text{Type}$ 
variable  $a : \alpha$ 
variables l1 l2 : list  $\alpha$ 

check cons  $\alpha$  a (nil  $\alpha$ )
check append  $\alpha$  (cons  $\alpha$  a (nil  $\alpha$ )) l1
check append  $\alpha$  (append  $\alpha$  (cons  $\alpha$  a (nil  $\alpha$ )) l1) l2

```

Because the constructors are polymorphic over types, we have to insert the type α as an argument repeatedly. But this information is redundant: one can infer the argument α in $\text{cons } \alpha \ a \ (\text{nil } \alpha)$ from the fact that the second argument, a , has type α . One can similarly infer the argument in $\text{nil } \alpha$, not from anything else in that expression, but from the fact that it is sent as an argument to the function cons , which expects an element of type $\text{list } \alpha$ in that position.

This is a central feature of dependent type theory: terms carry a lot of information, and often some of that information can be inferred from the context. In Lean, one uses an

underscore, `_`, to specify that the system should fill in the information automatically. This is known as an “implicit argument.”

```
check cons _ a (nil _)
check append _ (cons _ a (nil _)) l1
check append _ (append _ (cons _ a (nil _)) l1) l2
```

It is still tedious, however, to type all these underscores. When a function takes an argument that can generally be inferred from context, Lean allows us to specify that this argument should, by default, be left implicit. This is done by putting the arguments in curly braces, as follows:

```
namespace list
  constant cons :  $\Pi \{ \alpha : \text{Type } u \}, \alpha \rightarrow \text{list } \alpha \rightarrow \text{list } \alpha$ 
  constant nil :  $\Pi \{ \alpha : \text{Type } u \}, \text{list } \alpha$ 
  constant append :  $\Pi \{ \alpha : \text{Type } u \}, \text{list } \alpha \rightarrow \text{list } \alpha \rightarrow \text{list } \alpha$ 
end list

open hide.list

variable  $\alpha : \text{Type}$ 
variable a :  $\alpha$ 
variables l1 l2 : list  $\alpha$ 

check cons a nil
check append (cons a nil) l1
check append (append (cons a nil) l1) l2
```

All that has changed are the braces around $\alpha : \text{Type } u$ in the declaration of the variables. We can also use this device in function definitions:

```
universe variable u
def ident { $\alpha : \text{Type } u$ } (x :  $\alpha$ ) := x

variables  $\alpha \beta : \text{Type } u$ 
variables (a :  $\alpha$ ) (b :  $\beta$ )

check ident      --  $?M_1 \rightarrow ?M_1$ 
check ident a    --  $\alpha$ 
check ident b    --  $\beta$ 
```

This makes the first argument to `ident` implicit. Notationally, this hides the specification of the type, making it look as though `ident` simply takes an argument of any type. In fact, the function `id` is defined in the standard library in exactly this way. We have chosen a nontraditional name here only to avoid a clash of names.

Variables can also be declared implicit when they are declared with the `variables` command:

```

universe variable u

section
  variable {α : Type u}
  variable x : α
  def ident := x
end

variables α β : Type u
variables (a : α) (b : β)

check ident
check ident a
check ident b

```

This definition of `ident` has the same effect as the one above.

Lean has very complex mechanisms for instantiating implicit arguments, and we will see that they can be used to infer function types, predicates, and even proofs. The process of instantiating these “holes,” or “placeholders,” in a term is often known as *elaboration*. The presence of implicit arguments means that at times there may be insufficient information to fix the meaning of an expression precisely. An expression like `id` or `list.nil` is said to be *polymorphic*, because it can take on different meanings in different contexts. One can always specify the type `T` of an expression `e` by writing `(e : T)`. This instructs Lean’s elaborator to use the value `T` as the type of `e` when trying to resolve implicit arguments. The second pair of examples below use this mechanism to specify the desired types of the expressions `id` and `list.nil`:

```

check list.nil      -- list ?M1
check id            -- ?M1 → ?M1

check (list.nil : list N) -- list N
check (id : N → N)      -- N → N

```

Numerals are overloaded in Lean, but when the type of a numeral cannot be inferred, Lean assumes, by default, that it is a natural number. So the expressions in the first two `check` commands are elaborated in the same way, whereas the third `check` command interprets `2` as a raw numeral.

```

check 2             -- N
check (2 : N)       -- N
check (2 : num)     -- num

```

Sometimes, however, we may find ourselves in a situation where we have declared an argument to a function to be implicit, but now want to provide the argument explicitly. If `foo` is such a function, the notation `@foo` denotes the same function with all the arguments made explicit.

```

check @id      --  $\Pi \{ \alpha : \text{Type } u_1 \}, \alpha \rightarrow \alpha$ 
check @id  $\alpha$   --  $\alpha \rightarrow \alpha$ 
check @id  $\beta$    --  $\beta \rightarrow \beta$ 
check @id  $\alpha$  a  --  $\alpha$ 
check @id  $\beta$  b   --  $\beta$ 

```

Notice that now the first `check` command gives the type of the identifier, `id`, without inserting any placeholders. Moreover, the output indicates that the first argument is implicit.

Propositions and Proofs

By now, you have seen how to define some elementary notions in dependent type theory.

In this chapter, we will explain how mathematical propositions and proofs are expressed in the language of dependent type theory, so that you can start proving assertions about the objects and notations that have been defined. The encoding we use here is specific to the standard library; we will discuss proofs in *homotopy type theory* in a later chapter.

3.1 Propositions as Types

One strategy for proving assertions about objects defined in the language of dependent type theory is to layer an assertion language and a proof language on top of the definition language. But there is no reason to multiply languages in this way: dependent type theory is flexible and expressive, and there is no reason we cannot represent assertions and proofs in the same general framework.

For example, we could introduce a new type, `Prop`, to represent propositions, and constructors to build new propositions from others.

```

constant and : Prop → Prop → Prop
constant or  : Prop → Prop → Prop
constant not  : Prop → Prop
constant implies : Prop → Prop → Prop

variables p q r : Prop
check and p q           -- Prop
check or (and p q) r    -- Prop
check implies (and p q) (and q p) -- Prop

```

We could then introduce, for each element $p : \text{Prop}$, another type $\text{Proof } p$, for the type of proofs of p . An “axiom” would be constant of such a type.

```
constant Proof : Prop → Type

constant and_comm :  $\Pi$  p q : Prop, Proof (implies (and p q) (and q p))

variables p q : Prop
check and_comm p q      -- Proof (implies (and p q) (and q p))
```

In addition to axioms, however, we would also need rules to build new proofs from old ones. For example, in many proof systems for propositional logic, we have the rule of modus ponens:

From a proof of $\text{implies } p \ q$ and a proof of p , we obtain a proof of q .

We could represent this as follows:

```
constant modus_ponens (p q : Prop) : Proof (implies p q) → Proof p → Proof q
```

Systems of natural deduction for propositional logic also typically rely on the following rule:

Suppose that, assuming p as a hypothesis, we have a proof of q . Then we can “cancel” the hypothesis and obtain a proof of $\text{implies } p \ q$.

We could render this as follows:

```
constant implies_intro (p q : Prop) : (Proof p → Proof q) → Proof (implies p q).
```

This approach would provide us with a reasonable way of building assertions and proofs. Determining that an expression t is a correct proof of assertion p would then simply be a matter of checking that t has type $\text{Proof } p$.

Some simplifications are possible, however. To start with, we can avoid writing the term Proof repeatedly by conflating $\text{Proof } p$ with p itself. In other words, whenever we have $p : \text{Prop}$, we can interpret p as a type, namely, the type of its proofs. We can then read $t : p$ as the assertion that t is a proof of p .

Moreover, once we make this identification, the rules for implication show that we can pass back and forth between $\text{implies } p \ q$ and $p \rightarrow q$. In other words, implication between propositions p and q corresponds to having a function that takes any element of p to an element of q . As a result, the introduction of the connective implies is entirely redundant: we can use the usual function space constructor $p \rightarrow q$ from dependent type theory as our notion of implication.

This is the approach followed in the Calculus of Inductive Constructions, and hence in Lean as well. The fact that the rules for implication in a proof system for natural deduction correspond exactly to the rules governing abstraction and application for functions is an instance of the *Curry-Howard isomorphism*, sometimes known as the *propositions-as-types* paradigm. In fact, the type `Prop` is syntactic sugar for `Type 0`, the very bottom of the type hierarchy described in the last chapter. `Prop` has some special features, but like the other type universes, it is closed under the arrow constructor: if we have $p \rightarrow q : \text{Prop}$, then $p \rightarrow q : \text{Prop}$.

There are at least two ways of thinking about propositions as types. To some who take a constructive view of logic and mathematics, this is a faithful rendering of what it means to be a proposition: a proposition p represents a sort of data type, namely, a specification of the type of data that constitutes a proof. A proof of p is then simply an object $t : p$ of the right type.

Those not inclined to this ideology can view it, rather, as a simple coding trick. To each proposition p we associate a type, which is empty if p is false and has a single element, say $*$, if p is true. In the latter case, let us say that (the type associated with) p is *inhabited*. It just so happens that the rules for function application and abstraction can conveniently help us keep track of which elements of *Prop* are inhabited. So constructing an element $t : p$ tells us that p is indeed true. You can think of the inhabitant of p as being the “fact that p is true.” A proof of $p \rightarrow q$ uses “the fact that p is true” to obtain “the fact that q is true.”

Indeed, if $p : \text{Prop}$ is any proposition, Lean’s kernel treats any two elements $t_1 \ t_2 : p$ as being definitionally equal, much the same way as it treats $(\lambda \ x, \ t)s$ and $t[s/x]$ as definitionally equal. This is known as “proof irrelevance,” and is consistent with the interpretation in the last paragraph. It means that even though we can treat proofs $t : p$ as ordinary objects in the language of dependent type theory, they carry no information beyond the fact that p is true.

The two ways we have suggested thinking about the propositions-as-types paradigm differ in a fundamental way. From the constructive point of view, proofs are abstract mathematical objects that are *denoted* by suitable expressions in dependent type theory. In contrast, if we think in terms of the coding trick described above, then the expressions themselves do not denote anything interesting. Rather, it is the fact that we can write them down and check that they are well-typed that ensures that the proposition in question is true. In other words, the expressions *themselves* are the proofs.

In the exposition below, we will slip back and forth between these two ways of talking, at times saying that an expression “constructs” or “produces” or “returns” a proof of a proposition, and at other times simply saying that it “is” such a proof. This is similar to the way that computer scientists occasionally blur the distinction between syntax and semantics by saying, at times, that a program “computes” a certain function, and at other times speaking as though the program “is” the function in question.

In any case, all that really matters is that the bottom line is clear. To formally express

a mathematical assertion in the language of dependent type theory, we need to exhibit a term $p : \mathbf{Prop}$. To *prove* that assertion, we need to exhibit a term $t : p$. Lean’s task, as a proof assistant, is to help us to construct such a term, t , and to verify that it is well-formed and has the correct type.

3.2 Working with Propositions as Types

In the propositions-as-types paradigm, theorems involving only \rightarrow can be proved using lambda abstraction and application. In Lean, the `theorem` command introduces a new theorem:

```
constants p q : Prop

theorem t1 : p → q → p := λ hp : p, λ hq : q, hp
```

This looks exactly like the definition of the constant function in the last chapter, the only difference being that the arguments are elements of **Prop** rather than **Type**. Intuitively, our proof of $p \rightarrow q \rightarrow p$ assumes p and q are true, and uses the first hypothesis (trivially) to establish that the conclusion, p , is true.

Note that the `theorem` command is really a version of the `definition` command: under the propositions and types correspondence, proving the theorem $p \rightarrow q \rightarrow p$ is really the same as defining an element of the associated type. To the kernel type checker, there is no difference between the two.

There are a few pragmatic differences between definitions and theorems, however.

In normal circumstances, it is never necessary to unfold the “definition” of a theorem; by proof irrelevance, any two proofs of that theorem are definitionally equal. Once the proof of a theorem is complete, typically we only need to know that the proof exists; it doesn’t matter what the proof is. In light of that fact, Lean tags proofs as *irreducible*, which serves as a hint to the parser (more precisely, the *elaborator*) that there is generally no need to unfold it when processing a file. Moreover, for efficiency purposes, Lean treats theorems as axiomatic constants within the file in which they are defined. This makes it possible to process and check theorems in parallel, since theorems later in a file do not make use of the contents of earlier proofs.

As with definitions, the `print` command will show you the proof of a theorem.

```
theorem t1 : p → q → p := λ hp : p, λ hq : q, hp

print t1
```

(To save space, the online version of Lean does not store proofs of theorems in the library, so you cannot print them in the browser interface.)

Notice that the lambda abstractions $hp : p$ and $hq : q$ can be viewed as temporary assumptions in the proof of $t1$. Lean provides the alternative syntax `assume` for such a lambda abstraction:

```
theorem t1 : p → q → p :=
  assume hp : p,
  assume hq : q,
  hp
```

Lean also allows us to specify the type of the final term hp , explicitly, with a `show` statement.

```
theorem t1 : p → q → p :=
  assume hp : p,
  assume hq : q,
  show p, from hp
```

Adding such extra information can improve the clarity of a proof and help detect errors when writing a proof. The `show` command does nothing more than annotate the type, and, internally, all the presentations of $t1$ that we have seen produce the same term. Lean also allows you to use the alternative syntax `proposition`, `lemma`, or `corollary` instead of `theorem`:

```
lemma t1 : p → q → p :=
  assume hp : p,
  assume hq : q,
  show p, from hp
```

As with ordinary definitions, one can move the lambda-abstracted variables to the left of the colon:

```
theorem t1 (hp : p) (hq : q) : p := hp

check t1 -- p → q → p
```

Now we can apply the theorem $t1$ just as a function application.

```
axiom hp : p

theorem t2 : q → p := t1 hp
```

Here, the `axiom` command is alternative syntax for `constant`. Declaring a “constant” $hp : p$ is tantamount to declaring that p is true, as witnessed by hp . Applying the theorem $t1 : p \rightarrow q \rightarrow p$ to the fact $hp : p$ that p is true yields the theorem $t2 : q \rightarrow p$.

Notice, by the way, that the original theorem `t1` is true for *any* propositions `p` and `q`, not just the particular constants declared. So it would be more natural to define the theorem so that it quantifies over those, too:

```
theorem t1 (p q : Prop) (hp : p) (hq : q) : p := hp
check t1
```

The type of `t1` is now $\forall p\ q : \text{Prop}, p \rightarrow q \rightarrow p$. We can read this as the assertion “for every pair of propositions `p` `q`, we have $p \rightarrow q \rightarrow p$ ”. The symbol \forall is alternate syntax for Π , and later we will see how `Pi` types let us model universal quantifiers more generally. For the moment, however, we will focus on theorems in propositional logic, generalized over the propositions. We will tend to work in sections with variables over the propositions, so that they are generalized for us automatically.

When we generalize `t1` in that way, we can then apply it to different pairs of propositions, to obtain different instances of the general theorem.

```
theorem t1 (p q : Prop) (hp : p) (hq : q) : p := hp

variables p q r s : Prop

check t1 p q      -- p → q → p
check t1 r s      -- r → s → r
check t1 (r → s) (s → r) -- (r → s) → (s → r) → r → s

variable h : r → s
check t1 (r → s) (s → r) h -- (s → r) → r → s
```

Remember that under the propositions-as-types correspondence, a variable `h` of type $r \rightarrow s$ can be viewed as the hypothesis, or premise, that $r \rightarrow s$ holds. For that reason, Lean offers the alternative syntax, `premise`, for `variable`.

```
premise h : r → s
check t1 (r → s) (s → r) h
```

As another example, let us consider the composition function discussed in the last chapter, now with propositions instead of types.

```
variables p q r s : Prop

theorem t2 (h1 : q → r) (h2 : p → q) : p → r :=
  assume h3 : p,
  show r, from h1 (h2 h3)
```

As a theorem of propositional logic, what does `t2` say?

Lean allows the alternative syntax `premise` and `premises` for `variable` and `variables`. This makes sense, of course, for variables whose type is an element of `Prop`. It is also often

useful to use numeric unicode subscripts, entered as `\0`, `\1`, `\2`, ..., for hypotheses. The following definition of `t2` has the same net effect as the preceding one.

```
variables p q r s : Prop
premises (h1 : q → r) (h2 : p → q)

theorem t2 : p → r :=
assume h3 : p,
show r, from h1 (h2 h3)
```

3.3 Propositional Logic

Lean defines all the standard logical connectives and notation. The propositional connectives come with the following notation:

Ascii	Unicode	Emacs shortcut for unicode	Definition
true			true
false			false
not	¬	\not, \neg	not
∧	∧	\and	and
∨	∨	\or	or
→	→	\to, \r, \implies	
↔	↔	\iff, \lr	iff

They all take values in `Prop`.

```
variables p q : Prop

check p → q → p ∧ q
check ¬p → p ↔ false
check p ∨ q → q ∨ p
```

The order of operations is fairly standard: unary negation \neg binds most strongly, then \wedge and \vee , and finally \rightarrow and \leftrightarrow . For example, $a \wedge b \rightarrow c \vee d \wedge e$ means $(a \wedge b) \rightarrow (c \vee (d \wedge e))$. Remember that \rightarrow associates to the right (nothing changes now that the arguments are elements of `Prop`, instead of some other `Type`), as do the other binary connectives. So if we have $p \ q \ r : \text{Prop}$, the expression $p \rightarrow q \rightarrow r$ reads “if p , then if q , then r .” This is just the “curried” form of $p \wedge q \rightarrow r$.

In the last chapter we observed that lambda abstraction can be viewed as an “introduction rule” for \rightarrow . In the current setting, it shows how to “introduce” or establish an implication. Application can be viewed as an “elimination rule,” showing how to “eliminate” or use an implication in a proof. The other propositional connectives are defined in the standard library in the file `init.datatypes`, and each comes with its canonical introduction and elimination rules.

Conjunction

The expression `and.intro h1 h2` creates a proof for $p \wedge q$ using proofs $h1 : p$ and $h2 : q$. It is common to describe `and.intro` as the *and-introduction* rule. In the next example we use `and.intro` to create a proof of $p \rightarrow q \rightarrow p \wedge q$.

```
example (hp : p) (hq : q) : p ∧ q := and.intro hp hq
check assume (hp : p) (hq : q), and.intro hp hq
```

The `example` command states a theorem without naming it or storing it in the permanent context. Essentially, it just checks that the given term has the indicated type. It is convenient for illustration, and we will use it often.

The expression `and.elim_left H` creates a proof of p from a proof $h : p \wedge q$. Similarly, `and.elim_right H` is a proof of q . They are commonly known as the right and left *and-elimination* rules.

```
example (h : p ∧ q) : p := and.elim_left h
example (h : p ∧ q) : q := and.elim_right h
```

Because they are so commonly used, the standard library provides the abbreviations `and.left` and `and.right` for `and.elim_left` and `and.elim_right`, respectively.

We can now prove $p \wedge q \rightarrow q \wedge p$ with the following proof term.

```
example (h : p ∧ q) : q ∧ p :=
and.intro (and.right h) (and.left h)
```

Notice that *and-introduction* and *and-elimination* are similar to the pairing and projection operations for the cartesian product. The difference is that given $hp : p$ and $hq : q$, `and.intro hp hq` has type $p \wedge q : \text{Prop}$, while `pair hp hq` has type $p \times q : \text{Type}$. The similarity between \wedge and \times is another instance of the Curry-Howard isomorphism, but in contrast to implication and the function space constructor, \wedge and \times are treated separately in Lean. With the analogy, however, the proof we have just constructed is similar to a function that swaps the elements of a pair.

We will see in a later chapter that certain types in Lean are *structures*, which is to say, the type is defined with a single canonical *constructor* which builds an element of the type from a sequence of suitable arguments. For every $p \ q : \text{Prop}$, $p \wedge q$ is an example: the canonical way to construct an element is to apply `and.intro` to suitable arguments $hp : p$ and $hq : q$. Lean allows us to use *anonymous constructor* notation $\langle \text{arg1}, \text{arg2}, \dots \rangle$ in situations like these, when the relevant type can be inferred from the context. In particular, we can often write $\langle hp, hq \rangle$ instead of `and.intro hp hq`:

```
variables p q : Prop
premises (hp : p) (hq : q)

check (⟨hp, hq⟩ : p ∧ q)
```

These angle brackets are obtained by typing `\<` and `\>`, respectively. Alternatively, you can use ASCII equivalents `(|` and `|)`:

```
variables p q : Prop
premises (hp : p) (hq : q)

example : p ∧ q := (|hp, hq|)
```

Lean provides another useful syntactic gadget. Given an expression `e` of type `foo` (possibly applied to some arguments), the notation `e^.bar` is shorthand for `foo.bar e`. This provides a convenient way of accessing functions without opening a namespace. For example, the following three expressions all mean the same thing:

```
variable l : list ℕ

check list.head l
check l^.head
check l.head
```

As a result, given `h : p ∧ q`, we can write `h^.left` for `and.left h` and `h^.right` for `and.right h`. We can therefore rewrite the sample proof above conveniently as follows:

```
example (h : p ∧ q) : q ∧ p :=
  ⟨h^.right, h^.left⟩
```

There is a fine line between brevity and obfuscation, and omitting information in this way can sometimes make a proof harder to read. But for straightforward constructions like the one above, when the type of `h` and the goal of the construction are salient, the notation is clean and effective.

It is common to iterate constructions like “and.” Lean also allows you to flatten nested constructors that associate to the right, so that these two proofs are equivalent:

```
example (h : p ∧ q) : q ∧ p ∧ q :=
  ⟨h^.right, ⟨h^.left, h^.right⟩⟩

example (h : p ∧ q) : q ∧ p ∧ q :=
  ⟨h^.right, h^.left, h^.right⟩
```

This is often useful as well.

Disjunction

The expression `or.intro_left q hp` creates a proof of $p \vee q$ from a proof `hp : p`. Similarly, `or.intro_right p hq` creates a proof for $p \vee q$ using a proof `hq : q`. These are the left and right *or-introduction* rules.

```
example (hp : p) : p ∨ q := or.intro_left q hp
example (hq : q) : p ∨ q := or.intro_right p hq
```

The *or-elimination* rule is slightly more complicated. The idea is that we can prove r from $p \vee q$, by showing that r follows from p and that r follows from q . In other words, it is a proof “by cases.” In the expression `or.elim hpq hpr hqr`, `or.elim` takes three arguments, `hpq : p ∨ q`, `hpr : p → r` and `hqr : q → r`, and produces a proof of r . In the following example, we use `or.elim` to prove $p \vee q \rightarrow q \vee p$.

```
example (h : p ∨ q) : q ∨ p :=
or.elim h
  (assume hp : p,
    show q ∨ p, from or.intro_right q hp)
  (assume hq : q,
    show q ∨ p, from or.intro_left p hq)
```

In most cases, the first argument of `or.intro_right` and `or.intro_left` can be inferred automatically by Lean. Lean therefore provides `or.inr` and `or.inl` as shorthands for `or.intro_right _` and `or.intro_left _`. Thus the proof term above could be written more concisely:

```
example (h : p ∨ q) : q ∨ p := or.elim h (λ hp, or.inr hp) (λ hq, or.inl hq)
```

Notice that there is enough information in the full expression for Lean to infer the types of `hp` and `hq` as well. But using the type annotations in the longer version makes the proof more readable, and can help catch and debug errors.

Because `or` has two constructors, we cannot use anonymous constructor notation. But we can still write `h^.elim` instead of `or.elim h`:

```
example (h : p ∨ q) : q ∨ p :=
h^.elim
  (assume hp : p, or.inr hp)
  (assume hq : q, or.inl hq)
```

Once again, you should exercise judgment as to whether such abbreviations enhance or diminish readability.

Negation and Falsity

The expression `not.intro h` produces a proof of $\neg p$ from $h : p \rightarrow \text{false}$. That is, we obtain $\neg p$ if we can derive a contradiction from p . Similarly, the expression `hnp hp` produces a proof of `false` from `hp : p` and `hnp : $\neg p$` . The next example uses both these rules to produce a proof of $(p \rightarrow q) \rightarrow \neg q \rightarrow \neg p$.

```
example (hpq : p → q) (hnq : ¬q) : ¬p :=
  assume hp : p,
  show false, from hnq (hpq hp)
```

The connective `false` has a single elimination rule, `false.elim`, which expresses the fact that anything follows from a contradiction. This rule is sometimes called *ex falso* (short for *ex falso sequitur quodlibet*), or the *principle of explosion*.

```
example (hp : p) (hnp : ¬p) : q := false.elim (hnp hp)
```

The arbitrary fact, q , that follows from falsity is an implicit argument in `false.elim` and is inferred automatically. This pattern, deriving an arbitrary fact from contradictory hypotheses, is quite common, and is represented by `absurd`.

```
example (hp : p) (hnp : ¬p) : q := absurd hp hnp
```

Here, for example, is a proof of $\neg p \rightarrow q \rightarrow (q \rightarrow p) \rightarrow r$:

```
example (hnp : ¬p) (hq : q) (hqp : q → p) : r :=
  absurd (hqp hq) hnp
```

Incidentally, just as `false` has only an elimination rule, `true` has only an introduction rule, `true.intro : true`, sometimes abbreviated `trivial : true`. In other words, `true` is simply true, and has a canonical proof, `trivial`.

Logical Equivalence

The expression `iff.intro h1 h2` produces a proof of $p \leftrightarrow q$ from `h1 : $p \rightarrow q$` and `h2 : $q \rightarrow p$` . The expression `iff.elim_left H` produces a proof of $p \rightarrow q$ from `h : $p \leftrightarrow q$` . Similarly, `iff.elim_right H` produces a proof of $q \rightarrow p$ from `h : $p \leftrightarrow q$` . Here is a proof of $p \wedge q \leftrightarrow q \wedge p$:

```
theorem and_swap : p ∧ q ↔ q ∧ p :=
  iff.intro
    (assume h : p ∧ q,
     show q ∧ p, from and.intro (and.right h) (and.left h))
    (assume h : q ∧ p,
```

```

show p ∧ q, from and.intro (and.right h) (and.left h))

check and_swap p q    -- p ∧ q ↔ q ∧ p

```

Because they represent a form of *modus ponens*, `iff.elim_left` and `iff.elim_right` can be abbreviated `iff.mp` and `iff.mpr`, respectively. In the next example, we use that theorem to derive $q \wedge p$ from $p \wedge q$:

```

premise h : p ∧ q
example : q ∧ p := iff.mp (and_swap p q) h

```

Because `iff` is defined internally from `and`, we can use the anonymous constructor notation to construct a proof of $p \leftrightarrow q$ from proofs of the forward and backward directions. We can also use the `.^` notation with `mp` and `mpr`. The previous examples can therefore be written concisely as follows:

```

theorem and_swap : p ∧ q ↔ q ∧ p :=
  ⟨ λ h, ⟨h^.right, h^.left⟩, λ h, ⟨h^.right, h^.left⟩ ⟩

example (h : p ∧ q) : q ∧ p := (and_swap p q)^.mp h

```

3.4 Introducing Auxiliary Subgoals

This is a good place to introduce another device Lean offers to help structure long proofs, namely, the `have` construct, which introduces an auxiliary subgoal in a proof. Here is a small example, adapted from the last section:

```

variables p q : Prop

example (h : p ∧ q) : q ∧ p :=
  have hp : p, from and.left h,
  have hq : q, from and.right h,
  show q ∧ p, from and.intro hq hp

```

Internally, the expression `have h : p, from s, t` produces the term $(\lambda (h : p), t) s$. In other words, s is a proof of p , t is a proof of the desired conclusion assuming $h : p$, and the two are combined by a lambda abstraction and application. This simple device is extremely useful when it comes to structuring long proofs, since we can use intermediate `have`'s as stepping stones leading to the final goal.

Lean also supports a structured way of reasoning backwards from a goal, which models the “suffices to show” construction in ordinary mathematics. The next example simply permutes that last two lines in the previous proof.

```
variables p q : Prop

example (h : p ∧ q) : q ∧ p :=
have hp : p, from and.left h,
suffices hq : q, from and.intro hq hp,
show q, from and.right h
```

Writing `suffices hq : q` leaves us with two goals. First, we have to show that it indeed suffices to show `q`, by proving the original goal of $q \wedge p$ with the additional hypothesis `hq : q`. Finally, we have to show `q`.

3.5 Classical Logic

The introduction and elimination rules we have seen so far are all constructive, which is to say, they reflect a computational understanding of the logical connectives based on the propositions-as-types correspondence. Ordinary classical logic adds to this the law of the excluded middle, $p \vee \neg p$. To use this principle, you have to open the classical namespace.

```
open classical

variable p : Prop
check em p
```

Intuitively, the constructive “or” is very strong: asserting $p \vee q$ amounts to knowing which is the case. If RH represents the Riemann hypothesis, a classical mathematician is willing to assert $RH \vee \neg RH$, even though we cannot yet assert either disjunct.

One consequence of the law of the excluded middle is the principle of double-negation elimination:

```
theorem dne {p : Prop} (h : ¬¬p) : p :=
or.elim (em p)
  (assume hp : p, hp)
  (assume hnp : ¬p, absurd hnp h)
```

Double-negation elimination allows one to prove any proposition, `p`, by assuming `¬p` and deriving `false`, because that amounts to proving `¬¬p`. In other words, double-negation elimination allows one to carry out a proof by contradiction, something which is not generally possible in constructive logic. As an exercise, you might try proving the converse, that is, showing that `em` can be proved from `dne`.

The classical axioms also gives you access to additional patterns of proof that can be justified by appeal to `em`. For example, one can carry out a proof by cases:

```
example (h : ¬¬p) : p :=
by_cases
```

```
(assume h1 : p, h1)
(assume h1 : ¬p, absurd h1 h)
```

Or you can carry out a proof by contradiction:

```
example (h : ¬¬p) : p :=
by_contradiction
  (assume h1 : ¬p,
    show false, from h h1)
```

If you are not used to thinking constructively, it may take some time for you to get a sense of where classical reasoning is used. It is needed in the following example because, from a constructive standpoint, knowing that p and q are not both true does not necessarily tell you which one is false:

```
example (h : ¬ (p ∧ q)) : ¬ p ∨ ¬ q :=
or.elim (em p)
  (assume hp : p,
    or.inr
      (show ¬q, from
        assume hq : q,
          h ⟨hp, hq⟩))
  (assume hp : ¬p,
    or.inl hp)
```

We will see later that there *are* situations in constructive logic where principles like excluded middle and double-negation elimination are permissible, and Lean supports the use of classical reasoning in such contexts.

There are additional classical axioms that are not included by default in the standard library. We will discuss these in detail in a later chapter.

3.6 Examples of Propositional Validities

Lean's standard library contains proofs of many valid statements of propositional logic, all of which you are free to use in proofs of your own. In this section, we will review some common identities, and encourage you to try proving them on your own using the rules above.

The following is a long list of assertions in propositional logic. Prove as many as you can, using the rules introduced above to replace the `sorry` placeholders by actual proofs. The ones that require classical reasoning are grouped together at the end, while the rest are constructively valid.

```
open classical
```

```

variables p q r s : Prop

-- commutativity of  $\wedge$  and  $\vee$ 
example : p  $\wedge$  q  $\leftrightarrow$  q  $\wedge$  p := sorry
example : p  $\vee$  q  $\leftrightarrow$  q  $\vee$  p := sorry

-- associativity of  $\wedge$  and  $\vee$ 
example : (p  $\wedge$  q)  $\wedge$  r  $\leftrightarrow$  p  $\wedge$  (q  $\wedge$  r) := sorry
example : (p  $\vee$  q)  $\vee$  r  $\leftrightarrow$  p  $\vee$  (q  $\vee$  r) := sorry

-- distributivity
example : p  $\wedge$  (q  $\vee$  r)  $\leftrightarrow$  (p  $\wedge$  q)  $\vee$  (p  $\wedge$  r) := sorry
example : p  $\vee$  (q  $\wedge$  r)  $\leftrightarrow$  (p  $\vee$  q)  $\wedge$  (p  $\vee$  r) := sorry

-- other properties
example : (p  $\rightarrow$  (q  $\rightarrow$  r))  $\leftrightarrow$  (p  $\wedge$  q  $\rightarrow$  r) := sorry
example : ((p  $\vee$  q)  $\rightarrow$  r)  $\leftrightarrow$  (p  $\rightarrow$  r)  $\wedge$  (q  $\rightarrow$  r) := sorry
example :  $\neg$ (p  $\vee$  q)  $\leftrightarrow$   $\neg$ p  $\wedge$   $\neg$ q := sorry
example :  $\neg$ p  $\vee$   $\neg$ q  $\rightarrow$   $\neg$ (p  $\wedge$  q) := sorry
example :  $\neg$ (p  $\wedge$   $\neg$  p) := sorry
example : p  $\wedge$   $\neg$ q  $\rightarrow$   $\neg$ (p  $\rightarrow$  q) := sorry
example :  $\neg$ p  $\rightarrow$  (p  $\rightarrow$  q) := sorry
example : ( $\neg$ p  $\vee$  q)  $\rightarrow$  (p  $\rightarrow$  q) := sorry
example : p  $\vee$  false  $\leftrightarrow$  p := sorry
example : p  $\wedge$  false  $\leftrightarrow$  false := sorry
example :  $\neg$ (p  $\leftrightarrow$   $\neg$ p) := sorry
example : (p  $\rightarrow$  q)  $\rightarrow$  ( $\neg$ q  $\rightarrow$   $\neg$ p) := sorry

-- these require classical reasoning
example : (p  $\rightarrow$  r  $\vee$  s)  $\rightarrow$  ((p  $\rightarrow$  r)  $\vee$  (p  $\rightarrow$  s)) := sorry
example :  $\neg$ (p  $\wedge$  q)  $\rightarrow$   $\neg$ p  $\vee$   $\neg$ q := sorry
example :  $\neg$ (p  $\rightarrow$  q)  $\rightarrow$  p  $\wedge$   $\neg$ q := sorry
example : (p  $\rightarrow$  q)  $\rightarrow$  ( $\neg$ p  $\vee$  q) := sorry
example : ( $\neg$ q  $\rightarrow$   $\neg$ p)  $\rightarrow$  (p  $\rightarrow$  q) := sorry
example : p  $\vee$   $\neg$ p := sorry
example : (((p  $\rightarrow$  q)  $\rightarrow$  p)  $\rightarrow$  p) := sorry

```

The `sorry` identifier magically produces a proof of anything, or provides an object of any data type at all. Of course, it is unsound as a proof method – for example, you can use it to prove `false` – and Lean produces severe warnings when files use or import theorems which depend on it. But it is very useful for building long proofs incrementally. Start writing the proof from the top down, using `sorry` to fill in subproofs. Make sure Lean accepts the term with all the `sorry`'s; if not, there are errors that you need to correct. Then go back and replace each `sorry` with an actual proof, until no more remain.

Here is another useful trick. Instead of using `sorry`, you can use an underscore `_` as a placeholder. Recall that this tells Lean that the argument is implicit, and should be filled in automatically. If Lean tries to do so and fails, it returns with an error message “don’t know how to synthesize placeholder.” This is followed by the type of the term it is expecting, and all the objects and hypothesis available in the context. In other words, for each unresolved placeholder, Lean reports the subgoal that needs to be filled at that point. You can then construct a proof by incrementally filling in these placeholders.

For reference, here are two sample proofs of validities taken from the list above.

```

open classical

variables p q r : Prop

-- distributivity
example : p ∧ (q ∨ r) ↔ (p ∧ q) ∨ (p ∧ r) :=
iff.intro
  (assume h : p ∧ (q ∨ r),
    have hp : p, from h^.left,
    or.elim (h^.right)
      (assume hq : q,
        show (p ∧ q) ∨ (p ∧ r), from or.inl ⟨hp, hq⟩)
      (assume hr : r,
        show (p ∧ q) ∨ (p ∧ r), from or.inr ⟨hp, hr⟩))
  (assume h : (p ∧ q) ∨ (p ∧ r),
    or.elim h
      (assume hpq : p ∧ q,
        have hp : p, from hpq^.left,
        have hq : q, from hpq^.right,
        show p ∧ (q ∨ r), from ⟨hp, or.inl hq⟩)
      (assume hpr : p ∧ r,
        have hp : p, from hpr^.left,
        have hr : r, from hpr^.right,
        show p ∧ (q ∨ r), from ⟨hp, or.inr hr⟩))

-- an example that requires classical reasoning
example : ¬(p ∧ ¬q) → (p → q) :=
assume h : ¬(p ∧ ¬q),
assume hp : p,
show q, from
  or.elim (em q)
    (assume hq : q, hq)
    (assume hnq : ¬q, absurd (and.intro hp hnq) h)

```

Quantifiers and Equality

The last chapter introduced you to methods that construct proofs of statements involving the propositional connectives. In this chapter, we extend the repertoire of logical constructions to include the universal and existential quantifiers, and the equality relation.

4.1 The Universal Quantifier

Notice that if α is any type, we can represent a unary predicate p on α as an object of type $\alpha \rightarrow \mathbf{Prop}$. In that case, given $x : \alpha$, $p\ x$ denotes the assertion that p holds of x . Similarly, an object $r : \alpha \rightarrow \alpha \rightarrow \mathbf{Prop}$ denotes a binary relation on α : given $x\ y : \alpha$, $r\ x\ y$ denotes the assertion that x is related to y .

The universal quantifier, $\forall x : \alpha, p\ x$ is supposed to denote the assertion that “for every $x : \alpha$, $p\ x$ ” holds. As with the propositional connectives, in systems of natural deduction, “forall” is governed by an introduction and elimination rule. Informally, the introduction rule states:

Given a proof of $p\ x$, in a context where $x : \alpha$ is arbitrary, we obtain a proof
 $\forall x : \alpha, p\ x$.

The elimination rule states:

Given a proof $\forall x : \alpha, p\ x$ and any term $t : \alpha$, we obtain a proof of $p\ t$.

As was the case for implication, the propositions-as-types interpretation now comes into play. Remember the introduction and elimination rules for Pi types:

Given a term \mathbf{t} of type $\beta \mathbf{x}$, in a context where $\mathbf{x} : \alpha$ is arbitrary, we have $(\lambda \mathbf{x} : \alpha, \mathbf{t}) : \Pi \mathbf{x} : \alpha, \beta \mathbf{x}$.

The elimination rule states:

Given a term $\mathbf{s} : \Pi \mathbf{x} : \alpha, \beta \mathbf{x}$ and any term $\mathbf{t} : \alpha$, we have $\mathbf{s} \mathbf{t} : \beta \mathbf{t}$.

In the case where $\mathbf{p} \mathbf{x}$ has type **Prop**, if we replace $\Pi \mathbf{x} : \alpha, \beta \mathbf{x}$ with $\forall \mathbf{x} : \alpha, \mathbf{p} \mathbf{x}$, we can read these as the correct rules for building proofs involving the universal quantifier.

The Calculus of Inductive Constructions therefore identifies Π and \forall in this way. If \mathbf{p} is any expression, $\forall \mathbf{x} : \alpha, \mathbf{p}$ is nothing more than alternative notation for $\Pi \mathbf{x} : \alpha, \mathbf{p}$, with the idea that the former is more natural than the latter in cases where \mathbf{p} is a proposition. Typically, the expression \mathbf{p} will depend on $\mathbf{x} : \alpha$. Recall that, in the case of ordinary function spaces, we could interpret $\alpha \rightarrow \beta$ as the special case of $\Pi \mathbf{x} : \alpha, \beta$ in which β does not depend on \mathbf{x} . Similarly, we can think of an implication $\mathbf{p} \rightarrow \mathbf{q}$ between propositions as the special case of $\forall \mathbf{x} : \mathbf{p}, \mathbf{q}$ in which the expression \mathbf{q} does not depend on \mathbf{x} .

Here is an example of how the propositions-as-types correspondence gets put into practice.

```
variables (α : Type) (p q : α → Prop)

example : (∀ x : α, p x ∧ q x) → ∀ y : α, p y :=
assume h : ∀ x : α, p x ∧ q x,
take y : α,
show p y, from (h y).left
```

As a notational convention, we give the universal quantifier the widest scope possible, so parentheses are needed to limit the quantifier over \mathbf{x} to the hypothesis in the example above. The canonical way to prove $\forall \mathbf{y} : \alpha, \mathbf{p} \mathbf{y}$ is to take an arbitrary \mathbf{y} , and prove $\mathbf{p} \mathbf{y}$. This is the introduction rule. Now, given that \mathbf{h} has type $\forall \mathbf{x} : \alpha, \mathbf{p} \mathbf{x} \wedge \mathbf{q} \mathbf{x}$, the expression $\mathbf{h} \mathbf{y}$ has type $\mathbf{p} \mathbf{y} \wedge \mathbf{q} \mathbf{y}$. This is the elimination rule. Taking the left conjunct gives the desired conclusion, $\mathbf{p} \mathbf{y}$.

Remember that expressions which differ up to renaming of bound variables are considered to be equivalent. So, for example, we could have used the same variable, \mathbf{x} , in both the hypothesis and conclusion, or chosen the variable \mathbf{z} instead of \mathbf{y} in the proof:

```
example : (∀ x : α, p x ∧ q x) → ∀ y : α, p y :=
assume h : ∀ x : α, p x ∧ q x,
take z : α,
show p z, from and.elim_left (h z)
```

as another example, here is how we can express the fact that a relation, \mathbf{r} , is transitive:

```

variables (α : Type) (r : α → α → Prop)
variable trans_r : ∀ x y z, r x y → r y z → r x z

variables (a b c : α)
variables (hab : r a b) (hbc : r b c)

check trans_r          -- ∀ (x y z : α), r x y → r y z → r x z
check trans_r a b c
check trans_r a b c hab
check trans_r a b c hab hbc

```

Think about what is going on here. When we instantiate `trans_r` at the values `a b c`, we end up with a proof of `r a b → r b c → r a c`. Applying this to the “hypothesis” `hab : r a b`, we get a proof of the implication `r b c → r a c`. Finally, applying it to the hypothesis `hbc` yields a proof of the conclusion `r a c`.

In situations like this, it can be tedious to supply the arguments `a b c`, when they can be inferred from `hab hbc`. For that reason, it is common to make these arguments implicit:

```

variables (α : Type) (r : α → α → Prop)
variable (trans_r : ∀ {x y z}, r x y → r y z → r x z)

variables (a b c : α)
variables (hab : r a b) (hbc : r b c)

check trans_r
check trans_r hab
check trans_r hab hbc

```

The advantage is that we can simply write `trans_r hab hbc` as a proof of `r a c`. A disadvantage is that Lean does not have enough information to infer the types of the arguments in the expressions `trans_r` and `trans_r hab`. The output of the `check` command contains expressions like `?z α r trans_r a b c hab hbc`. Such an expression indicates an arbitrary value, that may depend on any of the values listed (in this case, all the variables in the local context).

Here is an example of how we can carry out elementary reasoning with an equivalence relation:

```

variables (α : Type) (r : α → α → Prop)

variable refl_r : ∀ x, r x x
variable symm_r : ∀ {x y}, r x y → r y x
variable trans_r : ∀ {x y z}, r x y → r y z → r x z

example (a b c d : α) (hab : r a b) (hcb : r c b) (hcd : r c d) : r a d :=
trans_r (trans_r hab (symm_r hcb)) hcd

```

You might want to try to prove some of these equivalences:

```

variables (α : Type) (p q : α → Prop)

example : (∀ x, p x ∧ q x) ↔ (∀ x, p x) ∧ (∀ x, q x) := sorry
example : (∀ x, p x → q x) → (∀ x, p x) → (∀ x, q x) := sorry
example : (∀ x, p x) ∨ (∀ x, q x) → ∀ x, p x ∨ q x := sorry

```

You should also try to understand why the reverse implication is not derivable in the last example.

It is often possible to bring a component outside a universal quantifier, when it does not depend on the quantified variable (one direction of the second of these requires classical logic):

```

variables (α : Type) (p q : α → Prop)
variable r : Prop

example : α → ((∀ x : α, r) ↔ r) := sorry
example : (∀ x, p x ∨ r) ↔ (∀ x, p x) ∨ r := sorry
example : (∀ x, r → p x) ↔ (r → ∀ x, p x) := sorry

```

As a final example, consider the “barber paradox”, that is, the claim that in a certain town there is a (male) barber that shaves all and only the men who do not shave themselves. Prove that this implies a contradiction:

```

variables (men : Type) (barber : men) (shaves : men → men → Prop)

example (h : ∀ x : men, shaves barber x ↔ ¬ shaves x x) : false := sorry

```

It is the typing rule for Pi types, and the universal quantifier in particular, that distinguishes `Prop` from other types. Suppose we have $\alpha : \text{Type } i$ and $\beta : \text{Type } j$, where the expression β may depend on a variable $x : \alpha$. Then $\Pi x : \alpha, \beta$ is an element of `Type (imax i j)`, where `imax i j` is the maximum of `i` and `j` if `j` is not 0, and 0 otherwise.

The idea is as follows. If `j` is not 0, then $\Pi x : \alpha, \beta$ is an element of `Type (max i j)`. In other words, the type of dependent functions from α to β “lives” in the universe with smallest index greater-than or equal to the indices of the universes of α and β . Suppose, however, that β is of `Type 0`, that is, an element of `Prop`. In that case, $\Pi x : \alpha, \beta$ is an element of `Type 0` as well, no matter which type universe α lives in. In other words, if β is a proposition depending on α , then $\forall x : \alpha, \beta$ is again a proposition. This reflects the interpretation of `Prop` as the type of propositions rather than data, and it is what makes `Prop` *impredicative*.

The term “predicative” stems from foundational developments around the turn of the twentieth century, when logicians such as Poincaré and Russell blamed set-theoretic paradoxes on the “vicious circles” that arise when we define a property by quantifying over a collection that includes the very property being defined. Notice that if α is any type, we

can form the type $\alpha \rightarrow \mathbf{Prop}$ of all predicates on α (the “power type of α ”). The impredicativity of \mathbf{Prop} means that we can form propositions that quantify over $\alpha \rightarrow \mathbf{Prop}$. In particular, we can define predicates on α by quantifying over all predicates on α , which is exactly the type of circularity that was once considered problematic.

4.2 Equality

Let us now turn to one of the most fundamental relations defined in Lean’s library, namely, the equality relation. In a later chapter,

we will explain *how* equality is defined from the primitives of Lean’s logical framework.

In the meanwhile, here we explain how to use it.

Of course, a fundamental property of equality is that it is an equivalence relation:

```
check eq.refl    --  $\forall (a : ?M_1), a = a$ 
check eq.symm    --  $?M_2 = ?M_3 \rightarrow ?M_3 = ?M_2$ 
check eq.trans   --  $?M_2 = ?M_3 \rightarrow ?M_3 = ?M_4 \rightarrow ?M_2 = ?M_4$ 
```

Thus, for example, we can specialize the example from the previous section to the equality relation:

```
variables ( $\alpha$  : Type) (a b c d :  $\alpha$ )
premises (hab : a = b) (hcb : c = b) (hcd : c = d)

example : a = d :=
  eq.trans (eq.trans hab (eq.symm hcb)) hcd
```

If we “open” the eq namespace, the names become shorter:

```
open eq

example : a = d := trans (trans hab (symm hcb)) hcd
```

We can also use the projection notation:

```
example : a = d := (hab.trans hcb.symm).trans hcd
```

Reflexivity is more powerful than it looks. Recall that terms in the Calculus of Inductive Constructions have a computational interpretation, and that the logical framework treats terms with a common reduct as the same. As a result, some nontrivial identities can be proved by reflexivity:

```
variables (α β : Type)

example (f : α → β) (a : α) : (λ x, f x) a = f a := eq.refl _
example (a : α) (b : α) : (a, b).1 = a := eq.refl _
example : 2 + 3 = 5 := eq.refl _
```

This feature of the framework is so important that the library defines a notation `rfl` for `eq.refl _`:

```
example (f : α → β) (a : α) : (λ x, f x) a = f a := rfl
example (a : α) (b : α) : (a, b).1 = a := rfl
example : 2 + 3 = 5 := rfl
```

Equality is much more than an equivalence relation, however. It has the important property that every assertion respects the equivalence, in the sense that we can substitute equal expressions without changing the truth value. That is, given `h1 : a = b` and `h2 : P a`, we can construct a proof for `P b` using substitution: `eq.subst h1 h2`.

```
example (α : Type) (a b : α) (P : α → Prop) (h1 : a = b) (h2 : P a) : P b :=
eq.subst h1 h2

example (α : Type) (a b : α) (P : α → Prop) (h1 : a = b) (h2 : P a) : P b :=
h1 ► h2
```

The triangle in the second presentation is nothing more than notation for `eq.subst`, and you can enter it by typing `\t`.

The rule `eq.subst` is used to define the following auxiliary rules, which carry out more explicit substitutions. They are designed to deal with applicative terms, that is, terms of form `s t`. Specifically, `congr_arg` can be used to replace the argument, `congr_fun` can be used to replace the terms that is being applied, and `congr` can be used to replace both at once.

```
variable α : Type
variables a b : α
variables f g : α → ℕ
premise h1 : a = b
premise h2 : f = g

example : f a = f b := congr_arg f h1
example : f a = g a := congr_fun h2 a
example : f a = g b := congr h2 h1
```

Lean's library contains a large number of common identities, such as these:

```
variables (α : Type) [comm_ring α]
variables a b c d : α
```

```

example : a + 0 = a := add_zero a
example : 0 + a = a := zero_add a
example : a * 1 = a := mul_one a
example : 1 * a = a := one_mul a
example : -a + a = 0 := neg_add_self a
example : a + -a = 0 := add_neg_self a
example : a - a = 0 := sub_self a
example : a + b = b + a := add_comm a b
example : a + b + c = a + (b + c) := add_assoc a b c
example : a * b = b * a := mul_comm a b
example : a * b * c = a * (b * c) := mul_assoc a b c
example : a * (b + c) = a * b + a * c := mul_add a b c
example : a * (b + c) = a * b + a * c := left_distrib a b c -- alternative name
example : (a + b) * c = a * c + b * c := add_mul a b c
example : (a + b) * c = a * c + b * c := right_distrib a b c -- alternative name
example : a * (b - c) = a * b - a * c := mul_sub a b c
example : (a - b) * c = a * c - b * c := sub_mul a b c

```

Identities like these are designed to work in arbitrary instances of the relevant algebraic structures, using the type class mechanism that is described in [Chapter 10](#). [Chapter 6](#) provides some pointers as to how to find them in the library. Here is an example of a calculation in the natural numbers that uses substitution combined with associativity, commutativity, and distributivity of the natural numbers.

```

variables x y z : ℕ

example (x y z : ℕ) : x * (y + z) = x * y + x * z := mul_add x y z
example (x y z : ℕ) : x * (y + z) = x * y + x * z := mul_add x y z
example (x y z : ℕ) : x + y + z = x + (y + z) := add_assoc x y z

example (x y : ℕ) : (x + y) * (x + y) = x * x + y * x + x * y + y * y :=
have h1 : (x + y) * (x + y) = (x + y) * x + (x + y) * y, from mul_add (x + y) x y,
have h2 : (x + y) * (x + y) = x * x + y * x + (x * y + y * y),
  from (add_mul x y x) >| (add_mul x y y) >| h1,
h2.trans (add_assoc (x * x + y * x) (x * y) (y * y)).symm

```

It is often important to be able to carry out substitutions by hand, but it is tedious to prove examples like the one above in this way. Fortunately, Lean provides an environment that provides better support for such calculations, which we will turn to now.

4.3 Calculational Proofs

A calculational proof is just a chain of intermediate results that are meant to be composed by basic principles such as the transitivity of equality. In Lean, a calculation proof starts with the keyword `calc`, and has the following syntax:

```

calc
  <expr>_0 'op_1' <expr>_1 ':' <proof>_1

```

```
'...' 'op_2' <expr>_2 ':' <proof>_2
...
'...' 'op_n' <expr>_n ':' <proof>_n
```

Each `<proof>_i` is a proof for `<expr>_{i-1} op_i <expr>_i`. The `<proof>_i` may also be of the form `{ <pr> }`, where `<pr>` is a proof for some equality `a = b`. The form `{ <pr> }` is just syntactic sugar for `eq.subst <pr> (eq.refl <expr>_{i-1})`. In other words, we are claiming we can obtain `<expr>_i` by replacing `a` with `b` in `<expr>_{i-1}`.

Here is an example:

```
variables (a b c d e : ℕ)
variable h1 : a = b
variable h2 : b = c + 1
variable h3 : c = d
variable h4 : e = 1 + d

theorem T : a = e :=
calc
  a      = b      : h1
  ... = c + 1    : h2
  ... = d + 1    : congr_arg _ h3
  ... = 1 + d    : add_comm d (1 : ℕ)
  ... = e      : eq.symm h4
```

The style of writing proofs is most effective when it is used in conjunction with the `simp` and `rewrite` tactics, which are discussed in greater detail in the next chapter. For example, using the abbreviation `rw` for `rewrite`, the proof above could be written as follows:

```
variables (a b c d e : ℕ)
variable h1 : a = b
variable h2 : b = c + 1
variable h3 : c = d
variable h4 : e = 1 + d

include h1 h2 h3 h4
theorem T : a = e :=
calc
  a      = b      : by rw h1
  ... = c + 1    : by rw h2
  ... = d + 1    : by rw h3
  ... = 1 + d    : by rw add_comm
  ... = e      : by rw h4
```

In the next chapter, we will see that hypotheses can be introduced, renamed, and modified by tactics, so it is not always clear what the names in `rw h1` refer to (though, in this case, it is). For that reason, section variables and premises that only appear in a tactic command or block are not automatically add to the context. The `include` command takes care of that. Essentially, the `rewrite` tactic uses a given equality (which can a hypothesis,

a theorem name, or a complex term) to “rewrite” the goal. If doing so reduces the goal to an identity $t = t$, the tactic applies reflexivity to prove it.

Rewrites can be applied sequentially, so that the proof above can be shortened to this:

```
theorem T : a = e :=
calc
  a      = d + 1    : by rw [h1, h2, h3]
  ...    = 1 + d    : by rw add_comm
  ...    = e        : by rw h4
```

Or even this:

```
theorem T : a = e :=
by rw [h1, h2, h3, add_comm, h4]
```

The `simp` tactic, instead, rewrites the goal by applying the given identities repeatedly, in any order, anywhere they are applicable in a term. It also uses other rules that have been previously declared to the system, and applies associativity and commutativity wisely to put expressions in canonical forms. As a result, we can also prove the theorem as follows:

```
theorem T : a = e :=
by simp [h1, h2, h3, h4]
```

We will discuss variations of `rw` and `simp` in the next chapter.

The `calc` command can be configured for any relation that supports some form of transitivity. It can even combine different relations.

```
theorem T2 (a b c d : ℕ)
  (h1 : a = b) (h2 : b ≤ c) (h3 : c + 1 < d) : a < d :=
calc
  a      = b      : h1
  ...    < b + 1  : nat.lt_succ_self b
  ...    ≤ c + 1  : nat.succ_le_succ h2
  ...    < d      : h3
```

With `calc`, we can write the proof in the last section in a more natural and perspicuous way.

```
example (x y : ℕ) : (x + y) * (x + y) = x * x + y * x + x * y + y * y :=
calc
  (x + y) * (x + y) = (x + y) * x + (x + y) * y : by rw mul_add
  ...              = x * x + y * x + (x + y) * y : by rw add_mul
  ...              = x * x + y * x + (x * y + y * y) : by rw add_mul
  ...              = x * x + y * x + x * y + y * y : by rw -add_assoc
```

Here the negation before `add_assoc` tells rewrite to use the identity in the opposite direction. If brevity is what we are after, both `rw` and `simp` can do the job on their own:

```
example (x y : ℕ) : (x + y) * (x + y) = x * x + y * x + x * y + y * y :=
by rw [mul_add, add_mul, add_mul, -add_assoc]

example (x y : ℕ) : (x + y) * (x + y) = x * x + y * x + x * y + y * y :=
by simp [mul_add, add_mul, add_mul]
```

4.4 The Existential Quantifier

Finally, consider the existential quantifier, which can be written as either `exists x : α, p x` or $\exists x : \alpha, p x$. Both versions are actually notationally convenient abbreviations for a more long-winded expression, `Exists (λ x : α, p x)`, defined in Lean's library.

As you should by now expect, the library includes both an introduction rule and an elimination rule. The introduction rule is straightforward: to prove $\exists x : \alpha, p x$, it suffices to provide a suitable term `t` and a proof of `p t`. here are some examples:

```
open nat

example : ∃ x : ℕ, x > 0 :=
have h : 1 > 0, from zero_lt_succ 0,
exists.intro 1 h

example (x : ℕ) (h : x > 0) : ∃ y, y < x :=
exists.intro 0 h

example (x y z : ℕ) (hxy : x < y) (hyz : y < z) : ∃ w, x < w ∧ w < z :=
exists.intro y (and.intro hxy hyz)

check @exists.intro
```

We can use the anonymous constructor notation $\langle t, h \rangle$ for `exists.intro t h`, when the type is clear from the context.

```
example : ∃ x : ℕ, x > 0 :=
⟨1, zero_lt_succ 0⟩

example (x : ℕ) (h : x > 0) : ∃ y, y < x :=
⟨0, h⟩

example (x y z : ℕ) (hxy : x < y) (hyz : y < z) : ∃ w, x < w ∧ w < z :=
⟨y, hxy, hyz⟩
```

Note that `exists.intro` has implicit arguments: Lean has to infer the predicate $p : \alpha \rightarrow \text{Prop}$ in the conclusion $\exists x, p x$. This is not a trivial affair. For example, if we have `hg : g 0 0 = 0` and write `exists.intro 0 hg`, there are many possible values for

the predicate `p`, corresponding to the theorems $\exists x, g\ x\ x = x$, $\exists x, g\ x\ x = 0$, $\exists x, g\ x\ 0 = x$, etc. Lean uses the context to infer which one is appropriate. This is illustrated in the following example, in which we set the option `pp.implicit` to true to ask Lean’s pretty-printer to show the implicit arguments.

```
variable g : ℕ → ℕ → ℕ
variable hg : g 0 0 = 0

theorem gex1 : ∃ x, g x x = x := ⟨0, hg⟩
theorem gex2 : ∃ x, g x 0 = x := ⟨0, hg⟩
theorem gex3 : ∃ x, g 0 0 = x := ⟨0, hg⟩
theorem gex4 : ∃ x, g x x = 0 := ⟨0, hg⟩

set_option pp.implicit true -- display implicit arguments
check gex1
check gex2
check gex3
check gex4
```

We can view `exists.intro` as an information-hiding operation: we are “hiding” the witness to the body of the assertion. The existential elimination rule, `exists.elim`, performs the opposite operation. It allows us to prove a proposition `q` from $\exists x : \alpha, p\ x$, by showing that `q` follows from `p w` for an arbitrary value `w`. Roughly speaking, since we know there is an `x` satisfying `p x`, we can give it a name, say, `w`. If `q` does not mention `w`, then showing that `q` follows from `p w` is tantamount to showing the `q` follows from the existence of any such `x`. Here is an example:

```
variables (α : Type) (p q : α → Prop)

example (h : ∃ x, p x ∧ q x) : ∃ x, q x ∧ p x :=
exists.elim h
  (take w,
    assume hw : p w ∧ q w,
    show ∃ x, q x ∧ p x, from ⟨w, hw.right, hw.left⟩)
```

It may be helpful to compare the exists-elimination rule to the or-elimination rule: the assertion $\exists x : \alpha, p\ x$ can be thought of as a big disjunction of the propositions `p a`, as `a` ranges over all the elements of α .

Notice that an existential proposition is very similar to a sigma type, as described in [Section 2.8](#). The difference is that given `a : α` and `h : p a`, the term `exists.intro a h` has type $(\exists x : \alpha, p\ x) : \text{Prop}$ and `sigma.mk a h` has type $(\Sigma x : \alpha, p\ x) : \text{Type}$. The similarity between \exists and Σ is another instance of the Curry-Howard isomorphism.

Lean provides a more convenient way to eliminate from an existential quantifier with the `match` statement:

```
variables (α : Type) (p q : α → Prop)
```

```
example (h :  $\exists x, p\ x \wedge q\ x$ ) :  $\exists x, q\ x \wedge p\ x$  :=
match h with <w, hw> :=
  <w, hw.right, hw.left>
end
```

The `match` statement is part of Lean’s function definition system, which provides a convenient and expressive ways of defining complex functions. Once again, it is the Curry-Howard isomorphism that allows us to co-opt this mechanism for writing proofs as well. The `match` statement “destructs” the existential assertion into the components `w` and `hw`, which can then be used in the body of the statement to prove the proposition. We can annotate the types used in the match for greater clarity:

```
example (h :  $\exists x, p\ x \wedge q\ x$ ) :  $\exists x, q\ x \wedge p\ x$  :=
match h with <(w :  $\alpha$ ), (hw :  $p\ w \wedge q\ w$ )> :=
  <w, hw.right, hw.left>
end
```

We can even use the match statement to decompose the conjunction at the same time:

```
example (h :  $\exists x, p\ x \wedge q\ x$ ) :  $\exists x, q\ x \wedge p\ x$  :=
match h with <w, hpw, hqw> :=
  <w, hqw, hpw>
end
```

Lean will even allow us to use an implicit `match` in the `assume` statement:

```
example : ( $\exists x, p\ x \wedge q\ x$ )  $\rightarrow \exists x, q\ x \wedge p\ x$  :=
assume <w, hpw, hqw>, <w, hqw, hpw>
```

In the following example, we define `even a` as $\exists b, a = 2*b$, and then we show that the sum of two even numbers is an even number.

```
definition is_even (a : nat) :=  $\exists b, a = 2 * b$ 

theorem even_plus_even {a b : nat} (h1 : is_even a) (h2 : is_even b) : is_even (a + b) :=
exists.elim h1 (take w1, assume hw1 : a = 2 * w1,
exists.elim h2 (take w2, assume hw2 : b = 2 * w2,
  exists.intro (w1 + w2)
    (calc
      a + b = 2 * w1 + 2 * w2 : by rw [hw1, hw2]
      ... = 2*(w1 + w2)      : by rw mul_add)))
```

Using the various gadgets described in this chapter — the match statement, anonymous constructors, and the `rewrite` tactic, we can write this proof concisely as follows:

```

theorem even_plus_even {a b : nat} (h1 : is_even a) (h2 : is_even b) : is_even (a + b) :=
match h1, h2 with
| ⟨w1, hw1⟩, ⟨w2, hw2⟩ := ⟨w1 + w2, by rw [hw1, hw2, mul_add]⟩
end

```

Just as the constructive “or” is stronger than the classical “or,” so, too, is the constructive “exists” stronger than the classical “exists”. For example, the following implication requires classical reasoning because, from a constructive standpoint, knowing that it is not the case that every x satisfies $\neg p$ is not the same as having a particular x that satisfies p .

```

open classical

variables (α : Type) (p : α → Prop)

example (h : ¬ ∀ x, ¬ p x) : ∃ x, p x :=
by_contradiction
  (assume h1 : ¬ ∃ x, p x,
   have h2 : ∀ x, ¬ p x, from
     take x,
     assume h3 : p x,
     have h4 : ∃ x, p x, from ⟨x, h3⟩,
     show false, from h1 h4,
   show false, from h h2)

```

What follows are some common identities involving the existential quantifier. We encourage you to prove as many as you can. We are also leaving it to you to determine which are nonconstructive, and hence require some form of classical reasoning.

```

open classical

variables (α : Type) (p q : α → Prop)
variable a : α
variable r : Prop

example : (∃ x : α, r) → r := sorry
example : r → (∃ x : α, r) := sorry
example : (∃ x, p x ∧ r) ↔ (∃ x, p x) ∧ r := sorry
example : (∃ x, p x ∨ q x) ↔ (∃ x, p x) ∨ (∃ x, q x) := sorry

example : (∀ x, p x) ↔ ¬ (∃ x, ¬ p x) := sorry
example : (∃ x, p x) ↔ ¬ (∀ x, ¬ p x) := sorry
example : (¬ ∃ x, p x) ↔ (∀ x, ¬ p x) := sorry
example : (¬ ∀ x, p x) ↔ (∃ x, ¬ p x) := sorry

example : (∀ x, p x → r) ↔ (∃ x, p x) → r := sorry
example : (∃ x, p x → r) ↔ (∀ x, p x) → r := sorry
example : (∃ x, r → p x) ↔ (r → ∃ x, p x) := sorry

```

Notice that the declaration `variable a : α` amounts to the assumption that there is at least one element of type α . This assumption is needed in the second example, as well as in the last two.

Here are solutions to two of the more difficult ones:

```

example : ( $\exists x, p\ x \vee q\ x$ )  $\leftrightarrow$  ( $\exists x, p\ x$ )  $\vee$  ( $\exists x, q\ x$ ) :=
iff.intro
  (assume ⟨a, (h1 : p a  $\vee$  q a)⟩,
    or.elim h1
      (assume hpa : p a, or.inl ⟨a, hpa⟩)
      (assume hqa : q a, or.inr ⟨a, hqa⟩))
  (assume h : ( $\exists x, p\ x$ )  $\vee$  ( $\exists x, q\ x$ ),
    or.elim h
      (assume ⟨a, hpa⟩, ⟨a, (or.inl hpa)⟩)
      (assume ⟨a, hqa⟩, ⟨a, (or.inr hqa)⟩))

example : ( $\exists x, p\ x \rightarrow r$ )  $\leftrightarrow$  ( $\forall x, p\ x \rightarrow r$ ) :=
iff.intro
  (assume ⟨b, (hb : p b  $\rightarrow$  r)⟩,
    assume h2 :  $\forall x, p\ x$ ,
    show r, from hb (h2 b))
  (assume h1 : ( $\forall x, p\ x$ )  $\rightarrow$  r,
    show  $\exists x, p\ x \rightarrow r$ , from
      by_cases
        (assume hap :  $\forall x, p\ x$ , ⟨a,  $\lambda h'$ , h1 hap⟩)
        (assume hnep :  $\neg \forall x, p\ x$ ,
          by_contradiction
            (assume hnex :  $\neg \exists x, p\ x \rightarrow r$ ,
              have hap :  $\forall x, p\ x$ , from
                take x,
                by_contradiction
                  (assume hnp :  $\neg p\ x$ ,
                    have hex :  $\exists x, p\ x \rightarrow r$ ,
                      from ⟨x, (assume hp, absurd hp hnp)⟩,
                      show false, from hnex hex),
                    show false, from hnep hap)))
  )
  )

```

4.5 More on the Proof Language

We have seen that keywords like `assume`, `take`, `have`, and `show` make it possible to write formal proof terms that mirror the structure of informal mathematical proofs. In this section, we discuss some additional features of the proof language that are often convenient.

To start with, we can use anonymous “have” expressions to introduce an auxiliary goal without having to label it. We can refer to the last expression introduced in this way using the keyword `this`:

```

variable f :  $\mathbb{N} \rightarrow \mathbb{N}$ 
premise h :  $\forall x : \mathbb{N}, f\ x \leq f\ (x + 1)$ 

example : f 0  $\leq$  f 3 :=

```

```

have f 0 ≤ f 1, from h 0,
have f 0 ≤ f 2, from le_trans this (h 1),
show f 0 ≤ f 3, from le_trans this (h 2)

```

Often proofs move from one fact to the next, so this can be effective in eliminating the clutter of lots of labels.

When the goal can be inferred, we can also ask Lean instead to fill in the proof by writing `by assumption`:

```

variable f : ℕ → ℕ
premise h : ∀ x : ℕ, f x ≤ f (x + 1)

example : f 0 ≤ f 3 :=
have f 0 ≤ f 1, from h 0,
have f 0 ≤ f 2, from le_trans (by assumption) (h 1),
show f 0 ≤ f 3, from le_trans (by assumption) (h 2)

```

This tells Lean to use the `assumption` tactic, which, in turn, proves the goal by finding a suitable hypothesis in the local context. We will learn more about the `assumption` tactic in the next chapter.

We can also ask Lean to fill in the proof by writing `<p>`, where `p` is the proposition whose proof we want Lean to find in the context.

```

example : f 0 ≥ f 1 → f 1 ≥ f 2 → f 0 = f 2 :=
suppose f 0 ≥ f 1,
suppose f 1 ≥ f 2,
have f 0 ≥ f 2, from le_trans this <f 0 ≥ f 1>,
have f 0 ≤ f 2, from le_trans (h 0) (h 1),
show f 0 = f 2, from le_antisymm this <f 0 ≥ f 2>

```

You can type these corner quotes using `\f<` and `\f>`, respectively. The letter “f” is for “French,” since the unicode symbols can also be used as French quotation marks. In fact, the notation is defined in Lean as follows:

```

notation `<` p `>` := show p, by assumption

```

This approach is more robust than using `by assumption`, because the type of the assumption that needs to be inferred is given explicitly. It also makes proofs more readable. Here is a more elaborate example:

```

example : f 0 ≤ f 3 :=
have f 0 ≤ f 1, from h 0,
have f 1 ≤ f 2, from h 1,
have f 2 ≤ f 3, from h 2,
show f 0 ≤ f 3, from le_trans <f 0 ≤ f 1>
    (le_trans <f 1 ≤ f 2> <f 2 ≤ f 3>)

```

Keep in mind that use can use the French quotation marks in this way to refer to *anything* in the context, not just things that were introduced anonymously. Its use is also not limited to propositions, though using it for data is somewhat odd:

```
example (n : ℕ) : ℕ := <ℕ>
```

The **suppose** keyword acts as an anonymous assume:

```
example : f 0 ≥ f 1 → f 0 = f 1 :=
  suppose f 0 ≥ f 1,
  show f 0 = f 1, from le_antisymm (h 0) this
```

Notice that there is an asymmetry: you can use **have** with or without a label, but if you do not wish to name the assumption, you must use **suppose** rather than **assume**. The reason is that Lean allows us to write **assume h** to introduce a hypothesis without specifying it, leaving it to the system to infer to relevant assumption. An anonymous **assume** would thus lead to ambiguities when parsing expressions.

As with the anonymous **have**, when you use **suppose** to introduce an assumption, that assumption can also be invoked later in the proof by enclosing it in backticks.

```
example : f 0 ≥ f 1 → f 1 ≥ f 2 → f 0 = f 2 :=
  suppose f 0 ≥ f 1,
  suppose f 1 ≥ f 2,
  have f 0 ≥ f 2, from le_trans <f 2 ≤ f 1> <f 1 ≤ f 0>,
  have f 0 ≤ f 2, from le_trans (h 0) (h 1),
  show f 0 = f 2, from le_antisymm this <f 0 ≥ f 2>
```

Notice that **le_antisymm** is the assertion that if $a \leq b$ and $b \leq a$ then $a = b$, and $a \geq b$ is definitionally equal to $b \leq a$.

Tactics

In this chapter, we describe an alternative approach to constructing proofs, using *tactics*. A proof term is a representation of a mathematical proof; tactics are commands, or instructions, that describe how to build such a proof. Informally, we might begin a mathematical proof by saying “to prove the forward direction, unfold the definition, apply the previous lemma, and simplify.” Just as these are instructions that tell the reader how to find the relevant proof, tactics are instructions that tell Lean how to construct a proof term. They naturally support an incremental style of writing proofs, in which users decompose a proof and work on goals one step at a time.

We will describe proofs that consist of sequences of tactics as “tactic-style” proofs, to contrast with the ways of writing proof terms we have seen so far, which we will call “term-style” proofs. Each style has its own advantages and disadvantages. For example, tactic-style proofs can be harder to read, because they require the reader to predict or guess the results of each instruction. But they can also be shorter and easier to write. Moreover, tactics offer a gateway to using Lean’s automation, since automated procedures are themselves tactics.

5.1 Entering Tactic Mode

Conceptually, stating a theorem or introducing a **have** statement creates a goal, namely, the goal of constructing a term with the expected type. For example, the following creates the goal of constructing a term of type $p \wedge q \wedge p$, in a context with constants $p \ q : \text{Prop}$, $hp : p$ and $hq : q$:

```
theorem test (p q : Prop) (hp : p) (hq : q) : p ∧ q ∧ p :=
  sorry
```

We can write this goal as follows:

```
p : Prop, q : Prop, hp : p, hq : q ⊢ p ∧ q ∧ p
```

Indeed, if you replace the “sorry” by an underscore in the example above, Lean will report that it is exactly this goal that has been left unsolved.

Ordinarily, we meet such a goal by writing an explicit term. But wherever a term is expected, Lean allows us to insert instead a **begin** ... **end** block, followed by a sequence of commands, separated by commas. We can prove the theorem above in that way:

```
theorem test (p q : Prop) (hp : p) (hq : q) : p ∧ q ∧ p :=
begin
  apply and.intro,
  exact hp,
  apply and.intro,
  exact hq,
  exact hp
end
```

The **apply** tactic applies an expression, viewed as denoting a function with zero or more arguments. It unifies the conclusion with the expression in the current goal, and creates new goals for the remaining arguments, provided that no later arguments depend on them. In the example above, the command **apply and.intro** yields two subgoals:

```
p : Prop,
q : Prop,
hp : p,
hq : q
⊢ p

⊢ q ∧ p
```

For brevity, Lean only displays the context for the first goal, which is the one addressed by the next tactic command. The first goal is met with the command **exact hp**. The **exact** command is just a variant of **apply** which signals that the expression given should fill the goal exactly. It is good form to use it in a tactic proof, since its failure signals that something has gone wrong; but otherwise **apply** would work just as well.

You can see the resulting proof term with the **print** command:

```
print test
```

You can write a tactic script incrementally. If you run Lean on an incomplete tactic proof bracketed by **begin** and **end**, the system reports all the unsolved goals that remain.

If you are running Lean with its Emacs interface, you can see this information by putting your cursor on the `end` symbol, which should be underlined. In the Emacs interface, there is another extremely useful trick: if you put your cursor on a line of a tactic proof and press “C-c C-g”, Lean will show you the goal that remains at the end of the line.

Tactic commands can take compound expressions, not just single identifiers. The following is a shorter version of the preceding proof:

```
theorem test (p q : Prop) (hp : p) (hq : q) : p ∧ q ∧ p :=
begin
  apply and.intro hp,
  exact and.intro hq hp
end
```

Unsurprisingly, it produces exactly the same proof term.

```
print test
```

Whenever a proof term is expected, instead of using a `begin...end` block, you can write the `by` keyword followed by a single tactic:

```
theorem test (p q : Prop) (hp : p) (hq : q) : p ∧ q ∧ p :=
by exact and.intro hp (and.intro hq hp)
```

In the Lean Emacs mode, if you put your cursor on the “b” in “by” and press “C-c C-g”, Lean shows you the goal that the tactic is supposed to meet.

We will see below that hypothesis and be introduced, reverted, modified, and renamed over the course of a tactic block. As a result, it is impossible for the Lean parser to detect when an identifier that occurs in a tactic block refers to a section variable that should therefore be added to the context. As a result, you need to explicitly tell Lean to include the relevant entities:

```
variables {p q : Prop} (hp : p) (hq : q)

include hp hq

example : p ∧ q ∧ p :=
begin
  apply and.intro hp,
  exact and.intro hq hp
end
```

The `include` command tells Lean to include the indicated variables (as well as any variables they depend on) from that point on, until the end of the section or file. To limit the effect of an `include`, you can use the `omit` command afterwards:

```
include hp hq

example : p ∧ q ∧ p :=
begin
  apply and.intro hp,
  exact and.intro hq hp
end

omit hp hq

-- hp and hq are no longer included by default
```

Alternatively, you can use a section to delimit the scope.

```
section
include hp hq

example : p ∧ q ∧ p :=
begin
  apply and.intro hp,
  exact and.intro hq hp
end
end

-- hp and hq are no longer included by default
```

Another workaround is to find a way to refer to the variable in question before entering a tactic block:

```
example : p ∧ q ∧ p :=
let hp := hp, hq := hq in
begin
  apply and.intro hp,
  exact and.intro hq hp
end
```

Any mention of `hp` or `hq` at all will cause it to be added to the hypotheses in the example.

5.2 Basic Tactics

In addition to `apply` and `exact`, another useful tactic is `intro`, which introduces a hypothesis. What follows is an example of an identity from propositional logic that we proved [Section 3.5](#), now proved using tactics. We adopt the following convention regarding indentation: whenever a tactic introduces one or more additional subgoals, we indent another two spaces, until the additional subgoals are deleted. That rationale behind this convention, and other structuring mechanisms, will be discussed in [Section 5.4](#) below.

```

example (p q r : Prop) : p ∧ (q ∨ r) ↔ (p ∧ q) ∨ (p ∧ r) :=
begin
  apply iff.intro,
  intro h,
  apply or.elim (and.elim_right h),
  intro hq,
  apply or.intro_left,
  apply and.intro,
  exact and.elim_left h,
  exact hq,
  intro hr,
  apply or.intro_right,
  apply and.intro,
  exact and.elim_left h,
  exact hr,
  intro h,
  apply or.elim h,
  intro hpq,
  apply and.intro,
  exact and.elim_left hpq,
  apply or.intro_left,
  exact and.elim_right hpq,
  intro hpr,
  apply and.intro,
  exact and.elim_left hpr,
  apply or.intro_right,
  exact and.elim_right hpr
end

```

The `intro` command can more generally be used to introduce a variable of any type:

```

example (α : Type) : α → α :=
begin
  intro a,
  exact a
end

example (α : Type) : ∀ x : α, x = x :=
begin
  intro x,
  exact eq.refl x
end

```

It has a plural form, `intros`, which takes a list of names.

```

example : ∀ a b c : ℕ, a = b → a = c → c = b :=
begin
  intros a b c h1 h2,
  exact eq.trans (eq.symm h2) h1
end

```

The `intros` command can also be used without any arguments, in which case, it chooses names and introduces as many variables as it can. We will see an example of this in a moment.

The `assumption` tactic looks through the assumptions in context of the current goal, and if there is one matching the conclusion, it applies it.

```
example (h1 : x = y) (h2 : y = z) (h3 : z = w) : x = w :=
begin
  apply eq.trans h1,
  apply eq.trans h2,
  assumption -- applied h3
end
```

It will unify metavariables in the conclusion if necessary:

```
example (h1 : x = y) (h2 : y = z) (h3 : z = w) : x = w :=
begin
  apply eq.trans,
  assumption, -- solves x = ?b with h1
  apply eq.trans,
  assumption, -- solves ?b = w with h2
  assumption -- solves z = w with h3
end
```

The following example uses the `intros` command to introduce the three variables and two hypotheses automatically:

```
example : ∀ a b c : ℕ, a = b → a = c → c = b :=
begin
  intros,
  apply eq.trans,
  apply eq.symm,
  assumption,
  assumption
end
```

There are tactics `reflexivity`, `symmetry`, and `transitivity`, which apply the corresponding operation. Using `reflexivity`, for example, is more general than writing `apply eq.refl`, because it works for any relation that has been tagged with the `refl` attribute.

With that tactic, the previous proof can be written more elegantly as follows:

```
example : ∀ a b c : ℕ, a = b → a = c → c = b :=
begin
  intros,
  transitivity,
  symmetry,
  assumption,
  assumption
end
```

The **repeat** combinator can be used to simplify the last two lines:

```
example :  $\forall a\ b\ c : \mathbb{N}, a = b \rightarrow a = c \rightarrow c = b :=$ 
begin
  intros,
  apply eq.trans,
  apply eq.symm,
  repeat { assumption }
end
```

The curly braces introduce a new tactic block; they are equivalent to a using a nested **begin ... end** pair, as discussed in the next section.

There is variant of **apply** called **fapply** that is more aggressive in creating new subgoals for arguments. Here is an example of how it is used:

```
example :  $\exists a : \mathbb{N}, a = a :=$ 
begin
  fapply exists.intro,
  exact 0,
  apply rfl
end
```

Here, the command **fapply exists.intro** creates two goals. The first is to provide a natural number, **a**, and the second is to prove that **a = a**. Notice that the second goal depends on the first; solving the first goal instantiates a metavariable in the second.

Another tactic that is sometimes useful is the **generalize** tactic, which is, in a sense, an inverse to **intro**.

```
variables x y z :  $\mathbb{N}$ 

example :  $x = x :=$ 
begin
  generalize x z, -- goal is  $x : \mathbb{N} \vdash \forall (z : \mathbb{N}), z = z$ 
  intro y,       -- goal is  $x\ y : \mathbb{N} \vdash y = y$ 
  reflexivity
end
```

The **generalize** tactic generalizes the conclusion over the variable **x**, using a universal quantifier over **z**. We can generalize any term, not just a variable:

```
example :  $x + y + z = x + y + z :=$ 
begin
  generalize (x + y + z) w, -- goal is  $x\ y\ z : \mathbb{N} \vdash \forall (w : \mathbb{N}), w = w$ 
  intro u,                -- goal is  $x\ y\ z\ u : \mathbb{N} \vdash u = u$ 
  reflexivity
end
```

If the expression passed as the first argument to **generalize** is not found in the goal, **generalize** raises an error.

Notice that once we generalize over $x + y + z$, the variables $x\ y\ z : \mathbb{N}$ in the context become irrelevant. The **clear** tactic throws away elements of the context, when it is safe to do so:

```
example : x + y + z = x + y + z :=
begin
  generalize (x + y + z) w, -- goal is x y z : ℕ ⊢ ∀ (w : ℕ), w = w
  clear x y z,
  intro u,                  -- goal is u : ℕ ⊢ u = u
  reflexivity
end
```

Another useful tactic is the **revert** tactic, which moves an element of the context into the goal. When applied to a variable that occurs in the goal, it has the same effect as **generalize** and **clear**:

```
example (x : ℕ) : x = x :=
begin
  revert x,      -- goal is ⊢ ∀ (x : ℕ), x = x
  intro y,      -- goal is y : ℕ ⊢ y = y
  reflexivity
end
```

Moving a hypothesis into the goal yields an implication:

```
example (x y : ℕ) (h : x = y) : y = x :=
begin
  revert h,      -- goal is x y : ℕ ⊢ x = y → y = x
  intro h₁,      -- goal is x y : ℕ, h₁ : x = y ⊢ y = x
  symmetry,
  assumption
end
```

But **revert** is even more clever, in that it will revert not only an element of the context but also all the subsequent elements of the context that depend on it. For example, reverting x in the example above brings h along with it:

```
example (x y : ℕ) (h : x = y) : y = x :=
begin
  revert x,      -- goal is y : ℕ ⊢ ∀ (x : ℕ), x = y → y = x
  intros,
  symmetry,
  assumption
end
```

You can also revert multiple elements of the context at once:

```

example (x y : ℕ) (h : x = y) : y = x :=
begin
  revert x y,      -- goal is ⊢ ∀ (x y : ℕ), x = y → y = x
  intros,
  symmetry,
  assumption
end

```

5.3 More tactics

Some additional tactics are useful for constructing and destructing propositions and data. For example, when applied to a goal of the form $p \vee q$, the tactics `left` and `right` are equivalent `apply or.inl` and `apply or.inr`, respectively. Conversely, the `cases` tactic can be used to decompose a disjunction.

```

example (p q : Prop) : p ∨ q → q ∨ p :=
begin
  intro h,
  cases h with hp hq,
  -- case hp : p
  right, exact hp,
  -- case hq : q
  left, exact hq
end

```

After `cases h` is applied, there are two goals. In the first, the hypothesis $h : p \vee q$ is replaced by $hp : p$, and in the second, it is replaced by $hq : q$. The `cases` can also be used to decompose a conjunction.

```

example (p q : Prop) : p ∧ q → q ∧ p :=
begin
  intro h,
  cases h with hp hq,
  constructor, exact hq, exact hp
end

```

In this case, there is only one goal after the `cases` tactic is applied, with $h : p \wedge q$ replaced by a pair of assumptions, $hp : p$ and $hq : q$. The constructor applies the unique constructor for conjunction, `and.intro`. With these tactics, an example from the previous section can be rewritten as follows:

```

example (p q r : Prop) : p ∧ (q ∨ r) ↔ (p ∧ q) ∨ (p ∧ r) :=
begin
  apply iff.intro,
  intro h,
  cases h with hp hqr,
  cases hqr with hq hr,

```

```

    left, constructor, repeat { assumption },
    right, constructor, repeat { assumption },
  intro h,
  cases h with hpq hpr,
  cases hpq with hp hq,
    constructor, exact hp, left, exact hq,
  cases hpr with hp hr,
    constructor, exact hp, right, exact hr
end

```

We will see in [Chapter 7](#) that these tactics are quite general. The `cases` tactic can be used to decompose any element of an inductively defined type; `constructor` always applies the first constructor of an inductively type, and `left` and `right` can be used with inductively defined types with exactly two constructors. For example, we can use `cases` and `constructor` with an existential quantifier:

```

example (p q :  $\mathbb{N} \rightarrow \text{Prop}$ ) : ( $\exists x, p\ x$ )  $\rightarrow \exists x, p\ x \vee q\ x$  :=
begin
  intro h,
  cases h with x px,
  constructor, left, exact px
end

```

Here, the `constructor` tactic leaves the first component of the existential assertion, the value of x , implicit. It is represented by a metavariable, which should be instantiated later on. In the previous example, the proper value of the metavariable is determined by the tactic `exact px`, since px has type $p\ x$. If you want to specify a witness to the existential quantifier explicitly, you can use the `existsi` tactic instead:

```

example (p q :  $\mathbb{N} \rightarrow \text{Prop}$ ) : ( $\exists x, p\ x$ )  $\rightarrow \exists x, p\ x \vee q\ x$  :=
begin
  intro h,
  cases h with x px,
  existsi x, left, exact px
end

```

These tactics can be used on data just as well as propositions. In the next two examples, they are used to define functions which swap the components of the product and sum types:

```

universe variables u v

def swap_pair { $\alpha$  : Type u} { $\beta$  : Type v} :  $\alpha \times \beta \rightarrow \beta \times \alpha$  :=
begin
  intro p,
  cases p with ha hb,
  constructor, exact hb, exact ha
end

```

```
def swap_sum {α : Type u} {β : Type v} : α ⊕ β → β ⊕ α :=
begin
  intro p,
  cases p with ha hb,
  right, exact ha,
  left, exact hb
end
```

Note that up to the names we have chosen for the variables, the definitions are identical to the proofs of the analogous propositions for conjunction and disjunction. The `cases` tactic will also do a case distinction on a natural number:

```
open nat

example (P : ℕ → Prop) (h₀ : P 0) (h₁ : ∀ n, P (succ n)) (m : ℕ) : P m :=
begin
  cases m with m', exact h₀, exact h₁ m'
end
```

For further discussion, see [Chapter 7](#).

5.4 Structuring Tactic Proofs

Tactics often provide an efficient way of building a proof, but long sequences of instructions can obscure the structure of the argument. In this section, we describe some means that help provide structure to a tactic-style proof, making such proofs more readable and robust.

One thing that is nice about Lean’s proof-writing syntax is that it is possible to mix term-style and tactic-style proofs, and pass between the two freely. For example, the tactics `apply` and `exact` expect arbitrary terms, which you can write using `have`, `show`, and so on. Conversely, when writing an arbitrary Lean term, you can always invoke the tactic mode by inserting a `begin...end` block. The following is a somewhat toy example:

```
example (p q r : Prop) : p ∧ (q ∨ r) → (p ∧ q) ∨ (p ∧ r) :=
begin
  intro h,
  exact
    have hp : p, from h.left,
    have hqr : q ∨ r, from h.right,
    show (p ∧ q) ∨ (p ∧ r),
    begin
      cases hqr with hq hr,
      exact or.inl ⟨hp, hq⟩,
      exact or.inr ⟨hp, hr⟩
    end
end
```

The following is a more natural example:

```

example (p q r : Prop) : p ∧ (q ∨ r) ↔ (p ∧ q) ∨ (p ∧ r) :=
begin
  apply iff.intro,
  intro h,
  cases h^.right with hq hr,
  exact
    show (p ∧ q) ∨ (p ∧ r),
      from or.inl ⟨h^.left, hq⟩,
  exact
    show (p ∧ q) ∨ (p ∧ r),
      from or.inr ⟨h^.left, hr⟩,
  intro h,
  cases h with hpq hpr,
  exact
    show p ∧ (q ∨ r),
      from ⟨hpq^.left, or.inl hpq^.right⟩,
  exact show p ∧ (q ∨ r),
    from ⟨hpr^.left, or.inr hpr^.right⟩
end

```

With the `exact` tactic, we use `show` to indicate the goal at that point in the proof. In fact, this idiom is so useful that Lean offers the following abbreviation: in a tactic block, the expression `show p, from t` abbreviates `exact (show p, from t)`. Thus we could have written the previous example more concisely as follows:

```

example (p q r : Prop) : p ∧ (q ∨ r) ↔ (p ∧ q) ∨ (p ∧ r) :=
begin
  apply iff.intro,
  intro h,
  cases h^.right with hq hr,
  show (p ∧ q) ∨ (p ∧ r),
    from or.inl ⟨h^.left, hq⟩,
  show (p ∧ q) ∨ (p ∧ r),
    from or.inr ⟨h^.left, hr⟩,
  intro h,
  cases h with hpq hpr,
  show p ∧ (q ∨ r),
    from ⟨hpq^.left, or.inl hpq^.right⟩,
  show p ∧ (q ∨ r),
    from ⟨hpr^.left, or.inr hpr^.right⟩
end

```

This blurs the distinction between proof-term mode and tactic-mode, and so it is important to use indentation and the structuring mechanisms discussed below to make it clear where a proof term ends. The convention we have used for indentation will be explained momentarily.

In the same way, in a tactic block, Lean interprets `have p, from t1, t2` as an abbreviation for `exact (have p, from t1, t2)`. Thus the first example in this section could have been written more concisely as follows:

```

example (p q r : Prop) : p ∧ (q ∨ r) → (p ∧ q) ∨ (p ∧ r) :=
begin
  intro h,
  have hp : p, from h.left,
  have hqr : q ∨ r, from h.right,
  show (p ∧ q) ∨ (p ∧ r),
  begin
    cases hqr with hq hr,
    exact or.inl ⟨hp, hq⟩,
    exact or.inr ⟨hp, hr⟩
  end
end

```

You can also nest `begin...end` blocks within other `begin...end` blocks. In a nested block, Lean focuses on the first goal, and generates an error if it has not been fully solved at the end of the block. This can be helpful in indicating the separate proofs of multiple subgoals introduced by a tactic.

```

example (p q r : Prop) : p ∧ (q ∨ r) ↔ (p ∧ q) ∨ (p ∧ r) :=
begin
  apply iff.intro,
  begin
    intro h,
    cases h.right with hq hr,
    begin
      show (p ∧ q) ∨ (p ∧ r),
      from or.inl ⟨h.left, hq⟩
    end,
    show (p ∧ q) ∨ (p ∧ r),
    from or.inr ⟨h.left, hr⟩
  end,
  intro h,
  cases h with hpq hpr,
  begin
    show p ∧ (q ∨ r), from
      ⟨hpq.left, or.inl hpq.right⟩
  end,
  show p ∧ (q ∨ r), from
    ⟨hpr.left, or.inr hpr.right⟩
end

```

Here, we have introduced a new `begin...end` block whenever a tactic leaves more than one subgoal. You can check (using `C-c C-g` in Emacs mode, for example) that every line in this proof, there is only one goal visible. Notice that you still need to use a comma after a `begin...end` block when there are remaining goals to be discharged.

Within a `begin...end` block, you can abbreviate nested occurrences of `begin` and `end` with curly braces:

```

example (p q r : Prop) : p ∧ (q ∨ r) ↔ (p ∧ q) ∨ (p ∧ r) :=
begin
  apply iff.intro,
  { intro h,

```

```

    cases h^.right with hq hr,
  { show (p ∧ q) ∨ (p ∧ r),
    from or.inl ⟨h^.left, hq⟩ },
  show (p ∧ q) ∨ (p ∧ r),
    from or.inr ⟨h^.left, hr⟩ },
intro h,
cases h with hpq hpr,
{ show p ∧ (q ∨ r),
  from ⟨hpq^.left, or.inl hpq^.right⟩ },
show p ∧ (q ∨ r),
  from ⟨hpr^.left, or.inr hpr^.right⟩
end

```

This helps explain the convention on indentation we have adopted here: every time a tactic leaves more than one subgoal, we separate the remaining subgoals by enclosing them in blocks and indenting, until we are back down to one subgoal. Thus if the application of theorem `foo` to a single goal produces four subgoals, one would expect the proof to look like this:

```

begin
  apply foo,
  { ... proof of first goal ... },
  { ... proof of second goal ... },
  { ... proof of third goal ... },
  proof of final goal
end

```

Another reasonable convention is to enclose *all* the remaining subgoals in indented blocks, including the last one:

```

example (p q r : Prop) : p ∧ (q ∨ r) ↔ (p ∧ q) ∨ (p ∧ r) :=
begin
  apply iff.intro,
  { intro h,
    cases h^.right with hq hr,
    { show (p ∧ q) ∨ (p ∧ r),
      from or.inl ⟨h^.left, hq⟩ },
    { show (p ∧ q) ∨ (p ∧ r),
      from or.inr ⟨h^.left, hr⟩ }},
  { intro h,
    cases h with hpq hpr,
    { show p ∧ (q ∨ r),
      from ⟨hpq^.left, or.inl hpq^.right⟩ },
    { show p ∧ (q ∨ r),
      from ⟨hpr^.left, or.inr hpr^.right⟩ }}
end

```

With this convention, the proof using `foo` described above would look like this:

```

begin
  apply foo,

```

```

{ ... proof of first goal ... },
{ ... proof of second goal ... },
{ ... proof of third goal ... },
{ ... proof of final goal ....}
end

```

Both conventions are reasonable. The second convention has the effect that the text in a long proof gradually creeps to the right. Many theorems in mathematics have side conditions that can be dispelled quickly; using the first convention means that the proofs of these side conditions are indented until we return to the “linear” part of the proof.

You can simulate the effect of the `have` construct without leaving tactic mode using the `assert` tactic.

```

example (p q : Prop) : p ∧ q ↔ q ∧ p :=
begin
  apply iff.intro,
  { intro h,
    assert hp : p, exact h.left,
    assert hq : q, exact h.right,
    exact ⟨hq, hp⟩ },
  intro h,
  assert hp : p, exact h.right,
  assert hq : q, exact h.left,
  exact ⟨hp, hq⟩
end

```

Here, the first `assert` creates a new subgoal, `p`. After that subgoal is proved, we are left with the original subgoal, with the context augmented by `hp : p`. Tactics are used to prove both subgoals.

Another option is to use the `note` tactic, which allows you to insert a fact into the context, without having to state the proposition it proves.

```

example (p q : Prop) : p ∧ q ↔ q ∧ p :=
begin
  apply iff.intro,
  { intro h,
    note hp := h.left,
    note hq := h.right,
    exact ⟨hq, hp⟩ },
  intro h,
  note hp := h.right,
  note hq := h.left,
  exact ⟨hp, hq⟩
end

```

In general, if `e` has type `t`, then `note h := e` adds a hypothesis `h : t` to the context, without giving you access to the contents of `e`. If, instead, you need the contents of `e`, use `pose x := e` instead. This adds `x : t := e` to the context as a `let` definition that can be unfolded when needed. Here is an example:

```
example :  $\exists x : \mathbb{N}, x + 3 = 8 :=$ 
begin
  pose x := 5,
  existsi x,
  reflexivity
end
```

Just as the `assert` tactic can be used to simulate the benefits of `have`, the `change` tactic can be used to simulate the benefits of `show`. In the following example, the tactic `change q` affirms that the goal at that point is `q`, and the tactic `change p` affirms that the goal is `p`.

```
example (p q : Prop) : p  $\wedge$  q  $\rightarrow$  q  $\wedge$  p :=
begin
  intro h,
  split,
  { change q,
    exact h.right },
  change p,
  exact h.left
end
```

The name of the `change` tactic is explained by the fact that it can be used to replace a goal by any definitionally equivalent statement.

```
example (a b :  $\mathbb{N}$ ) (h : a = b) : a + 0 = b + 0 :=
begin
  change a = b,
  assumption
end
```

The `change` statement will also work if the expression you give it has metavariables, in which case, it tries to unify the expression with the goal.

```
example (a b c :  $\mathbb{N}$ ) (h1 : a = b) (h2 : b = c) : a = c :=
begin
  transitivity,
  change _ = b, assumption,
  assumption
end
```

In this example, after the `transitivity` tactic is applied, there are two goals, `a = ?m_1` and `?m_1 = c`. After the `change`, the two goals have been specialized to `a = b` and `b = c`.

5.5 Rewriting and the Simplifier

The `rewrite` tactic (abbreviated `rw`) and the `simp` tactic were introduced in [Section 4.3](#). In this section, we discuss them in greater detail.

The **rewrite** tactic provide a basic mechanism for applying substitutions to goals and hypotheses, providing a convenient and efficient way of working with equality. The most basic form of the tactic is **rewrite** *t*, where *t* is a term which conclusion is an equality. In the following example, we use this basic form to rewrite the goal using a hypothesis.

```
variables (f : ℕ → ℕ) (k : ℕ)

example (h1 : f 0 = 0) (h2 : k = 0) : f k = 0 :=
begin
  rw h2, -- replace k with 0
  rw h1  -- replace f 0 with 0
end
```

In the example above, the first use of **rw** replaces *k* with 0 in the goal $f\ k = 0$. Then, the second one replaces $f\ 0$ with 0. The tactic automatically closes any goal of the form $t = t$.

Multiple rewrites can be combined using the notation **rw** [*t*₁, ..., *t*_{*n*}], which is just shorthand for **rewrite** *t*₁, ..., **rewrite** *t*_{*n*}. The previous example can be written as follows:

```
variables (f : ℕ → ℕ) (k : ℕ)

example (h1 : f 0 = 0) (h2 : k = 0) : f k = 0 :=
begin
  rw [h2, h1]
end
```

By default, **rw** uses an equation in the forward direction, matching the left-hand side with an expression, and replacing it with the right-hand side. The notation **-t** can be used to instruct the tactic to use the equality *t* in the reverse direction.

```
variables (f : ℕ → ℕ) (a b : ℕ)

example (h1 : a = b) (h2 : f a = 0) : f b = 0 :=
begin
  rw [-h1, h2]
end
```

In this example, the term **-h₁** instructs the rewriter to replace *b* with *a*.

Sometimes the left-hand side of an identity can match more than one subterm in the pattern, in which case the **rewrite** tactic chooses the first match it finds when traversing the term. If that is not the one you want, you can use additional arguments to specify the appropriate subterm.

```
example (a b c : ℕ) : a + b + c = a + c + b :=
begin
  rw [add_assoc, add_comm b, -add_assoc]
```

```

end

example (a b c : ℕ) : a + b + c = a + c + b :=
begin
  rw [add_assoc, add_assoc, add_comm b]
end

example (a b c : ℕ) : a + b + c = a + c + b :=
begin
  rw [add_assoc, add_assoc, add_comm _ b]
end

```

In the first example above, the first step rewrites $a + b + c$ to $a + (b + c)$. Then next applies commutativity to the term $b + c$; without specifying the argument, the tactic would instead rewrite $a + (b + c)$ to $(b + c) + a$. Finally, the last step applies associativity in the reverse direction rewriting $a + (c + b)$ to $a + c + b$. The next two examples instead apply associativity to move the parenthesis to the right on both sides, and then switch b and c . Notice that the last example specifies that the rewrite should take place on the right-hand side by specifying the second argument to `add_comm`.

By default, the `rewrite` tactic affects only the goal. The notation `rw t at h` applies the rewrite `t` at hypothesis `h`.

```

variables (f : ℕ → ℕ) (a : ℕ)

example (h : a + 0 = 0) : f a = f 0 :=
begin
  rw add_zero at h, rw h
end

```

The first step, `rw add_zero at h`, rewrites the hypothesis $a + 0 = 0$ to $a = 0$. Then the new hypothesis $a = 0$ is used to rewrite the goal to $f 0 = f 0$.

The `rewrite` tactic is not restricted to propositions. In the following example, we use `rw h at t` to rewrite the hypothesis $t : \text{tuple } \alpha \ n$ to $v : \text{tuple } \alpha \ 0$.

```

universe variable u

definition tuple (α : Type u) (n : ℕ) := { l : list α // list.length l = n }

variables {α : Type u} {n : ℕ}

example (h : n = 0) (t : tuple α n) : tuple α 0 :=
begin
  rw h at t,
  exact t
end

```

Note that the rewrite tactic can carry out generic calculations in any algebraic structure. The following examples involve an arbitrary ring and an arbitrary group, respectively.

```

universe variable uu

example {α : Type uu} [ring α] (a b c : α) : a * 0 + 0 * b + c * 0 + 0 * a = 0 :=
begin
  rw [mul_zero, mul_zero, zero_mul, zero_mul],
  repeat { rw add_zero }
end

example {α : Type uu} [group α] {a b : α} (h : a * b = 1) : a-1 = b :=
by rw [-(mul_one a-1), -h, inv_mul_cancel_left]

```

Using the type class mechanism described in [Chapter 10](#), Lean identifies both abstract and concrete instances of the relevant algebraic structures, and instantiates the relevant facts accordingly.

Whereas `rewrite` is designed as a surgical tool for manipulating a goal, the simplifier offers a powerful form of automation. A number of identities in Lean’s library have been tagged with the `[simp]` attribute, and the `simp` tactic uses them to iteratively rewrite subterms in an expression.

```

variables (x y z : ℕ) (p : ℕ → Prop)
premise   (h : p (x * y))

example : (x + 0) * (0 + y * 1 + z * 0) = x * y :=
by simp

include h
example : p ((x + 0) * (0 + y * 1 + z * 0)) :=
begin simp at h, assumption end

```

In the first example, the left-hand side of the equality in the goal is simplified using the usual identities involving 0 and 1, reducing the goal to $x * y = x * y$. At that point, `simp` applies reflexivity to finish it off. In the second example, `simp` reduces the goal to $p(x * y)$, at which point the assumption `h` finishes it off.

As with `rw`, you can use the keyword `at` to simplify a hypothesis:

```

example (h : p ((x + 0) * (0 + y * 1 + z * 0))) : p (x * y) :=
begin simp at h, assumption end

```

For operations that are commutative and associative, like addition on the natural numbers, the simplifier uses these two facts to rewrite an expression, as well as *left commutativity*. In the case of addition the latter is expressed as follows: $x + (y + z) = y + (x + z)$. It may seem that commutativity and left-commutativity are problematic, in that repeated application of either causes looping. But the simplifier detects identities that permute their arguments, and uses a technique known as *ordered rewriting*. This means that the system maintains an internal ordering of terms, and only applies the identity if doing so decreases the order. With the three identities mentioned above, this has the effect

that all the parentheses in an expression are associated to the right, and the expressions are ordered in a canonical (though somewhat arbitrary) way. Two expressions that are equivalent up to associativity and commutativity are then rewritten to the same canonical form.

```
variables (x y z w : ℕ) (p : ℕ → Prop)

example : x * y + z * w * x = x * w * z + y * x :=
by simp

example (h : p (x * y + z * w * x)) : p (x * w * z + y * x) :=
begin simp, simp at h, assumption end
```

As with the rewriter, the simplifier behaves appropriately in algebraic structures:

```
variables {α : Type} [comm_ring α]

example (x y z : α) : (x - x) * y + z = z :=
begin simp end

example (x y z w : α) : x * y + z * w * x = x * w * z + y * x :=
by simp
```

Also as with the `rewrite` tactic, you can pass additional arguments to `simp`. These can either be names of theorems or expressions. The `simp` tactic does not recognize the `-t` syntax, so to use an identity in the other direction you need to use `eq.symm` explicitly. In any case, the additional rules are added to the collection of identities that are used to simplify a term.

```
def f (m n : ℕ) : ℕ := m + n + m

theorem f.def (m n : ℕ) : f m n = m + n + m := rfl

example {m n : ℕ} (h : n = 1) (h' : 0 = m) : (f m n) * m = m :=
by simp [h, h'.symm, f.def]
```

If we add the attribute `[simp]` to the theorem `f.def`, we do not need to include it.

```
def f (m n : ℕ) : ℕ := m + n + m

@[simp]
theorem f.def (m n : ℕ) : f m n = m + n + m := rfl

example {m n : ℕ} (h : n = 1) (h' : 0 = m) : (f m n) * m = m :=
by simp [h, h'.symm]
```

Interacting with Lean

You are now familiar with the fundamentals of dependent type theory, both as a language for defining mathematical objects and a language for constructing proofs. The one thing you are missing is a mechanism for defining new data types. We will fill this gap in the next chapter, which introduces the notion of an *inductive data type*. But first, in this chapter, we take a break from the mechanics of type theory to explore some pragmatic aspects of interacting with Lean.

6.1 Displaying Information

There are a number of ways in which you can query Lean for information about its current state and the objects and theorems that are available in the current context. You have already seen two of the most common ones, `check` and `eval`. Remember that `check` is often used in conjunction with the `@` operator, which makes all of the arguments to a theorem or definition explicit. In addition, you can use the `print` command to get information about any identifier. If the identifier denotes a definition or theorem, Lean prints the type of the symbol, and its definition. If it is a constant or an axiom, Lean indicates that fact, and shows the type.

```
-- examples with equality
check eq
check @eq
check eq.symm
check @eq.symm

print eq.symm

-- examples with and
```

```

check and
check and.intro
check @and.intro

-- examples with addition
check add
check @add
eval add 3 2
print add

-- a user-defined function
definition foo {α : Type} (x : α) : α := x

check foo
check @foo
eval foo
eval (foo @nat.zero)
print foo

```

There are other useful `print` commands:

<code>print definition</code>	: display definition
<code>print inductive</code>	: display an inductive type and its constructors
<code>print notation</code>	: display all notation
<code>print notation <tokens></code>	: display notation using any of the tokens
<code>print axioms</code>	: display assumed axioms
<code>print options</code>	: display options set by user or emacs mode
<code>print prefix <namespace></code>	: display all declarations in the namespace
<code>print classes</code>	: display all classes
<code>print instances <class name></code>	: display all instances of the given class
<code>print fields <structure></code>	: display all "fields" of a structure

We will discuss inductive types, structures, classes, instances in the next four chapters. Here are examples of how these commands are used:

```

print notation
print notation + * -
print axioms
print options
print prefix nat
print prefix nat.le
print classes
print instances ring
print fields ring

```

The behavior of the generic `print` command is determined by its argument, so that the following pairs of commands all do the same thing.

```

print add
print definition add

print +

```

```

print notation +

print nat
print inductive nat

print group
print inductive group

```

Moreover, both `print group` and `print inductive group` recognize that a group is a structure (see [Chapter 9](#)), and so print the fields as well.

6.2 Setting Options

Lean maintains a number of internal variables that can be set by users to control its behavior. The syntax for doing so is as follows:

```
set_option <name> <value>
```

One very useful family of options controls the way Lean’s *pretty-printer* displays terms. The following options take an input of true or false:

```

pp.implicit  : display implicit arguments
pp.universes : display hidden universe parameters
pp.coercions  : show coercions
pp.notation   : display output using defined notations
pp.beta       : beta reduce terms before displaying them

```

In Lean, *coercions* can be inserted automatically to cast an element of one data type to another, for example, to cast an element of `nat` to an element of `int`. We will say more about them later in this chapter.

As an example, the following settings yield much longer output:

```

import data.nat
open nat

set_option pp.implicit true
set_option pp.universes true
set_option pp.notation false
set_option pp.numerals false

check 2 + 2 = 4
eval (λ x, x + 2) = (λ x, x + 3)
check (λ x, x + 1) 1

set_option pp.beta false
check (λ x, x + 1) 1

```

Pretty printing additional information is often very useful when you are debugging a proof, or trying to understand a cryptic error message. Too much information can be overwhelming, though, and Lean’s defaults are generally sufficient for ordinary interactions.

6.3 Using the Library

To use Lean effectively you will inevitably need to make use of definitions and theorems in the library. Recall that the `import` command at the beginning of a file imports previously compiled results from other files, and that importing is transitive; if you import `foo` and `foo` imports `bar`, then the definitions and theorems from `bar` are available to you as well. But the act of opening a namespace, which provides shorter names, does not carry over. In each file, you need to open the namespaces you wish to use.

In general, it is important for you to be familiar with the library and its contents, so you know what theorems, definitions, notations, and resources are available to you. Below we will see that Emacs’ Lean mode can also help you find things you need, but studying the contents of the library directly is often unavoidable. Lean’s standard library can be found online, on github:

<https://github.com/leanprover/lean/tree/master/library>

You can see the contents of the directories and files using github’s browser interface. If you have installed Lean on your own computer, you can find the library in the `lean` folder, and explore it with your file manager. Comment headers at the top of each file provide additional information.

Lean’s library developers follow general naming guidelines to make it easier to guess the name of a theorem you need, or to find it using tab completion in Emacs’ Lean mode, which is discussed in the next section. Identifiers are generally `snake_case`, which is to say, they are composed of words written in lower case separated by underscores. For the most part, we rely on descriptive names. Often the name of theorem simply describes the conclusion:

```
open nat

check succ_ne_zero
check @mul_zero
check @mul_one
check @sub_add_eq_add_sub
check @le_iff_lt_or_eq
```

If only a prefix of the description is enough to convey the meaning, the name may be made even shorter:

```
check @neg_neg
check pred_succ
```

Sometimes, to disambiguate the name of theorem or better convey the intended reference, it is necessary to describe some of the hypotheses. The word “of” is used to separate these hypotheses:

```
check @nat.lt_of_succ_le
check @lt_of_not_ge
check @lt_of_le_of_ne
check @add_lt_add_of_lt_of_le
```

Sometimes the word “left” or “right” is helpful to describe variants of a theorem.

```
check @add_le_add_left
check @add_le_add_right
```

We can also use the word “self” to indicate a repeated argument:

```
check mul_inv_self
check neg_add_self
```

Remember that identifiers in Lean can be organized into hierarchical namespaces. For example, the theorem named `lt_of_succ_le` in the namespace `nat` has full name `nat.lt_of_succ_le`, but the shorter name is made available by the command `open nat`. We will see in [Chapter 7](#) and [Chapter 9](#) that defining structures and inductive data types in Lean generates associated operations, and these are stored in a namespace with the same name as the type under definition. For example, the product type comes with the following opens:

```
check @prod.mk
check @prod.fst
check @prod.snd
check @prod.rec
```

The first is used to construct a pair, whereas the next two, `prod.fst` and `prod.snd`, project the two elements. The last, `prod.rec`, provides another mechanism for defining functions on a product in terms of a function on the two components. Names like `prod.rec` are *protected*, which means that one has to use the full name even when the `prod` namespace is open.

With the propositions as types correspondence, it then makes sense that operations that construct, destruct, and otherwise make use of propositions are also grouped into namespaces. We have already seen that this is the case with the logical connectives:

```
check @and.intro
check @and.elim
check @and.left
```

```

check @and.right
check @or.inl
check @or.inr
check @or.elim
check @exists.intro
check @exists.elim
check @eq.refl
check @eq.subst

```

But it also applies to predicates and relations that can be introduced and eliminated in similar ways.

```

check @le.refl

```

6.4 Using Lean with Emacs

This tutorial is designed to be read alongside Lean’s web-browser interface, which runs a Javascript-compiled version of Lean inside your web browser. But there is a much more powerful interface to Lean that runs as a special mode in the Emacs text editor. This section describes some of the advantages and features of the Emacs interface.

If you have never used Emacs before, you should spend some time experimenting with it. Emacs is an extremely powerful text editor, but it can also be overwhelming. There are a number of introductory tutorials on the web. See, for example:

- [A Guided Tour of Emacs](#)
- [Absolute Beginner’s Guide to Emacs](#)
- [Introduction to Emacs Course \(PDF\)](#)

You can get pretty far simply using the menus at the top of the screen for basic editing and file management. Those menus list keyboard-equivalents for the commands. Notation like “C-x”, short for “control x,” means “hold down the control key while typing x.” The notation “M-x”, short for “Meta x,” means “hold down the Alt key while typing x,” or, equivalently, “press the Esc key, followed by x.” For example, the “File” menu lists “C-c C-s” as a keyboard-equivalent for the “save file” command.

There are a number of benefits to using the native version of Lean instead of the web interface. Perhaps the most important is file management. The web interface imports the entire standard library internally, which is why some examples in this tutorial have to put examples in a namespace, `hide`, to avoid conflicting with objects already defined in the standard library. Moreover, the web interface only operates on one file at a time. Using the Emacs editor, you can create and edit Lean theory files anywhere on your file system, as with any editor or word processor. From these files, you can import pieces of the library at will, as well as your own theories, defined in separate files.

To use the Emacs with Lean, you simply need to create a file with the extension “lean” and edit it. For example, you can create a file by typing `emacs my_file.lean` in a terminal window, in the directory where you want to keep the file. Assuming everything has been installed correctly, Emacs will start up in Lean mode, already checking your file in the background.

You can then start typing, or copy any of the examples in this tutorial. (In the latter case, make sure you include the `import` and `open` commands that are sometimes hidden in the text.) Lean mode offers syntax highlighting, so commands, identifiers, and so on are helpfully color-coded. Any errors that Lean detects are subtly underlined in red, and the editor adds an annotation to the left margin at lines where errors occur. As you continue to type and eliminate errors, these annotations magically disappear.

If you put the cursor on a highlighted error, Emacs displays the error message in at the bottom of the frame. Alternatively, if you type `C-c ! 1` while in Lean mode, Emacs opens a new window with a list of compilation errors. Lean relies on an Emacs package, *Flycheck*, for this functionality, as evidenced by the letters “FlyC” that appear in the Emacs information line. Flycheck offers a number of commands that begin with `C-c !`. For example, `C-c ! n` moves the cursor to the next error, and `C-c ! p` moves the cursor to the previous error. You can get to a help menu that lists these key bindings by clicking on the “FlyC” tag.

It may be disconcerting to see a perfectly good proof suddenly “break” when you change a single character. Moreover, changes can introduce errors downstream. But the error messages vanish quickly when correctness is restored. Lean is quite fast. It uses multiple cores to process a file, and caches previous work to speed up compilation. As a result, changes you make are registered almost instantaneously.

It is often inconvenient to have to put the cursor on a highlighted identifier to see an error message or the outcome of a `print` or `check` command. The keystrokes `C-c C-n` toggle **Lean-Next-Error** mode, in which the next message (or all the messages that occur on the line that the cursor is on, if there are any) appears in a buffer named `*lean-info*`. You can position this window anywhere you want using Emacs commands to splitting windows and loading buffers. Pressing `C-c C-n` again toggles the mode off.

The Emacs Lean mode also maintains a continuous dialog with the background Lean server and uses it to present useful information to you. For example, if you put your cursor on any identifier — a theorem name, a defined symbol, or a variable — Emacs displays its type in the information line at the bottom.

The Lean mode supports tab completion. In a context where Lean expects an identifier (e.g. a theorem name or a defined symbol), if you start typing and then hit the tab key, a popup window suggests possible matches or near-matches for the expression you have typed. This helps you find the theorems you need without having to browse the library.

If you put your cursor on an identifier and hit `M-.`, Emacs will take you to the identifier’s definition, whether it is in the same file, in another file in the project (see [Section 6.5](#) below), or in the library. This works even in an autocompletion popup window: if you start typing

an identifier, press the tab key, choose a completion from the list of options, and press “M-.”, you are taken to the symbol’s definition. If you have Emacs 25 or later, you can then press “M-,” to go back to the original location.

In tactic mode, if you put your cursor on a tactic (or the keyword `begin` or `end`) and type `C-c C-g`, Emacs will show you the goal in the `*lean-info*` buffer. Here is another useful trick: if you see some notation in a Lean file and you want to know how to enter it from the keyboard, put the cursor on the symbol and type `C-c C-k`.

Recall that typing an underscore in an expression asks Lean to infer a suitable value for the expression and fill it in automatically. In cases where Lean is unable to determine a value for the argument, the underscore is highlighted, and the error message indicates the type of the “hole” that needs to be filled. This can be extremely useful when constructing proofs incrementally. One can start typing a “proof sketch,” using either `sorry` or an underscore for details you intend to fill in later. Assuming the proof is correct modulo these missing pieces of information, the error message at an unfilled underscore tells you the type of the term you need to construct, typically an assertion you need to justify.

The Emacs Lean-mode commands are summarized in the online documentation:

<https://github.com/leanprover/lean/blob/master/src/emacs/README.md>

If for some reason the Lean background process does not seem to be responding (for example, the information line no longer shows you type information), type “`C-c C-r`” or “`M-x lean-server-restart-process`”, or choose “restart lean process” from the Lean menu, and with luck that will set things right again.

In Lean, the `exit` command halts processing of a file abruptly. Inserting an `exit` therefore prevents Lean from checking the file beyond that point.

6.5 Imports, Object Files, and Projects

At this point, it will be helpful to convey more information about the inner workings of Lean. A `.lean` file (read “dot Lean”) consists of instructions that tell Lean how to construct formal terms in dependent type theory. Processing this file is a matter of filling in missing or implicit information, constructing the relevant terms, and sending them to the type checker to confirm that they are well-formed and have the specified types. This is analogous to the compilation process for a programming language: the `.lean` file contains the source code that is then compiled down to machine representations of the desired formal objects. Lean caches the output of the compilation process in files with the extension `.olean`, for “object Lean”.

Assuming the directory that contains Lean is in your system path, you can run lean on a file `foo.lean` from a system command line by typing `lean foo.lean`. If `foo` imports other files, by default Lean looks for these in the standard library, which it finds relative to the directory from which it was invoked, and the current directory. You can change the default

or add additional directories by specifying the search paths in the `LEAN_PATH` environment variable. You can specify subdirectories using periods in the module name: for example, `import foo.bar.baz` looks for the file “foo/bar/baz.lean” relative to any of the locations listed in the search path. A leading period, as in `import .foo.bar`, indicates that the .lean file in question is specified relative to the current directory. Two leading periods, as in `import ..foo.bar`, indicates that the address is relative to the parent directory, and so on.

When processing `foo`, Lean uses any `.olean` files it can find for the imports as long as they are up to date with the source file. Otherwise, it recursively compiles the dependencies when necessary. Of course, it is more efficient if it can use the `.olean` files. The command `lean --make foo.lean` not only compiles `foo` but saves the results in `foo.olean`. You can compile more than one file at once, for example, with a command like `lean --make foo.lean bar.lean baz.lean`.

One often wants to create complex projects and arrange the source files in nested directories. If `bar` is a path to a directory, `lean --make bar` compiles all the files in that directory, descending recursively into subdirectories. The specified directory is added to the `LEAN_PATH`, so files in a project can specify imports in absolute terms from the base directory. The command `lean --make` is equivalent to `lean --make .`, which is to say, it compiles all the files within and below the current directory.

Similar considerations hold when running Lean from Emacs. Lean mode starts a background Lean process that not only checks files and provides the messages you see in Fly-check, but also responds to other queries for information. Lean processes the current file in the background, giving highest priority to theorems that are visible in the buffer, and updating compilation tasks as you continue to type. Otherwise, the process is the same as command-line compilation: Lean uses `.olean` files when they are available and up to date, and compiles imports recursively when necessary.

Because files can be opened in Emacs anywhere and at any time, it takes a bit more effort to help Lean mode identify the root directory when you open a file in a project. You can do this simply by creating a file named `.project` in that root directory. The contents of the file are ignored; the file serves only as a marker. (On variants of Unix, you can create an empty file by typing `touch .project` at a shell prompt.) When you open any `.lean` file in Emacs, Lean mode traverses the parent directories, and if it finds a `.project` file along the way, it takes that to be the root of the project.

In fact, Lean mode starts one Lean server for each project being edited. As a result, if you are editing files `a.lean` and `b.lean` in the same project, and `b.lean` depends on `a.lean`, then whenever you make a change in `b.lean` the result is immediately visible to `a.lean`.

Changes are not immediately available across projects, however. Suppose you have a project `foo` that depends on another project `bar` and you are editing both. If you want the changes in `bar` to be available to `foo`, simply save all the files in `bar`, switch to any file in `foo`, and either type `C-c C-r` or choose the corresponding Lean menu option to restart

the Lean server. Upon restarting, the server for `foo` will detect and use the new version of `bar`.

6.6 Notation and Abbreviations

Lean’s parser is an instance of a Pratt parser, a non-backtracking parser that is fast and flexible. You can read about Pratt parsers in a number of places online, such as here:

http://en.wikipedia.org/wiki/Pratt_parser <http://eli.thegreenplace.net/2010/01/02/top-down-operator-precedence-parsing>

Identifiers can include any alphanumeric characters, including Greek characters (other than Π , Σ , and λ , which, as we have seen, have a special meaning in the dependent type theory). They can also include subscripts, which can be entered by typing `_` followed by the desired subscripted character.

Lean’s parser is moreover extensible, which is to say, we can define new notation.

```
notation `[ a `**` b `]` := a * b + 1

definition mul_square (a b : ℕ) := a * a * b * b

infix `<*>`:50 := mul_square

eval [2 ** 3]
eval 2 <*> 3
```

In this example, the `notation` command defines a complex binary notation for multiplying and adding one. The `infix` command declares a new infix operator, with precedence 50, which associates to the left. (More precisely, the token is given left-binding power 50.) The command `infixr` defines notation which associates to the right, instead.

If you declare these notations in a namespace, the notation is only available when the namespace is open. You can declare temporary notation using the keyword `local`, in which case the notation is available in the current file, and moreover, within the scope of the current `namespace` or `section`, if you are in one.

```
local notation `[ a `**` b `]` := a * b + 1
local infix `<*>`:50 := λ a b : ℕ, a * a * b * b
```

The file `reserved_notation.lean` in the `init` folder of the library declares the left-binding powers of a number of common symbols that are used in the library.

https://github.com/leanprover/lean/blob/master/library/init/reserved_notation.lean

You are welcome to overload these symbols for your own use, but you cannot change their right-binding power.

Remember that you can direct the pretty-printer to suppress notation with the command `set_option pp.notation false`. You can also declare notation to be used for input purposes only with the `[parsing_only]` attribute:

```
notation [parsing_only] `[ a `**` b ` ]` := a * b + 1

variables a b : ℕ
check [a ** b]
```

The output of the `check` command displays the expression as `a * b + 1`. Lean also provides mechanisms for iterated notation, such as `[a, b, c, d, e]` to denote a list with the indicated elements. See the discussion of `list` in the next chapter for an example.

Inductive Types

We have seen that Lean’s formal foundation includes basic types, `Prop`, `Type.{1}`, `Type.{2}`, ..., and allows for the formation of dependent function types, $\Pi x : \alpha. \beta$. In the examples, we have also made use of additional types like `bool`, `nat`, and `int`, and type constructors, like `list`, and product, \times . In fact, in Lean’s library, every concrete type other than the universes and every type constructor other than `Pi` is an instance of a general family of type constructions known as *inductive types*. It is remarkable that it is possible to construct a substantial edifice of mathematics based on nothing more than the type universes, `Pi` types, and inductive types; everything else follows from those.

Intuitively, an inductive type is built up from a specified list of constructors. In Lean, the syntax for specifying such a type is as follows:

```
inductive foo : Type
| constructor1 : ... → foo
| constructor2 : ... → foo
...
| constructorn : ... → foo
```

The intuition is that each constructor specifies a way of building new objects of `foo`, possibly from previously constructed values. The type `foo` consists of nothing more than the objects that are constructed in this way. The first character `|` in an inductive declaration is optional. We can also separate constructors using a comma instead of `|`.

We will see below that the arguments to the constructors can include objects of type `foo`, subject to a certain “positivity” constraint, which guarantees that elements of `foo` are built from the bottom up. Roughly speaking, each `...` can be any `Pi` type constructed from `foo` and previously defined types, in which `foo` appears, if at all, only as the “target” of the `Pi` type. For more details, see [2].

We will provide a number of examples of inductive types. We will also consider slight generalizations of the scheme above, to mutually defined inductive types, and so-called *inductive families*.

As with the logical connectives, every inductive type comes with introduction rules, which show how to construct an element of the type, and elimination rules, which show how to “use” an element of the type in another construction. The analogy to the logical connectives should not come as a surprise; as we will see below, they, too, are examples of inductive type constructions. You have already seen the introduction rules for an inductive type: they are just the constructors that are specified in the definition of the type. The elimination rules provide for a principle of recursion on the type, which includes, as a special case, a principle of induction as well.

In the next chapter, we will describe Lean’s function definition package, which provides even more convenient ways to define functions on inductive types and carry out inductive proofs. But because the notion of an inductive type is so fundamental, we feel it is important to start with a low-level, hands-on understanding. We will start with some basic examples of inductive types, and work our way up to more elaborate and complex examples.

7.1 Enumerated Types

The simplest kind of inductive type is simply a type with a finite, enumerated list of elements.

```
inductive weekday : Type
| sunday : weekday
| monday : weekday
| tuesday : weekday
| wednesday : weekday
| thursday : weekday
| friday : weekday
| saturday : weekday
```

The `inductive` command creates a new type, `weekday`. The constructors all live in the `weekday` namespace.

```
check weekday.sunday
check weekday.monday

open weekday

check sunday
check monday
```

Think of the `sunday`, `monday`, ... as being distinct elements of `weekday`, with no other distinguishing properties. The elimination principle, `weekday.rec`, is defined at the same

time as the type `weekday` and its constructors. It is also known as a *recursor*, and it is what makes the type “inductive”: it allows us to define a function on `weekday` by assigning values corresponding to each constructor. The intuition is that an inductive type is exhaustively generated by the constructors, and has no elements beyond those they construct.

We will use a slight (automatically generated) variant, `weekday.rec_on`, which takes its arguments in a more convenient order. Note that the shorter versions of names like `weekday.rec` and `weekday.rec_on` are not made available by default when we open the `weekday` namespace, to avoid clashes. If we import `nat`, we can use `rec_on` to define a function from `weekday` to the natural numbers:

```
def number_of_day (d : weekday) : ℕ :=
  weekday.rec_on d 1 2 3 4 5 6 7

eval number_of_day weekday.sunday
eval number_of_day weekday.monday
eval number_of_day weekday.tuesday
```

The first (explicit) argument to `rec_on` is the element being “analyzed.” The next seven arguments are the values corresponding to the seven constructors. Note that `number_of_day weekday.sunday` evaluates to 1: the computation rule for `rec_on` recognizes that `sunday` is a constructor, and returns the appropriate argument.

Below we will encounter a more restricted variant of `rec_on`, namely, `cases_on`. When it comes to enumerated types, `rec_on` and `cases_on` are the same. You may prefer to use the label `cases_on`, because it emphasizes that the definition is really a definition by cases.

```
def number_of_day (d : weekday) : ℕ :=
  weekday.cases_on d 1 2 3 4 5 6 7
```

It is often useful to group definitions and theorems related to a structure in a namespace with the same name. For example, we can put the `number_of_day` function in the `weekday` namespace. We are then allowed to use the shorter name when we open the namespace.

The names `rec_on`, `cases_on`, `induction_on`, and so on are generated automatically. As noted above, they are *protected* to avoid name clashes. In other words, they are not provided by default when the namespace is opened. However, you can explicitly declare abbreviations for them using the `renaming` option when you open a namespace.

```
namespace weekday
  @[reducible]
  private def cases_on := @weekday.cases_on

  def number_of_day (d : weekday) : nat :=
    cases_on d 1 2 3 4 5 6 7
end weekday
```

```

eval weekday.number_of_day weekday.sunday

open weekday (renaming cases_on → cases_on)

eval number_of_day sunday
check cases_on

```

We can define functions from `weekday` to `weekday`:

```

namespace weekday
def next (d : weekday) : weekday :=
  weekday.cases_on d monday tuesday wednesday thursday friday saturday sunday

def previous (d : weekday) : weekday :=
  weekday.cases_on d saturday sunday monday tuesday wednesday thursday friday

eval next (next tuesday)
eval next (previous tuesday)

example : next (previous tuesday) = tuesday := rfl
end weekday

```

How can we prove the general theorem that `next (previous d) = d` for any `weekday` `d`? The induction principle parallels the recursion principle: we simply have to provide a proof of the claim for each constructor:

```

theorem next_previous (d: weekday) : next (previous d) = d :=
  weekday.induction_on d
    (show next (previous sunday) = sunday, from rfl)
    (show next (previous monday) = monday, from rfl)
    (show next (previous tuesday) = tuesday, from rfl)
    (show next (previous wednesday) = wednesday, from rfl)
    (show next (previous thursday) = thursday, from rfl)
    (show next (previous friday) = friday, from rfl)
    (show next (previous saturday) = saturday, from rfl)

```

In fact, `induction_on` is just a special case of `rec_on` where the target type is an element of `Prop`. In other words, under the propositions-as-types correspondence, the principle of induction is a type of definition by recursion, where what is being “defined” is a proof instead of a piece of data. We could equally well have used `cases_on`:

```

theorem next_previous (d: weekday) : next (previous d) = d :=
  weekday.cases_on d
    (show next (previous sunday) = sunday, from rfl)
    (show next (previous monday) = monday, from rfl)
    (show next (previous tuesday) = tuesday, from rfl)
    (show next (previous wednesday) = wednesday, from rfl)
    (show next (previous thursday) = thursday, from rfl)
    (show next (previous friday) = friday, from rfl)
    (show next (previous saturday) = saturday, from rfl)

```

While the `show` commands make the proof clearer and more readable, they are not necessary:

```
theorem next_previous (d: weekday) : next (previous d) = d :=
  weekday.cases_on d rfl rfl rfl rfl rfl rfl rfl
```

Some fundamental data types in the Lean library are instances of enumerated types.

```
inductive empty : Type

inductive unit : Type
| star : unit

inductive bool : Type
| ff : bool
| tt : bool
```

(To run these examples, we put them in a namespace called `hide`, so that a name like `bool` does not conflict with the `bool` in the standard library. This is necessary because these types are part of the Lean “prelude” that is automatically imported with the system is started.)

The type `empty` is an inductive data type with no constructors. The type `unit` has a single element, `star`, and the type `bool` represents the familiar boolean values. As an exercise, you should think about what the introduction and elimination rules for these types do. As a further exercise, we suggest defining boolean operations `band`, `bor`, `bnot` on the boolean, and verifying common identities. Note that defining a binary operation like `band` will require nested cases splits:

```
def band (b1 b2 : bool) : bool :=
  bool.cases_on b1
    ff
    (bool.cases_on b2 ff tt)
```

Similarly, most identities can be proved by introducing suitable case splits, and then using `rfl`.

7.2 Constructors with Arguments

Enumerated types are a very special case of inductive types, in which the constructors take no arguments at all. In general, a “construction” can depend on data, which is then represented in the constructed argument. Consider the definitions of the product type and sum type in the library:

```

universe variables u v

inductive prod (α : Type u) (β : Type v)
| mk : α → β → prod

inductive sum (α : Type u) (β : Type v)
| inl {} : α → sum
| inr {} : β → sum

```

Notice that we do not include the types α and β in the target of the constructors. For the moment, ignore the annotation `{}` after the constructors `inl` and `inr`; we will explain that below. In the meanwhile, think about what is going on in these examples. The product type has one constructor, `prod.mk`, which takes two arguments. To define a function on `prod α β`, we can assume the input is of the form `prod.mk a b`, and we have to specify the output, in terms of `a` and `b`. We can use this to define the two projections for `prod`; remember that the standard library defines notation $\alpha \times \beta$ for `prod α β` and `(a, b)` for `prod.mk a b`.

```

def fst {α : Type u} {β : Type v} (p : α × β) : α :=
prod.rec_on p (λ a b, a)

def snd {α : Type u} {β : Type v} (p : α × β) : β :=
prod.rec_on p (λ a b, b)

```

The function `fst` takes a pair, `p`. Applying the recursor `prod.rec_on p` (`fun a b, a`) interprets `p` as a pair, `prod.mk a b`, and then uses the second argument to determine what to do with `a` and `b`. Remember that you can enter the symbol for a product by typing `\times`. Recall also from [Section 2.8](#) that to give these definitions the greatest generality possible, we allow the types α and β to belong to any universe.

Here is another example:

```

def prod_example (p : bool × ℕ) : ℕ :=
prod.rec_on p (λ b n, cond b (2 * n) (2 * n + 1))

eval prod_example (tt, 3)
eval prod_example (ff, 3)

```

The `cond` function is a boolean conditional: `cond b t1 t2` return `t1` if `b` is true, and `t2` otherwise. (It has the same effect as `bool.rec_on b t2 t1`.) The function `prod_example` takes a pair consisting of a boolean, `b`, and a number, `n`, and returns either `2 * n` or `2 * n + 1` according to whether `b` is true or false.

In contrast, the `sum` type has *two* constructors, `inl` and `inr` (for “insert left” and “insert right”), each of which takes *one* (explicit) argument. To define a function on `sum α β`, we have to handle two cases: either the input is of the form `inl a`, in which case we

have to specify an output value in terms of `a`, or the input is of the form `inr b`, in which case we have to specify an output value in terms of `b`.

```
def sum_example (s :  $\mathbb{N} \oplus \mathbb{N}$ ) :  $\mathbb{N}$  :=
  sum.cases_on s ( $\lambda n, 2 * n$ ) ( $\lambda n, 2 * n + 1$ )

eval sum_example (sum.inl 3)
eval sum_example (sum.inr 3)
```

This example is similar to the previous one, but now an input to `sum_example` is implicitly either of the form `inl n` or `inr n`. In the first case, the function returns `2 * n`, and the second case, it returns `2 * n + 1`. You can enter the symbol for the sum by typing `\oplus`.

In the section after next we will see what happens when the constructor of an inductive type takes arguments from the inductive type itself. What characterizes the examples we consider in this section is that this is not the case: each constructor relies only on previously specified types.

Notice that a type with multiple constructors is disjunctive: an element of `sum α β` is either of the form `inl a` or of the form `inr b`. A constructor with multiple arguments introduces conjunctive information: from an element `prod.mk a b` of `prod α β` we can extract `a` and `b`. An arbitrary inductive type can include both features, by having any number of constructors, each of which takes any number of arguments.

A type, like `prod`, with only one constructor is purely conjunctive: the constructor simply packs the list of arguments into a single piece of data, essentially a tuple where the type of subsequent arguments can depend on the type of the initial argument. We can also think of such a type as a “record” or a “structure”. In Lean, these two words are synonymous, and provide alternative syntax for inductive types with a single constructor.

```
structure prod ( $\alpha$   $\beta$  : Type) :=
  mk :: (fst :  $\alpha$ ) (snd :  $\beta$ )
```

The `structure` command simultaneously introduces the inductive type, `prod`, its constructor, `mk`, the usual eliminators (`rec`, `rec_on`), as well as the projections, `fst` and `snd`, as defined above.

If you do not name the constructor, Lean uses `mk` as a default. For example, the following defines a record to store a color as a triple of RGB values:

```
record color := (red : nat) (green : nat) (blue : nat)
def yellow := color.mk 255 255 0
eval color.red yellow
```

The definition of `yellow` forms the record with the three values shown, and the projection `color.red` returns the red component. The `structure` command is especially useful for defining algebraic structures, and Lean provides substantial infrastructure to support working with them. Here, for example, is the definition of a semigroup:

```

universe variable u

structure Semigroup :=
  (carrier : Type u)
  (mul : carrier → carrier → carrier)
  (mul_assoc : ∀ a b c, mul (mul a b) c = mul a (mul b c))

```

We will see more examples in [Chapter 9](#).

Notice that the product type depends on parameters $\alpha \ \beta : \text{Type}$ which are arguments to the constructors as well as `prod`. Lean detects when these arguments can be inferred from later arguments to a constructor, and makes them implicit in that case. Sometimes an argument can only be inferred from the return type, which means that it could not be inferred by parsing the expression from bottom up, but may be inferrable from context. In that case, Lean does not make the argument implicit by default, but will do so if we add the annotation `{}` after the constructor. We used that option, for example, in the definition of `sum`:

```

inductive sum (α : Type u) (β : Type v)
| inl {} : α → sum
| inr {} : β → sum

```

as a result, the argument α to `inl` and the argument β to `inr` are left implicit.

We have already discussed sigma types, also known as the dependent product:

```

inductive sigma {α : Type u} (β : α → Type v)
| dpair : Π a : α, β a → sigma

```

Two more examples of inductive types in the library are the following:

```

inductive option (α : Type u)
| none {} : option
| some   : α → option

inductive inhabited (α : Type u)
| mk : α → inhabited

```

In the semantics of dependent type theory, there is no built-in notion of a partial function. Every element of a function type $\alpha \rightarrow \beta$ or a Pi type $\Pi x : \alpha, \beta$ is assumed to have a value at every input. The `option` type provides a way of representing partial functions. An element of `option` β is either `none` or of the form `some b`, for some value $b : \beta$. Thus we can think of an element f of the type $\alpha \rightarrow \text{option } \beta$ as being a partial function from α to β : for every $a : \alpha$, $f \ a$ either returns `none`, indicating the $f \ a$ is “undefined”, or `some b`.

An element of `inhabited` α is simply a witness to the fact that there is an element of α . Later, we will see that `inhabited` is an example of a *type class* in Lean: Lean can be instructed that suitable base types are inhabited, and can automatically infer that other constructed types are inhabited on that basis.

As exercises, we encourage you to develop a notion of composition for partial functions from α to β and β to γ , and show that it behaves as expected. We also encourage you to show that `bool` and `nat` are inhabited, that the product of two inhabited types is inhabited, and that the type of functions to an inhabited type is inhabited.

7.3 Inductively Defined Propositions

Inductively defined types can live in any type universe, including the bottom-most one, `Prop`. In fact, this is exactly how the logical connectives are defined.

```
inductive false : Prop

inductive true : Prop
| intro : true

inductive and (a b : Prop) : Prop
| intro : a → b → and

inductive or (a b : Prop) : Prop
| intro_left : a → or
| intro_right : b → or
```

You should think about how these give rise to the introduction and elimination rules that you have already seen. There are rules that govern what the eliminator of an inductive type can eliminate *to*, that is, what kinds of types can be the target of a recursor. Roughly speaking, what characterizes inductive types in `Prop` is that one can only eliminate to other types in `Prop`. This is consistent with the understanding that if $p : \text{Prop}$, an element $hp : p$ carries no data. There is a small exception to this rule, however, which we will discuss below, in the section on inductive families.

Even the existential quantifier is inductively defined:

```
inductive Exists {α : Type u} (p : α → Prop) : Prop
| intro : ∀ (a : α), p a → Exists

def exists.intro := @Exists.intro
```

Keep in mind that the notation $\exists x : \alpha, p$ is syntactic sugar for `Exists` $(\lambda x : \alpha, p)$.

The definitions of `false`, `true`, `and`, and `or` are perfectly analogous to the definitions of `empty`, `unit`, `prod`, and `sum`. The difference is that the first group yields elements of `Prop`, and the second yields elements of `Type i` for i greater than 0. In a similar way, $\exists x : \alpha, p$ is a `Prop`-valued variant of $\Sigma x : \alpha, p$.

This is a good place to mention another inductive type, denoted $\{x : \alpha \mid p\}$, which is sort of a hybrid between $\exists x : \alpha, P$ and $\Sigma x : \alpha, P$.

```
inductive subtype {α : Type u} (p : α → Prop)
| tag : Π x : α, p x → subtype
```

The notation $\{x : \alpha \mid p\}$ is syntactic sugar for `subtype (λ x : α, p)`. It is modeled after subset notation in set theory: the idea is that $\{x : \alpha \mid p\}$ denotes the collection of elements of α that have property p .

7.4 Defining the Natural Numbers

The inductively defined types we have seen so far are “flat”: constructors wrap data and insert it into a type, and the corresponding recursor unpacks the data and acts on it. Things get much more interesting when the constructors act on elements of the very type being defined. A canonical example is the type `nat` of natural numbers:

```
inductive nat : Type
| zero : nat
| succ : nat → nat
```

There are two constructors. We start with `zero : nat`; it takes no arguments, so we have it from the start. In contrast, the constructor `succ` can only be applied to a previously constructed `nat`. Applying it to `zero` yields `succ zero : nat`. Applying it again yields `succ (succ zero) : nat`, and so on. Intuitively, `nat` is the “smallest” type with these constructors, meaning that it is exhaustively (and freely) generated by starting with `zero` and applying `succ` repeatedly.

As before, the recursor for `nat` is designed to define a dependent function `f` from `nat` to any domain, that is, an element `f` of $\prod n : \text{nat}, C\ n$ for some $C : \text{nat} \rightarrow \text{Type}$. It has to handle two cases: the case where the input is `zero`, and the case where the input is of the form `succ n` for some `n : nat`. In the first case, we simply specify a target value with the appropriate type, as before. In the second case, however, the recursor can assume that a value of `f` at `n` has already been computed. As a result, the next argument to the recursor specifies a value for `f (succ n)` in terms of `n` and `f n`. If we check the type of the recursor,

```
check @nat.rec_on
```

we find the following:

```
Π {C : nat → Type} (n : nat),
  C nat.zero → (Π (a : nat), C a → C (nat.succ a)) → C n
```

The implicit argument, C , is the codomain of the function being defined. In type theory it is common to say C is the **motive** for the elimination/recursion. The next argument, $n : \text{nat}$, is the input to the function. It is also known as the **major premise**. Finally, the two arguments after specify how to compute the zero and successor cases, as described above. They are also known as the **minor premises**.

Consider, for example, the addition function $\text{add } m \ n$ on the natural numbers. Fixing m , we can define addition by recursion on n . In the base case, we set $\text{add } m \ \text{zero}$ to m . In the successor step, assuming the value $\text{add } m \ n$ is already determined, we define $\text{add } m \ (\text{succ } n)$ to be $\text{succ } (\text{add } m \ n)$.

```
namespace nat

def add (m n : nat) : nat :=
nat.rec_on n m (λ n add_m_n, succ add_m_n)

-- try it out
eval add (succ zero) (succ (succ zero))

end nat
```

It is useful to put such definitions into a namespace, `nat`. We can then go on to define familiar notation in that namespace. The two defining equations for addition now hold definitionally:

```
instance : has_zero nat := has_zero.mk zero
instance : has_add nat := has_add.mk add

theorem add_zero (m : nat) : m + 0 = m := rfl
theorem add_succ (m n : nat) : m + succ n = succ (m + n) := rfl
```

We will explain how the `instance` command works in [Chapter 10](#). In the examples below, we will henceforth use Lean’s version of the natural numbers.

Proving a fact like $0 + m = m$, however, requires a proof by induction. As observed above, the induction principle is just a special case of the recursion principle, when the codomain $C \ n$ is an element of `Prop`. It represents the familiar pattern of an inductive proof: to prove $\forall n, C \ n$, first prove $C \ 0$, and then, for arbitrary n , assume $ih : C \ n$ and prove $C \ (\text{succ } n)$.

```
theorem zero_add (n : ℕ) : 0 + n = n :=
nat.induction_on n
  (show 0 + 0 = 0, from rfl)
  (take n,
    assume ih : 0 + n = n,
    show 0 + succ n = succ n, from
      calc
        0 + succ n = succ (0 + n) : rfl
        ... = succ n : by rewrite ih)
```

In the example above, we encourage you to replace `induction_on` with `rec_on` and observe that the theorem is still accepted by Lean. As we have seen above, `induction_on` is just a special case of `rec_on`.

For another example, let us prove the associativity of addition, $\forall m\ n\ k, m + n + k = m + (n + k)$. (The notation $+$, as we have defined it, associates to the left, so $m + n + k$ is really $(m + n) + k$.) The hardest part is figuring out which variable to do the induction on. Since addition is defined by recursion on the second argument, k is a good guess, and once we make that choice the proof almost writes itself:

```

theorem add_assoc (m n k : ℕ) : m + n + k = m + (n + k) :=
nat.induction_on k
  (show m + n + 0 = m + (n + 0), from rfl)
  (take k,
    assume ih : m + n + k = m + (n + k),
    show m + n + succ k = m + (n + succ k), from
      calc
        m + n + succ k = succ (m + n + k) : rfl
        ... = succ (m + (n + k)) : by rewrite ih
        ... = m + succ (n + k) : rfl
        ... = m + (n + succ k) : rfl)

```

For another example, suppose we try to prove the commutativity of addition. Choosing induction on the second argument, we might begin as follows:

```

theorem add_comm (m n : nat) : m + n = n + m :=
nat.induction_on n
  (show m + 0 = 0 + m, by rewrite nat.zero_add)
  (take n,
    assume ih : m + n = n + m,
    calc
      m + succ n = succ (m + n) : rfl
      ... = succ (n + m) : by rewrite ih
      ... = succ n + m : sorry)

```

At this point, we see that we need another supporting fact, namely, that `succ (n + m) = succ n + m`. We can prove this by induction on m :

```

theorem succ_add (m n : nat) : succ m + n = succ (m + n) :=
nat.induction_on n
  (show succ m + 0 = succ (m + 0), from rfl)
  (take n,
    assume ih : succ m + n = succ (m + n),
    show succ m + succ n = succ (m + succ n), from
      calc
        succ m + succ n = succ (succ m + n) : rfl
        ... = succ (succ (m + n)) : by rewrite ih
        ... = succ (m + succ n) : rfl)

```

We can then replace the `sorry` in the previous proof with `succ_add`.

As an exercise, try defining other operations on the natural numbers, such as multiplication, the predecessor function (with `pred 0 = 0`), truncated subtraction (with `n - m = 0` when `m` is greater than or equal to `n`), and exponentiation. Then try proving some of their basic properties, building on the theorems we have already proved.

7.5 Tactics

Given the fundamental importance of inductive types in Lean, it should not be surprising that there are a number of tactics described to work with them effectively. We describe some of them here.

The `cases` tactic works on elements of an inductively defined type, and does what the name suggests: it decomposes the element according to each of the possible constructors. In its most basic form, it is applied to an element `x` in the local context. It then reduces the goal to cases in which `x` is replaced by each of the constructions.

```
open nat
variable p : ℕ → Prop

example (hz : p 0) (hs : ∀ n, p (succ n)) : ∀ n, p n :=
begin
  intro n,
  cases n,
  { exact hz }, -- goal is p 0
  apply hs     -- goal is a : ℕ ⊢ p (succ a)
end
```

There are extra bells and whistles. For one thing, `cases` allows you to choose the names for the arguments to the constructors using a `with` clause. In the next example, for example, we choose the name `m` for the argument to `succ`, so that the second case refers to `succ m`. More importantly, the `cases` tactic will detect any items in the local context that depend on the target variable. It reverts these elements, does the split, and reintroduces them. In the example below, notice that the hypothesis `h : n ≠ 0` becomes `h : 0 ≠ 0` in the first branch, and `h : succ m ≠ 0` in the second.

```
example (n : ℕ) (h : n ≠ 0) : succ (pred n) = n :=
begin
  cases n with m,
  -- first goal: h : 0 ≠ 0 ⊢ succ (pred 0) = 0
  { apply (absurd rfl h) },
  -- second goal: h : succ m ≠ 0 ⊢ succ (pred (succ a)) = succ a
  reflexivity
end
```

Notice that `cases` can be used to produce data as well as prove propositions.

```
def f (n : ℕ) : ℕ :=
begin
  cases n, exact 3, exact 7
end

example : f 0 = 3 := rfl
example : f 5 = 7 := rfl
```

Once again, cases will revert and dependencies in the context, split, and then reintroduce them.

```
universe variable u

definition tuple (α : Type u) (n : ℕ) := { l : list α // list.length l = n }

variables {α : Type u} {n : ℕ}

def f {n : ℕ} (t : tuple α n) : ℕ :=
begin
  cases n, exact 3, exact 7
end

def my_tuple : tuple ℕ 3 := ⟨[0, 1, 2], rfl⟩

example : f my_tuple = 7 := rfl
```

If there are multiple constructors with arguments, you can provide `cases` with a list of all the names, arranged sequentially:

```
inductive foo : Type
| bar1 : ℕ → ℕ → foo
| bar2 : ℕ → ℕ → ℕ → foo

def silly (x : foo) : ℕ :=
begin
  cases x with a b c d e,
  exact b,    -- a, b, c are in the context
  exact e     -- d, e are in the context
end
```

You can also use `cases` with an arbitrary expression. Assuming that expression occurs in the goal, the cases tactic will generalize over the expression, introduce the resulting universally quantified variable, and case on that.

```
open nat
variable p : ℕ → Prop

example (hz : p 0) (hs : ∀ n, p (succ n)) (m k : ℕ) : p (m + 3 * k) :=
begin
  cases (m + 3 * k),
  { exact hz }, -- goal is p 0
```

```

    apply hs          -- goal is a : ℕ ⊢ p (succ a)
end

```

Think of this as saying “split on cases as to whether $m + 3 * k$ is zero or the successor of some number.” The result is functionally equivalent to the following:

```

example (hz : p 0) (hs : ∀ n, p (succ n)) (m k : ℕ) : p (m + 3 * k) :=
begin
  generalize (m + 3 * k) n,
  intro n,
  cases n,
  { exact hz }, -- goal is p 0
  apply hs      -- goal is a : ℕ ⊢ p (succ a)
end

```

Notice that the expression $m + 3 * k$ is erased by `generalize`; all that matters is whether it is of the form 0 or `succ a`. This form of `cases` will *not* revert any hypotheses that also mention the expression in equation (in this case, $m + 3 * k$). If such a term appears in a hypothesis and you want to generalize over that as well, you need to `revert` it explicitly.

If the expression you case on does not appear in the goal, the `cases` tactic uses `assert` to put the type of the expression into the context. Here is an example:

```

example (p : Prop) (m n : ℕ) (h1 : m < n → p) (h2 : m ≥ n → p) : p :=
begin
  cases lt_or_ge m n with hlt hge,
  { exact h1 hlt },
  exact h2 hge
end

```

The theorem `lt_or_ge m n` says $m < n \vee m \geq n$, and it is natural to think of the proof above as splitting on these two cases. In the first branch, we have the hypothesis $h_1 : m < n$, and in the second we have the hypothesis $h_2 : m \geq n$. The proof above is functionally equivalent to the following:

```

example (p : Prop) (m n : ℕ) (h1 : m < n → p) (h2 : m ≥ n → p) : p :=
begin
  assert h : m < n ∨ m ≥ n,
  { exact lt_or_ge m n },
  cases h with hlt hge,
  { exact h1 hlt },
  exact h2 hge
end

```

After the first two lines, we have $h : m < n \vee m \geq n$ as a hypothesis, and we simply do cases on that.

Here is another example, where we use the decidability of equality on the natural numbers to split on the cases $m = n$ and $m \neq n$.

```

check nat.sub_self

example (m n : ℕ) : m - n = 0 ∨ m ≠ n :=
begin
  cases decidable.em (m = n) with heq hne,
  { rw heq,
    left, exact nat.sub_self n },
  right, exact hne
end

```

Remember that if you open `classical`, you can use the law of the excluded middle for any proposition at all. But using type class inference (see [Chapter 10](#)), Lean can actually find the relevant decision procedure, which means that you can use the case split in a computable function.

```

def f (m k : ℕ) : ℕ :=
begin
  cases m - k, exact 3, exact 7
end

example : f 5 7 = 3 := rfl
example : f 10 2 = 7 := rfl

```

Aspects of computability will be discussed in a later chapter.

7.6 Other Inductive Types

Let us consider some more examples of inductively defined types. For any type, α , the type `list α` of lists of elements of α is defined in the library.

```

inductive list (α : Type u)
| nil {} : list
| cons : α → list → list

namespace list

variable {α : Type}

notation h :: t := cons h t

def append (s t : list α) : list α :=
list.rec t (λ x l u, x::u) s

notation s ++ t := append s t

theorem nil_append (t : list α) : nil ++ t = t := rfl

theorem cons_append (x : α) (s t : list α) : x::s ++ t = x::(s ++ t) := rfl

end list

```

A list of elements of type α is either the empty list, `nil`, or an element $h : \alpha$ followed by a list $t : \text{list } \alpha$. We define the notation $h :: t$ to represent the latter. The first element, h , is commonly known as the “head” of the list, and the remainder, t , is known as the “tail.” Recall that the notation `{}` in the definition of the inductive type ensures that the argument to `nil` is implicit. In most cases, it can be inferred from context. When it cannot, we have to write `@nil α` to specify the type α .

Lean allows us to define iterative notation for lists:

```
inductive list ( $\alpha : \text{Type } u$ )
| nil {} : list
| cons :  $\alpha \rightarrow \text{list} \rightarrow \text{list}$ 

namespace list

notation `[: foldr `[,` (h t, cons h t) nil) `[: := 1

section
  open nat
  check [1, 2, 3, 4, 5]
  check ([1, 2, 3, 4, 5] : list num)
end

end list
```

In the first check, Lean assumes that `[1, 2, 3, 4, 5]` is a list of natural numbers. The `(t : list num)` expression forces Lean to interpret `t` as a list of numerals.

As an exercise, prove the following:

```
theorem append_nil (t : list  $\alpha$ ) : t ++ nil = t := sorry

theorem append_assoc (r s t : list  $\alpha$ ) : r ++ s ++ t = r ++ (s ++ t) := sorry
```

Try also defining the function `length : $\Pi \alpha : \text{Type}, \text{list } \alpha \rightarrow \text{nat}$` that returns the length of a list, and prove that it behaves as expected (for example, `length (s ++ t) = length s + length t`).

For another example, we can define the type of binary trees:

```
inductive binary_tree
| leaf : binary_tree
| node : binary_tree  $\rightarrow$  binary_tree  $\rightarrow$  binary_tree
```

In fact, we can even define the type of countably branching trees:

```
inductive cbtree
| leaf : cbtree
| sup : ( $\mathbb{N} \rightarrow \text{cbtree}$ )  $\rightarrow$  cbtree
```

```
namespace cbtree

def succ (t : cbtree) : cbtree :=
  sup (λ n, t)

def omega : cbtree :=
  sup (λ n, nat.rec_on n leaf (λ n t, succ t))

end cbtree
```

Induction and Recursion

Other than the type universes and Pi types, inductively defined types provide the only means of defining new types in the Calculus of Inductive Constructions. We have also seen that, fundamentally, the constructors and the recursors provide the only means of defining functions on these types. By the propositions-as-types correspondence, this means that induction is the fundamental method of proof for these types.

Working with induction and recursion is therefore fundamental to working in the Calculus of Inductive Constructions. For that reason Lean provides more natural ways of defining recursive functions, performing pattern matching, and writing inductive proofs. Behind the scenes, these are “compiled” down to recursors.

Thus, the function definition package, which performs this reduction, is not part of the trusted code base.

8.1 Pattern Matching

The `cases_on` recursor can be used to define functions and prove theorems by cases. But complicated definitions may use several nested `cases_on` applications, and may be hard to read and understand. Pattern matching provides a more convenient and standard way of defining functions and proving theorems. Lean supports a very general form of pattern matching called *dependent pattern matching*.

A pattern-matching definition is of the following form:

```

definition [name] [parameters] : [domain] → [codomain]
| [patterns_1] := [value_1]
...
| [patterns_n] := [value_n]

```

The parameters are fixed, and each assignment defines the value of the function for a different case specified by the given pattern. As a first example, we define the function `sub2` for natural numbers:

```
open nat

def sub2 : nat → nat
| 0      := 0
| 1      := 0
| (a+2)  := a

example : sub2 5 = 3 := rfl
```

The default compilation method guarantees that the pattern matching equations hold definitionally.

```
example : sub2 0 = 0 := rfl

example : sub2 1 = 0 := rfl

example (a : nat) : sub2 (a + 2) = a := rfl
```

We can use the command `print sub2` to see how our definition was compiled into recursors.

```
print sub2
```

We will say a term is a *constructor application* if it is of the form `c a_1 ... a_n` where `c` is the constructor of some inductive data type. Note that in the definition `sub2`, the terms `1` and `a+2` are not constructor applications. However, the compiler normalizes them at compilation time, and obtains the constructor applications `succ zero` and `succ (succ a)` respectively. This normalization step is just a convenience that allows us to write definitions resembling the ones found in textbooks. There is no magic here: the compiler simply uses the kernel's ordinary evaluation mechanism. If we had written `2+a`, the definition would be rejected since `2+a` does not normalize into a constructor application.

In the next example, we use pattern-matching to define Boolean negation `bnot`, and proving `bnot (bnot b) = b`.

```
def bnot : bool → bool
| tt := ff
| ff := tt

theorem bnot_bnot : ∀ (b : bool), bnot (bnot b) = b
| tt := rfl    -- proof that bnot (bnot tt) = tt
| ff := rfl    -- proof that bnot (bnot ff) = ff
```

As described in [Chapter 7](#), Lean inductive data types can be parametric. The following example defines the `tail` function using pattern matching. The argument $\alpha : \text{Type}$ is a parameter and occurs before the colon to indicate it does not participate in the pattern matching. Lean allows parameters to occur after `:`, but it cannot pattern match on them.

```
open list

def tail1 {α : Type} : list α → list α
| nil      := nil
| (h :: t) := t

-- Parameter α may occur after ':'
def tail2 : Π {α : Type}, list α → list α
| α nil      := nil
| α (h :: t) := t
```

8.2 Structural Recursion and Induction

The function definition package supports structural recursion, that is, recursive applications where one of the arguments is a subterm of the corresponding term on the left-hand-side. Later, we describe how to compile recursive equations using well-founded recursion. The main advantage of the default compilation method is that the recursive equations hold definitionally.

Here are some examples from the last chapter, written in the new style:

```
def add : nat → nat → nat
| m zero      := m
| m (succ n) := succ (add m n)

local infix `+` := add

theorem add_zero (m : nat) : m + zero = m := rfl
theorem add_succ (m n : nat) : m + succ n = succ (m + n) := rfl

theorem zero_add : ∀ n, zero + n = n
| zero      := rfl
| (succ n) := congr_arg succ (zero_add n)

def mul : nat → nat → nat
| n zero      := zero
| n (succ m) := mul n m + m
```

The “definition” of `zero_add` makes it clear that proof by induction is really a form of induction in Lean.

As with definition by pattern matching, parameters to a structural recursion or induction may appear before the colon. Such parameters are simply added to the local context before the definition is processed. For example, the definition of addition may be written as follows:

```
def add (m : nat) : nat → nat
| zero   := m
| (succ n) := succ (add n)
```

This may seem a little odd, but you should read the definition as follows: “Fix m , and define the function which adds something to m recursively, as follows. To add zero, return m . To add the successor of n , first add n , and then take the successor.” The mechanism for adding parameters to the local context is what makes it possible to process match expressions within terms, as described below.

A more interesting example of structural recursion is given by the Fibonacci function `fib`.

```
def fib : nat → nat
| 0   := 1
| 1   := 1
| (a+2) := fib (a+1) + fib a

-- the defining equations hold definitionally
example : fib 0 = 1 := rfl
example : fib 1 = 1 := rfl
example (a : nat) : fib (a+2) = fib (a+1) + fib a := rfl
```

Another classic example is the list `append` function.

```
def append {α : Type} : list α → list α → list α
| []     l := l
| (h::t) l := h :: append t l

example : append [(1 : ℕ), 2, 3] [4, 5] = [1, 2, 3, 4, 5] := rfl
```

8.3 Dependent Pattern-Matching

All the examples we have seen so far can be easily written using `cases_on` and `rec_on`. However, this is not the case with indexed inductive families, such as `vector α n`. A lot of boilerplate code needs to be written to define very simple functions such as `map`, `zip`, and `unzip` using recursors.

To understand the difficulty, consider what it would take to define a function `tail` which takes a vector $v : \text{vector } \alpha \text{ (succ } n\text{)}$ and deletes the first element. A first thought might be to use the `cases_on` function:

```
open nat

inductive vector (α : Type) : nat → Type
| nil {} : vector 0
| cons   : Π {n}, α → vector n → vector (succ n)
```

```

open vector
local notation h :: t := cons h t

check @vector.cases_on
-- Π {α : Type}
-- {C : Π (a : ℕ), vector α a → Type}
-- {a : ℕ}
-- (n : vector α a),
-- (e1 : C 0 nil)
-- (e2 : Π {n : ℕ} (a : α) (a_1 : vector α n), C (succ n) (cons a a_1)),
-- C a n

```

But what value should we return in the `nil` case? Something funny is going on: if `v` has type `vector α (succ n)`, it *can't* be `nil`, but it is not clear how to tell that to `cases_on`.

One standard solution is to define an auxiliary function:

```

def tail_aux {α : Type} {n m : nat} (v : vector α m) :
  m = succ n → vector α n :=
vector.cases_on v
  (assume H : 0 = succ n, nat.no_confusion H)
  (take m (a : α) w : vector α m,
    assume H : succ m = succ n,
      nat.no_confusion H (λ H1 : m = n, eq.rec_on H1 w))

def tail {α : Type} {n : nat} (v : vector α (succ n)) : vector α n :=
tail_aux v rfl

```

In the `nil` case, `m` is instantiated to 0, and `no_confusion` makes use of the fact that `0 = succ n` cannot occur. Otherwise, `v` is of the form `a :: w`, and we can simply return `w`, after casting it from a vector of length `m` to a vector of length `n`.

The difficulty in defining `tail` is to maintain the relationships between the indices. The hypothesis `e : m = succ n` in `tail_aux` is used to “communicate” the relationship between `n` and the index associated with the minor premise. Moreover, the `zero = succ n` case is “unreachable,” and the canonical way to discard such a case is to use `no_confusion`.

The `tail` function is, however, easy to define using recursive equations, and the function definition package generates all the boilerplate code automatically for us.

Here are a number of examples:

```

def head {α : Type} : Π {n}, vector α (succ n) → α
| n (h :: t) := h

def tail {α : Type} : Π {n}, vector α (succ n) → vector α n
| n (h :: t) := t

lemma eta {α : Type} : ∀ {n} (v : vector α (succ n)), head v :: tail v = v
| n (h::t) := rfl

def map {α β γ : Type} (f : α → β → γ)
  : Π {n : nat}, vector α n → vector β n → vector γ n

```

```

| 0      nil      nil      := nil
| (succ n) (a::va) (b::vb) := f a b :: map va vb

def zip {α β : Type} : Π {n}, vector α n → vector β n → vector (α × β) n
| 0      nil nil      := nil
| (succ n) (a::va) (b::vb) := (a, b) :: zip va vb

```

Note that we can omit recursive equations for “unreachable” cases such as `head nil`. The automatically generated definitions for indexed families are far from straightforward. For example:

```

print map
print map._main

```

The `map` function is even more tedious to define by hand than the `tail` function. We encourage you to try it, using `rec_on`, `cases_on` and `no_confusion`.

8.4 Variations on Pattern Matching

We say that a set of recursive equations *overlaps* when there is an input that more than one left-hand-side can match. In the following definition the input `0 0` matches the left-hand-side of the first two equations. Should the function return `1` or `2`?

```

def f : nat → nat → nat
| 0      y      := 1
| x      0      := 2
| (x+1) (y+1) := 3

```

Overlapping patterns are often used to succinctly express complex patterns in data, and they are allowed in Lean. Lean handles the ambiguity by using the first applicable equation. In the example above, the following equations hold definitionally:

```

variables (a b : nat)

example : f 0      0      = 1 := rfl
example : f 0      (a+1) = 1 := rfl
example : f (a+1) 0      = 2 := rfl
example : f (a+1) (b+1) = 3 := rfl

```

Lean also supports *wildcard patterns*, also known as *anonymous variables*. They are used to create patterns where we don’t care about the value of a specific argument. In the function `f` defined above, the values of `x` and `y` are not used in the right-hand-side. Here is the same example using wildcards:

```

def f : nat → nat → nat
| 0 _ := 1
| _ 0 := 2
| _ _ := 3
variables (a b : nat)
example : f 0 0 = 1 := rfl
example : f 0 (a+1) = 1 := rfl
example : f (a+1) 0 = 2 := rfl
example : f (a+1) (b+1) = 3 := rfl

```

Some functional languages support *incomplete patterns*. In these languages, the interpreter produces an exception or returns an arbitrary value for incomplete cases. We can simulate the arbitrary value approach using the `inhabited` type class, discussed in [Chapter 10](#). Roughly, an element of `inhabited α` is simply a witness to the fact that there is an element of α ; in [Chapter 10](#) we will see that Lean can be instructed that suitable base types are inhabited, and can automatically infer that other constructed types are inhabited on that basis. On this basis, the standard library provides an arbitrary element, `arbitrary α` , of any inhabited type.

We can also use the type `option α` to simulate incomplete patterns. The idea is to return `some a` for the provided patterns, and use `none` for the incomplete cases. The following example demonstrates both approaches.

```

def f1 : nat → nat → nat
| 0 _ := 1
| _ 0 := 2
| _ _ := arbitrary nat -- the "incomplete" case

variables (a b : nat)

example : f1 0 0 = 1 := rfl
example : f1 0 (a+1) = 1 := rfl
example : f1 (a+1) 0 = 2 := rfl
example : f1 (a+1) (b+1) = arbitrary nat := rfl

def f2 : nat → nat → option nat
| 0 _ := some 1
| _ 0 := some 2
| _ _ := none -- the "incomplete" case

example : f2 0 0 = some 1 := rfl
example : f2 0 (a+1) = some 1 := rfl
example : f2 (a+1) 0 = some 2 := rfl
example : f2 (a+1) (b+1) = none := rfl

```

8.5 Inaccessible Terms

Sometimes an argument in a dependent matching pattern is not essential to the definition, but nonetheless has to be included to specialize the type of the expression appropriately.

Lean allows users to mark such subterms as *inaccessible* for pattern matching. These annotations are essential, for example, when a term occurring in the left-hand side is neither a variable nor a constructor application, because these are not suitable targets for pattern matching. We can view such inaccessible terms as “don’t care” components of the patterns. You can declare a subterm inaccessible by writing `.t`. If `t` is a composite term, we need to write `.(t)`.

The following example can be found in `cite{goguen:et:al:06}`. We declare an inductive type that defines the property of “being in the image of `f`”. You can view an element of the type `image_of f b` as evidence that `b` is in the image of `f`, whereby the constructor `imf` is used to build such evidence. We can then define any function `f` with an “inverse” which takes anything in the image of `f` to an element that is mapped to it. The typing rules forces us to write `f a` for the first argument, but this term is not a variable nor a constructor application, and plays no role in the pattern-matching definition. To define the function `inverse` below, we *have to* mark `f a` inaccessible.

```
variables {α β : Type}
inductive image_of (f : α → β) : β → Type
| imf : Π a, image_of (f a)

open image_of

def inverse {f : α → β} : Π b, image_of f b → α
| .(f a) (imf .f a) := a
```

In the example above, the inaccessible annotation makes it clear that `f` is *not* a pattern matching variable.

8.6 Match Expressions

Lean also provides a compiler for *match-with* expressions found in many functional languages. It uses essentially the same infrastructure used to compile recursive equations.

```
def is_not_zero (a : nat) : bool :=
match a with
| 0      := ff
| (n+1) := tt
end

-- We can use recursive equations and match
variable {α : Type}
variable p : α → bool

definition filter : list α → list α
| []      := []
| (a :: l) :=
  match p a with
  | tt := a :: filter l
```

```
| ff := filter l
end

example : filter is_not_zero [1, 0, 0, 3, 0] = [1, 3] := rfl
```

8.7 Well-Founded Recursion

[TODO: write this section.]

Structures and Records

We have seen that Lean’s foundational system includes inductive types. We have, moreover, noted that it is a remarkable fact that it is possible to construct a substantial edifice of mathematics based on nothing more than the type universes, Pi types, and inductive types; everything else follows from those. The Lean standard library contains many instances of inductive types (e.g., `nat`, `prod`, `list`), and even the logical connectives are defined using inductive types.

Remember that a non-recursive inductive type that contains only one constructor is called a *structure* or *record*. The product type is a structure, as is the dependent product type, that is, the Sigma type. In general, whenever we define a structure `S`, we usually define *projection* functions that allow us to “destruct” each instance of `S` and retrieve the values that are stored in its fields. The functions `prod.pr1` and `prod.pr2`, which return the first and second elements of a pair, are examples of such projections.

When writing programs or formalizing mathematics, it is not uncommon to define structures containing many fields. The `structure` command, available in Lean, provides infrastructure to support this process. When we define a structure using this command, Lean automatically generates all the projection functions. The `structure` command also allows us to define new structures based on previously defined ones. Moreover, Lean provides convenient notation for defining instances of a given structure.

9.1 Declaring Structures

The `structure` command is essentially a “front end” for defining inductive data types. Every `structure` declaration introduces a namespace with the same name. The general form is as follows:

```
structure <name> <parameters> <parent-structures> : Type :=
  <constructor> :: <fields>
```

Most parts are optional. Here is an example:

```
structure point (α : Type) :=
mk :: (x : α) (y : α)
```

Values of type `point` are created using `point.mk a b`, and the fields of a point `p` are accessed using `point.x p` and `point.y p`. The structure command also generates useful recursors and theorems. Here are some of the constructions generated for the declaration above.

```
check point           -- a Type
check point.rec_on    -- the recursor
check point.induction_on -- then recursor to Prop
check point.x         -- a projection / field accessor
check point.y         -- a projection / field accessor
```

You can obtain the complete list of generated constructions using the command `print prefix`.

```
print prefix point
```

Here are some simple theorems and expressions that use the generated constructions. As usual, you can avoid the prefix `point` by using the command `open point`.

```
eval point.x (point.mk 10 20)
eval point.y (point.mk 10 20)

open point

example (α : Type) (a b : α) : x (mk a b) = a :=
rfl

example (α : Type) (a b : α) : y (mk a b) = b :=
rfl
```

Given `p : point nat`, the notation `p^.x` is shorthand for `point.x p`. This provides a convenient way of accessing the fields of a structure.

```
def p := point.mk 10 20

check p^.x -- nat
eval p^.x -- 10
eval p^.y -- 20
```

If the constructor is not provided, then a constructor is named `mk` by default.

```
structure prod (α : Type) (β : Type) :=
  (pr1 : α) (pr2 : β)

check prod.mk
```

The keyword `record` is an alias for `structure`.

```
record point (α : Type) :=
  mk :: (x : α) (y : α)
```

You can provide universe levels explicitly. The annotations in the next example force the parameters α and β to be types from the same universe, and set the return type to also be in the same universe.

```
structure {u} prod (α : Type u) (β : Type u) : Type (max 1 u) :=
  (pr1 : α) (pr2 : β)

set_option pp.universes true
check prod.mk
```

The `set_option` command above instructs Lean to display the universe levels.

We use `max 1 1` as the resultant universe level to ensure the universe level is never 0 even when the parameter α and β are propositions. Recall that in Lean, `Type 0` is `Prop`, which is impredicative and proof irrelevant.

We can use the anonymous constructor notation to build structure values whenever the expected type is known.

```
structure {u} prod (α : Type u) (β : Type u) : Type (max 1 u) :=
  (pr1 : α) (pr2 : β)

example : prod nat nat :=
  ⟨1, 2⟩

check (⟨1, 2⟩ : prod nat nat)
```

9.2 Objects

We have been using constructors to create elements of a structure (or record) type. For structures containing many fields, this is often inconvenient, because we have to remember the order in which the fields were defined. Lean therefore provides the following alternative notations for defining elements of a structure type.

```
{ structure-name . <field-name> := <expr>)* }
or
{<field-name> := <expr>)*}
```

The prefix `structure-name .` can be omitted whenever the name of the structure can be inferred from the expected type. For example, we use this notation to define “points.” The order that the fields are specified does not matter, so all the expressions below define the same point.

```
structure point (α : Type) :=
mk :: (x : α) (y : α)

check { point . x := 10, y := 20 } -- point ℕ
check { point . y := 20, x := 10 }
check ({x := 10, y := 20} : point nat)

example : point nat :=
{ y := 20, x := 10 }
```

If the value of a field is not specified, Lean tries to infer it. If the unspecified fields cannot be inferred, Lean signs an error indicating the corresponding placeholder could not be synthesized.

```
structure my_struct :=
mk :: {α : Type} {β : Type} (a : α) (b : β)

check { my_struct . a := 10, b := true }
```

Record update is another common operation. It consists in creating a new record object by modifying the value of one or more fields. Lean provides a variation of the notation described above for record updates.

```
{ record-obj with <field-name> := <expr>)* }
```

The semantics is simple: record objects `<record-obj>` provide the values for the unspecified fields. If more than one record object is provided, then they are visited in order until Lean finds one that contains the unspecified field. Lean raises an error if any of the field names remain unspecified after all the objects are visited.

```
structure point (α : Type) :=
mk :: (x : α) (y : α)

def p : point nat :=
{x := 1, y := 2}

eval {p with y := 3}
eval {p with x := 3}
```

9.3 Inheritance

We can *extend* existing structures by adding new fields. This feature allow us to simulate a form of *inheritance*.

```

structure point (α : Type) :=
mk :: (x : α) (y : α)

inductive color
| red | green | blue

structure color_point (α : Type) extends point α :=
mk :: (c : color)

```

We can “rename” fields inherited from parent structures using the `renaming` clause.

```

structure prod (α : Type) (β : Type) :=
pair :: (pr1 : α) (pr2 : β)

-- Rename fields pr1 and pr2 to x and y respectively.
structure point3 (α : Type) extends prod α α renaming pr1→x pr2→y :=
mk :: (z : α)

check point3.x
check point3.y
check point3.z

```

In the next example, we define a structure using multiple inheritance, and then define an object using objects of the parent structures.

```

structure point (α : Type) :=
(x : α) (y : α) (z : α)

structure rgb_val :=
(red : nat) (green : nat) (blue : nat)

structure red_green_point (α : Type) extends point α, rgb_val :=
(no_blue : blue = 0)

definition p : point nat := {x := 10, y := 10, z := 20}
definition rgp : red_green_point nat :=
{p with red := 200, green := 40, blue := 0, no_blue := rfl}

example : rgp^.x = 10 := rfl
example : rgp^.red = 200 := rfl

```

Type Classes

We have seen that Lean’s elaborator provides helpful automation, filling in information that is tedious to enter by hand. In this section we will explore a simple but powerful technical device known as *type class inference*, which provides yet another mechanism for the elaborator to supply missing information.

The notion of a *type class* originated with the *Haskell* programming language. Many of the original uses carry over, but, as we will see, the realm of interactive theorem proving raises even more possibilities for their use.

10.1 Type Classes and Instances

Any family of types can be marked as a *type class*. Then we can declare particular elements of a type class to be *instances*. These provide hints to the elaborator: any time the elaborator is looking for an element of a type class, it can consult a table of declared instances to find a suitable element.

More precisely, there are three steps involved:

- First, we declare a family of inductive types to be a type class.
- Second, we declare instances of the type class.
- Finally, we mark some implicit arguments with square brackets instead of curly brackets, to inform the elaborator that these arguments should be inferred by the type class mechanism.

Here is a somewhat frivolous example:

```

attribute [class] nat

instance nat_one :  $\mathbb{N}$  := 1
/- The command instance is syntax sugar for
def nat_one :  $\mathbb{N}$  := 1
attribute [instance, reducible] nat_one
-/

def foo [x :  $\mathbb{N}$ ] : nat := x

check @foo
eval foo

example : foo = 1 := rfl

```

Here we declare `nat` to be a class with a “canonical” instance 1. Then we declare `foo` to be, essentially, the identity function on the natural numbers, but we mark the argument implicit, and indicate that it should be inferred by type class inference. When we write `foo`, the preprocessor interprets it as `foo ?x`, where `?x` is an implicit argument. But when the elaborator gets hold of the expression, it sees that `?x : \mathbb{N}` is supposed to be solved by type class inference. It looks for a suitable element of the class, and it finds the instance `one`. Thus, when we evaluate `foo`, we simply get 1.

It is tempting to think of `foo` as defined to be equal to 1, but that is misleading. Every time we write `foo`, the elaborator searches for a value. If we declare other instances of the class, that can change the value that is assigned to the implicit argument. This can result in seemingly paradoxical behavior. For example, we might continue the development above as follows:

```

instance nat_two :  $\mathbb{N}$  := 2

eval foo

example : foo  $\neq$  1 :=
 $\lambda$  h : 2 = 1, nat.no_confusion h ( $\lambda$  h : 1 = 0, nat.no_confusion h)

```

Now the “same” expression `foo` evaluates to 2. Whereas before we could prove `foo = 1`, now we can prove `foo \neq 1`, because the inferred implicit argument has changed. When searching for a suitable instance of a type class, the elaborator tries the most recent instance declaration first, by default. We will see below, however, that it is possible to give individual instances higher or lower priority.

As with other attributes, you can assign the `class` or `instance` attributes in a definition, or after the fact, with an `attribute` command. As usual, the assignments `attribute [class] foo` and `attribute [instance] foo`. To limit the scope of an assignment to the current file, use the `local attribute` variant.

The reason the example is frivolous is that there is rarely a need to “infer” a natural number; we can just hard-code the choice of 1 or 2 into the definition of `foo`. Type classes

become useful when they depend on parameters, in which case, the value that is inferred depends on these parameters.

Let us work through a simple example. Many theorems hold under the additional assumption that a type is inhabited, which is to say, it has at least one element. For example, if α is a type, $\exists x : \alpha, x = x$ is true only if α is inhabited. Similarly, it often happens that we would like a definition to return a default element in a “corner case.” For example, we would like the expression `head l` to be of type α when `l` is of type `list α` ; but then we are faced with the problem that `head l` needs to return an “arbitrary” element of α in the case where `l` is the empty list, `nil`.

For purposes like this, the standard library defines a type class `inhabited : Type → Type`, to enable type class inference to infer a “default” or “arbitrary” element of an inhabited type. We will carry out a similar development in the examples that follow, using a namespace `hide` to avoid conflicting with the definitions in the standard library.

Let us start with the first step of the program above, declaring an appropriate class:

```
class inhabited (α : Type) :=
  (value : α)
/- The command 'class' above is shorthand for

@[class] structure inhabited (α : Type) :=
  (value : α)
-/
```

An element of the class `inhabited α` is simply an expression of the form `inhabited.mk a`, for some element $a : \alpha$. The projection `inhabited.value` will allow us to “extract” such an element of α from an element of `inhabited α` .

The second step of the program is to populate the class with some instances:

```
instance Prop_inhabited : inhabited Prop :=
  inhabited.mk true

instance bool_inhabited : inhabited bool :=
  inhabited.mk tt

instance nat_inhabited : inhabited nat :=
  inhabited.mk 0

instance unit_inhabited : inhabited unit :=
  inhabited.mk ()
```

In the Lean standard library, we regularly use the anonymous constructor when defining instances. It is particularly useful when the class name is long.

```
instance Prop_inhabited : inhabited Prop :=
  ⟨true⟩

instance bool_inhabited : inhabited bool :=
```

```

<tt>

instance nat_inhabited : inhabited nat :=
<0>

instance unit_inhabited : inhabited unit :=
<()>

```

This arranges things so that when type class inference is asked to infer an element `?M : Prop`, it can find the element `true` to assign to `?M`, and similarly for the elements `tt`, `0`, and `()` of the types `bool`, `nat`, and `unit`, respectively.

The final step of the program is to define a function that infers an element `s : inhabited α` and puts it to good use. The following function simply extracts the corresponding element `a : α` :

```

definition default ( $\alpha$  : Type) [s : inhabited  $\alpha$ ] :  $\alpha$  :=
@inhabited.value  $\alpha$  s

```

This has the effect that given a type expression α , whenever we write `default α` , we are really writing `default α ?s`, leaving the elaborator to find a suitable value for the metavariable `?s`. When the elaborator succeeds in finding such a value, it has effectively produced an element of type α , as though by magic.

```

check default Prop    -- Prop
check default nat     -- N
check default bool    -- bool
check default unit    -- unit

```

In general, whenever we write `default α` , we are asking the elaborator to synthesize an element of type α .

Notice that we can “see” the value that is synthesized with `eval`:

```

eval default Prop    -- true
eval default nat     -- 0
eval default bool    -- tt
eval default unit    -- ()

```

Sometimes we want to think of the default element of a type as being an *arbitrary* element, whose specific value should not play a role in our proofs. For that purpose, we can write `arbitrary α` instead of `default α` . The definition of `arbitrary` is the same as that of `default`, but is marked `irreducible` to discourage the elaborator from unfolding it. This does not preclude proofs from making use of the value, however, so the use of `arbitrary` rather than `default` functions primarily to signal intent.

10.2 Chaining Instances

If that were the extent of type class inference, it would not be all the impressive; it would be simply a mechanism of storing a list of instances for the elaborator to find in a lookup table. What makes type class inference powerful is that one can *chain* instances. That is, an instance declaration can in turn depend on an implicit instance of a type class. This causes class inference to chain through instances recursively, backtracking when necessary, in a Prolog-like search.

For example, the following definition shows that if two types α and β are inhabited, then so is their product:

```
instance prod_inhabited {α β : Type} [inhabited α] [inhabited β]
  : inhabited (prod α β) :=
  ⟨(default α, default β)⟩
```

With this added to the earlier instance declarations, type class instance can infer, for example, a default element of `nat × bool × unit`:

```
check default (nat × bool)
eval default (nat × bool)
```

Given the expression `default (nat × bool)`, the elaborator is called on to infer an implicit argument `?M : inhabited (nat × bool)`. The instance `prod_inhabited` reduces this to inferring `?M1 : inhabited nat` and `?M2 : inhabited bool`. The first one is solved by the instance `nat_inhabited`. The second uses `bool_inhabited`.

Similarly, we can inhabit function spaces with suitable constant functions:

```
instance inhabited_fun (α : Type) {β : Type} [inhabited β] : inhabited (α → β) :=
  ⟨(λ a : α, default β)⟩

check default (nat → nat × bool)
eval default (nat → nat × bool)
```

In this case, type class inference finds the default element `λ (a : nat), (0, tt)`.

As an exercise, try defining default instances for other types, such as sum types and the list type.

10.3 Decidable Propositions

Let us consider another example of a type class defined in the standard library, namely the type class of `decidable` propositions. Roughly speaking, an element of `Prop` is said to be decidable if we can decide whether it is true or false. The distinction is only useful in constructive mathematics; classically, every proposition is decidable. Nonetheless, as

we will see, the implementation of the type class allows for a smooth transition between constructive and classical logic.

In the standard library, `decidable` is defined formally as follows:

```
class inductive decidable (p : Prop) : Type
| is_false : ¬p → decidable
| is_true  : p → decidable
```

Logically speaking, having an element `t : decidable p` is stronger than having an element `t : p ∨ ¬p`; it enables us to define values of an arbitrary type depending on the truth value of `p`. For example, for the expression `if p then a else b` to make sense, we need to know that `p` is decidable. That expression is syntactic sugar for `ite p a b`, where `ite` is defined as follows:

```
def ite (c : Prop) [d : decidable c] {α : Type} (t e : α) : α :=
decidable.rec_on d (λ hnc, e) (λ hc, t)
```

The standard library also contains a variant of `ite` called `dite`, the dependent if-then-else expression. It is defined as follows:

```
definition dite (c : Prop) [d : decidable c] {α : Type} (t : c → α) (e : ¬ c → α) : α :=
decidable.rec_on d (λ hnc : ¬ c, e hnc) (λ hc : c, t hc)
```

That is, in `dite c t e`, we can assume `hc : c` in the “then” branch, and `hnc : ¬ c` in the “else” branch. To make `dite` more convenient to use, Lean allows us to write `if h : c then t else e` instead of `dite c (λ h : c, t) (λ h : ¬ c, e)`.

In the standard library, we cannot prove that every proposition is decidable. But we can prove that *certain* propositions are decidable. For example, we can prove that basic operations like equality and comparisons on the natural numbers and the integers are decidable. Moreover, decidability is preserved under propositional connectives:

```
check @and.decidable
-- Π {p q : Prop} [hp : decidable p] [hq : decidable q], decidable (p ∧ q)

check @or.decidable
check @not.decidable
check @implies.decidable
```

Thus we can carry out definitions by cases on decidable predicates on the natural numbers:

```
open nat

definition step (a b x : ℕ) : ℕ :=
if x < a ∨ x > b then 0 else 1
```

```
set_option pp.implicit true
print definition step
```

Turning on implicit arguments shows that the elaborator has inferred the decidability of the proposition $x < a \vee x > b$, simply by applying appropriate instances.

With the classical axioms, we can prove that every proposition is decidable. When you import the classical axioms, then, `decidable p` has an instance for every `p`, and the elaborator infers that value quickly. Thus all theorems in the standard library that rely on decidability assumptions are freely available in the classical library.

10.4 Overloading with Type Classes

We now consider the application of type classes that motivates their use in functional programming languages like Haskell, namely, to overload notation in a principled way. In Lean, a symbol like `+` can be given entirely unrelated meanings, a phenomenon that is sometimes called “ad-hoc” overloading. Typically, however, we use the `+` symbol to denote a binary function from a type to itself, that is, a function of type $\alpha \rightarrow \alpha \rightarrow \alpha$ for some type α . We can use type classes to infer an appropriate addition function for suitable types α . We will see in the next section that this is especially useful for developing algebraic hierarchies of structures in a formal setting.

We can declare a type class `has_add α` as follows:

```
universe variables u

class has_add ( $\alpha$  : Type u) :=
  (add :  $\alpha \rightarrow \alpha \rightarrow \alpha$ )

def add { $\alpha$  : Type u} [has_add  $\alpha$ ] :  $\alpha \rightarrow \alpha \rightarrow \alpha$  := has_add.add

local notation a `+` b := add a b
```

The class `has_add α` is supposed to be inhabited exactly when there is an appropriate addition function for α . The `add` function is designed to find an instance of `has_add α` for the given type, α , and apply the corresponding binary addition function. The notation `a + b` thus refers to the addition that is appropriate to the type of `a` and `b`. We can then declare instances for `nat`, and `bool`:

```
instance nat_has_add : has_add nat :=
  ⟨nat.add⟩

instance bool_has_add : has_add bool :=
  ⟨bor⟩

check 2 + 2    -- nat
check tt + ff  -- bool
```

As with `inhabited` and `decidable`, the power of type class inference stems not only from the fact that the class enables the elaborator to look up appropriate instances, but also from the fact that it can chain instances to infer complex addition operations. For example, assuming that there are appropriate addition functions for types α and β , we can define addition on $\alpha \times \beta$ pointwise:

```
instance prod_has_add {α : Type u} {β : Type v} [has_add α] [has_add β] : has_add (α × β) :=
  ⟨λ ⟨a₁, b₁⟩ ⟨a₂, b₂⟩, ⟨a₁+a₂, b₁+b₂⟩⟩

check (1, 2) + (3, 4)    -- ℕ × ℕ
eval  (1, 2) + (3, 4)    -- ⟨4, 6⟩
```

We can similarly define pointwise addition of functions:

```
instance fun_has_add {α : Type u} {β : Type v} [has_add β] : has_add (α → β) :=
  ⟨λ f g x, f x + g x⟩

check (λ x : nat, 1) + (λ x, 2)    -- ℕ → ℕ
eval (λ x : nat, 1) + (λ x, 2)    -- λ (x : ℕ), 3
```

As an exercise, try defining instances of `has_add` for lists, and show that they have the work as expected.

10.5 Managing Type Class Inference

You can ask Lean for information about the classes and instances that are currently in scope:

```
print classes
print instances inhabited
```

At times, you may find that the type class inference fails to find an expected instance, or, worse, falls into an infinite loop and times out. To help debug in these situations, Lean enables you to request a trace of the search:

```
set_option trace.class_instances true
```

If you add this to your file in Emacs mode and use `C-c C-x` to run an independent Lean process on your file, the output buffer will show a trace every time the type class resolution procedure is subsequently triggered.

You can also limit the search depth (the default is 32):

```
set_option class.instance_max_depth 5
```

Remember also that in the Emacs Lean mode, tab completion works in `set_option`, to help you find suitable options.

As noted above, the type class instances in a given context represent a Prolog-like program, which gives rise to a backtracking search. Both the efficiency of the program and the solutions that are found can depend on the order in which the system tries the instance. Instances which are declared last are tried first. Moreover, if instances are declared in other modules, the order in which they are tried depends on the order in which namespaces are opened. Instances declared in namespaces which are opened later are tried earlier.

You can change the order that type classes instances are tried by assigning them a *priority*. When an instance is declared, it is assigned a priority value `std.priority.default`, defined to be 1000 in module `init.priority` in both the standard and hott libraries. You can assign other priorities when defining an instance, and you can later change the priority with the `attribute` command. The following example illustrates how this is done:

```
class foo :=
(a : nat) (b : nat)

@[priority std.priority.default+1]
instance i1 : foo :=
⟨1, 1⟩

instance i2 : foo :=
⟨2, 2⟩

example : foo.a = 1 := rfl

@[priority std.priority.default+20]
instance i3 : foo :=
⟨3, 3⟩

example : foo.a = 3 := rfl

attribute [instance, priority 10] i3

example : foo.a = 1 := rfl

attribute [instance, priority std.priority.default-10] i1

example : foo.a = 2 := rfl
```

Bibliography

- [1] Thierry Coquand and Gerard Huet. The calculus of constructions. *Inf. Comput.*, 76(2-3):95–120, February 1988.
- [2] Peter Dybjer. Inductive families. *Formal Asp. Comput.*, 6(4):440–465, 1994.
- [3] Frank Pfenning and Christine Paulin-Mohring. Inductively defined types in the calculus of constructions. In Michael G. Main, Austin Melton, Michael W. Mislove, and David A. Schmidt, editors, *Mathematical Foundations of Programming Semantics, 5th International Conference, Tulane University, New Orleans, Louisiana, USA, March 29 - April 1, 1989, Proceedings*, volume 442 of *Lecture Notes in Computer Science*, pages 209–228. Springer, 1989.